

BASIC computerspellen voor MSX-computers



M.Th.A.M. Vijftigschild

BASIC-computerspellen voor MSX-computers



M.Th.A.M. Vijftigschild

Kluwer Software-reeks

**BASIC-computerspellen
voor MSX-computers**

M.Th.A.M. Vijftigschild

BASIC-computerspellen voor MSX-computers



Kluwer Technische Boeken B.V. Deventer-Antwerpen

Omslag en illustraties: Wim Niessink

ISBN 90 201 2001 8
D/1987/0108/179
NUGI 434

© 1987 Kluwer Technische Boeken B.V. Deventer

1^o druk 1987

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar gemaakt door middel van druk, fotokopie, microfilm of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

No part of this book may be reproduced in any form, by print, photoprint, microfilm or any other means without written permission from the publisher.

Ondanks alle aan de samenstelling van de tekst bestede zorg, kan noch de redactie noch de uitgever aansprakelijkheid aanvaarden voor eventuele schade, die zou kunnen voortvloeien uit enige fout die in deze uitgave zou kunnen voorkomen.

Woord vooraf

In 1982 verscheen de eerste druk van het oorspronkelijke werk 'BASIC-computerspellen'. Het boek was in de daaraan voorafgaande jaren ontstaan en kon op de meeste microcomputers worden gebruikt. Hierbij moest de gebruiker uiteraard kleine aanpassingen maken.

Al snel bleek behoefte aan bewerkingen voor een aantal populaire computers. Dat heeft onder andere geresulteerd in een speciale versie van dit boek; alle listings zijn omgezet voor gebruik op MSX-computers.

Wanneer men de programmatuur van het originele werk met dit boek vergelijkt, valt op dat de structuur hetzelfde is gebleven. In alle programma's is nu echter kleur verwerkt en in de meeste programma's is tevens gebruik gemaakt van sprites en geluid. De overige aanpassingen hebben voornamelijk betrekking op de speciale mogelijkheden van MSX-computers, die in dit boek dan ook volledig worden uitgebuit.

Als men het boek doorbladert, zal het opvallen dat de meeste programma's volgens een zeer strakke modulaire structuur zijn opgebouwd. Voor deze opbouw is heel bewust gekozen. De bedoeling is niet alleen de gebruiker een aantal kant-en-klare spelprogramma's aan te bieden, maar hem tevens in staat te stellen op eenvoudige wijze wijzigingen en aanpassingen aan te brengen. Om de gebruiker op weg te helpen, zijn bij een aantal programmabeschrijvingen suggesties toegevoegd. Deze suggesties hebben tot doel de gebruiker te laten zien hoe hij of zij eventuele wijzigingen kan aanbrengen.

Bij de opzet van dit boek zijn we ervan uitgegaan dat een beeldscherm als uitvoereenheid wordt gebruikt en het toetsenbord als invoereenheid.

Sommige programma's worden ingeleid met een uitgebreid verhaal om de spelsituatie te verduidelijken.

In het eerste hoofdstuk wordt uitgelegd hoe de programma's die in dit boek staan, zijn ontwikkeld. Meestal wordt dit soort programma's recht toe recht aan geschreven, maar dat is in dit geval niet gebeurd. De gebruikte methoden zijn ook toepasbaar voor het ontwikkelen van andere applicatieprogramma-tuur.

Inhoud

1. Inleiding 9

2. Standaardspellen 11

- 2.1 Mijn computer 11
- 2.2 31 fiches 15
- 2.3 Dobbelsteen 20
- 2.4 Sul en Xantippe 24
- 2.5 Hoog-laag 30
- 2.6 Gokken 34
- 2.7 Letters 38
- 2.8 Galgje 43
- 2.9 Mastermind 50
- 2.10 Vier op een rij 56

3. Science fiction-spellen 63

- 3.1 Maanlander 63
- 3.2 Ruimtevaartuig 69
- 3.3 De blijde 78
- 3.4 De Compusaurus 85
- 3.5 De zandplaneet Silico 94
- 3.6 De slag bij de Shrixheuvels 102

4. Simulatiespellen 112

- 4.1 De kolenmijn 112
- 4.2 Boerderij Zeldenrust 122
- 4.3 De stadhouder van Flipflopland 132

1. Inleiding

Bij het bestuderen van de programma's in dit boek zal het opvallen dat de meeste hiervan volgens dezelfde structuur zijn opgebouwd. Deze methode heeft als groot voordeel dat elk programma in logische eenheden is op te delen. De verschillende eenheden zijn via een stuurmodule met elkaar verbonden. In de programmatuur zitten nog meer standaards ingebouwd, maar die zullen alleen bij nauwkeurige bestudering opvallen.

Om de actieve programmeur behulpzaam te zijn, zal wat dieper worden ingegaan op het ontwikkelen van een spel tot een computerprogramma.

Zoals al werd opgemerkt, zijn de meeste programma's in dit boek via een gestructureerd modulaire manier opgebouwd. Dit heeft tot gevolg dat de meeste programma's ook uiterlijk sterk op elkaar lijken. Dat laatste is ook de bedoeling. Als men de opbouw van één programma door heeft, kent men de opbouw van alle andere programma's. De programma's beginnen allemaal met één of meer NON-EXECUTABLE-statements. Dit zijn statements die voor de uitvoering van een BASIC-programma noodzakelijk zijn, maar waarvan de gebruiker niet merkt dat ze ook inderdaad worden uitgevoerd. Het zijn dus statements zoals DATA, DIM en RANDOM. Direct na deze statements volgt een GOTO naar een hoog regelnummer. Dit is meestal de enige GOTO die in het programma voorkomt. De spronginstructie zorgt ervoor dat de stuurmodule wordt gestart, deze houdt het gehele programma onder controle. Via subroutines worden de verschillende programmamodules aangeroepen. Het voordeel van deze methode is dat de noodzakelijke programmamodules onafhankelijk van elkaar kunnen worden ontwikkeld.

Elk spelprogramma bestaat uit een invoer-, bewerkings- en uitvoermodule. Deze modules zijn ook terug te vinden in de programma's, vaak met deze aanduidingen. Behalve de reeds genoemde modules komen ook de instructiemodule en de initialisatiemodule voor. De volgorde waarin deze modules in het programma komen te staan, is niet van belang, omdat de volgorde door de stuurmodule wordt bepaald.

Bij het bewerken van de programma's voor de MSX-computer bleek de modulaire opbouw opnieuw een groot voordeel. In principe kwam het er vaak op neer dat alleen de uitvoermodules herschreven hoefden te worden. Wanneer er van sprites en/of geluid gebruik werd gemaakt, werden aparte modules voor realisatie hiervan toegevoegd.

Een idee voor een spel wordt meestal eerst op papier uitgewerkt. Meestal kan men dan de spelregels al bepalen. Is men zover dan kan de instructiemodule met de spelregels worden gemaakt. Is de module ingetikt dan kan deze worden getest.

Zo, het eerste stukje programma is klaar. Waarschijnlijk heeft u al begrepen dat elke module als een programmaatje wordt beschouwd. Zo'n programmamodule heeft een logisch begin en een logisch eind. Bij de ontwikkeling van deze programmatuur is ervan uitgegaan dat de module wordt aangeroepen met het regelnummer waarop het eerste statement van de module staat. De module wordt bij het hoogste regelnummer van de module verlaten; dit is bij

de RETURN-opdracht. Op deze manier bereikt men dat er slechts één begin en één einde in de module is. Men voorkomt op deze manier de zogenaamde 'spaghetti-structuur' die helaas maar al te vaak in BASIC-programma's wordt aangetroffen. De bedoeling van al deze modules is te zorgen dat fouten in de programmatuur zeer snel kunnen worden opgelost. Het zal duidelijk zijn dat in het algemeen elke module maar één functie heeft.

De instructiemodule is al aan de orde geweest. Handig is het om daarna met de uitvoermodule te beginnen. De reden is dat men op die manier automatisch de variabelen kan bepalen die men in het programma nodig heeft. Tevens kan men dan bepalen welke variabele via welke module komt.

Nadat de uitvoermodule is ingevoerd, kan deze worden getest. Zeker wanneer er hele plaatjes moeten worden opgebouwd, zal het vrij veel testwerk geven om alles netjes op zijn plaats te krijgen. Werkt de module dan hoeft men er verder niet meer naar om te kijken.

De module die logisch volgt op de uitvoermodule is de invoermodule. In deze module worden de gegevens opgehaald die van het toetsenbord komen. Zou men een spel ontwerpen met joysticks dan haalt men de gegevens op van die joysticks. In deze module worden niet alleen de gegevens opgehaald, maar wordt tevens nagegaan of de waarden aan de gestelde eisen voldoen. Afhankelijk van het soort programma kan men als het fout gaat direct om nieuwe invoer vragen of men kan opnieuw een plaatje opbouwen en dan de invoermodule weer aanroepen. Dit hangt af van de methode waarop men de stuurmodule wil structureren. Een aantal variabelen dient in een spelprogramma een vaste startwaarde te krijgen, dit wordt gedaan in de initialisatiemodule. Deze module wordt slechts eenmaal aangeroepen aan het begin van het programma.

Er blijft nog één module over om te maken: de bewerkingsmodule. Deze vormt het hart van elk programma. Hierin worden de verschillende berekeningen uitgevoerd en de verschillende strings opgebouwd die in de uitvoermodule naar het beeldscherm worden gestuurd.

Als de bewerkingsmodule klaar is, kan er worden getest. Treden er fouten op dan zit de fout meestal in de berekeningsmodule.

Deze methode heeft behalve de overzichtelijkheid ook als voordeel dat er later vrij eenvoudig kleinere berekeningsmodules kunnen worden bijgevoegd. Deze kunnen dan óf via de stuurmodule óf via de grote berekeningsmodule worden aangeroepen.

Wanneer men later modules toevoegt, moet men zich wel aan de gestelde eisen houden.

Nu alle modules klaar zijn, moet de zaak nog aan elkaar worden geknoopt. Dit gebeurt via de stuurmodule. Om een inzicht te krijgen hoe zoiets eruit ziet, is het programma 'De zandplaneet Silico' een goed voorbeeld.

Het veelvuldig gebruik van REM-statements bevordert, zeker bij het ontwikkelen, de leesbaarheid van een programma. De programmatuur in dit boek is er dan ook rijkelijk mee voorzien. De gebruiker kan aan de hand van de REM-statements als direct een goed inzicht krijgen in de werking van een programma en wat de verschillende variabelen voorstellen. De REM-statements achter een BASIC-statement kunnen het beste worden weggelaten als men het programma overtuikt. Eventueel kunnen uiteraard alle REM-statements worden weggelaten. Laat men de regels die met een REM-statement beginnen weg dan wordt het echter wel erg ingewikkeld om de afbakening tussen de verschillende modules te vinden.

2. Standaardspellen

2.1 Mijn computer

Het programma 'Mijn computer' is een vrij eenvoudig demonstratieprogramma. Er wordt om de voornaam van de speler gevraagd. Met deze ingevoerde naam wordt een golvend display opgebouwd.

Na deze display wordt om de geboortedatum gevraagd. Deze geboortedatum wordt met de 'formule van Zeller' omgezet in de geboortedag.

Programma

```
10 REM ***** MIJN COMPUTER *****
20 REM *** SCHERM INITIALISATIE ***
30 REM *****
40 SCREEN 0 : 'SCHERM
50 WIDTH 40 : ' 40 COLOMEN
60 KEY OFF : ' FUNCTIETOETSEN
AFZETTEN
70 CLS : ' SCHERM LEEG
MAKEN
80 BEEP : ' PSG REGISTERS
WISSEN
90 GOSUB 1210 : ' OPENING
PROGRAMMA
100 COLOR 1,14,14
110 REM *** HOOFDPROGRAMMA ***
120 LOCATE 5,10
130 PRINT "GEEF EENS EEN GETAL ";:BEEP
140 INPUT A
150 CLS
160 TIME = 0
170 IF TIME < A THEN GOTO 170 ELSE IF A > 1 AND A < 15 THEN
COLOR,A,A
180 SCREEN 3:OPEN "GRP:" AS £ 1
190 PRESET(56,16):PRINT£1,"HALLO"
200 PRESET(36,72):PRINT£1,"IK BEN"
210 PRESET(80,120):PRINT£1,"MSX"
220 REM *** WACHTLUS ***
230 TIME = 0
240 IF TIME < 300 THEN GOTO 240
250 CLOSE:SCREEN 0:CLS
260 PRINT
270 PRINT " Ik ben de nieuwe MSX-computer van ...."
280 PRINT " Volgens ..... ben ik zeer knap."
290 PRINT " Af en toe laat ik wel eens een steek"
300 PRINT " vallen."
310 PRINT:PRINT
320 PRINT " HOE HEET JIJ";:BEEP
```



```

330 INPUT A$
340 CLS
350 PLAY"S1M200002","S1M200004","S1M200006"
360 FOR I = 1 TO 360 STEP 4
370 X% = ABS(INT(COS(I * 22/7/180)*30))+1
380 Y% = ABS(INT(SIN(I * 22/7/180)*30))+1
390 IF X% < Y% THEN 410
400 SWAP X%,Y%
410 PRINT TAB(X%);"HALLO ";A$
420 K = INT(RND(-TIME)*12)+2
430 IF (K=13 OR K=11) OR K=8 OR K=3 THEN COLOR K+1,K
440 IF I < 150 THEN PLAY"A","B","C"
450 IF I > 150 THEN PLAY"B","C","D"
460 NEXT I
470 PLAY"V15S1M200002AA.", "V12S1M200004BB.", "V10S1M200006CC."
"
480 PRINT
490 B$="ABCDEFGHIJKLMNOPQRSTUVWXYZ"
500 X = INT(RND(-TIME) * 26 + 1)
510 B$=LEFT$(B$,X)
520 B$=RIGHT$(B$,1)
530 C$=RIGHT$(A$,LEN(A$)-1)
540 C$=B$+C$
550 SCREEN 0:COLOR 1,11,12
560 PRINT
570 PRINT " HOE WAS JE NAAM OOK AL WEER ";C$;"?"
580 LOCATE 10,6:PRINT "INVOER > ";:BEEP
590 LINE INPUT B$
600 REM ** BIJZOEKEN VAN DE NAAM VAN DE DAG **
610 CLS
620 IF A$ = B$ THEN PRINT:PRINT:GOTO 660
630 PRINT:PRINT
640 PRINT " Je moet niet jokken ";A$;"!!!!"
650 PRINT " Ik heb je wel door."
660 PRINT " Ik maak ook weleens fouten, zoals"
670 PRINT " je ziet !!"
680 PRINT:PRINT
690 PRINT " Wanneer ben je geboren ";" ";B$;" " ?"
700 PRINT:PRINT " INVOER <D,M,J> ";:BEEP
710 INPUT D,M,J
720 H = 0
730 IF (D < 0) OR (D > 31) GOTO 760
740 IF (M < 1) OR (M > 12) GOTO 760
750 IF (J > 1900) AND (J < 2000) GOTO 880
760 H = 1
770 SCREEN 3:OPEN "GRP:" AS £ 1
780 FOR I = 1 TO 5
790 PRESET(40,56):PRINT £ 1, "EEND!!"
800 TIME = 0
810 IF TIME < 25 GOTO 810
820 CLS
830 TIME = 0
840 IF TIME < 15 GOTO 840
850 NEXT I
860 CLOSE:CLS:SCREEN 0
870 PRINT " DE DATUM DIE JE OPGEEFT KLOPT NIET!!"
880 IF H = 1 THEN GOTO 680

```


```

890 REM ** BEPALING DAG VAN DE WEEK **
900 IF M > 2 GOTO 930
910 J = J - 1
920 M = M + 12
930 E = INT(J/100)
940 Y = J - E * 100
950 A = INT(2.6001 * (M - 2) - .2) + D + Y + INT(Y/4) +
INT(E/4) - 2 * E
960 A = A - INT(A/7) * 7 + 1
970 RESTORE
980 FOR I = 1 TO A
990 READ B$
1000 NEXT I
1010 COLOR 1,2,2:CLS
1020 LOCATE 1,6
1030 PRINT"Volgens mij ben je geboren op een"
1040 PRINT" ";B$;"!!"
1050 PRINT:PRINT" KLOP HE!?" ";
1060 LINE INPUT A$
1070 IF LEFT$(A$,1) <> "N" AND LEFT$(A$,1) <> "n" GOTO
1130
1080 COLOR 15,6,6:LOCATE 1,12
1090 PRINT"VRAAG HET DAN MAAR EENS AAN JE MOEDER!"
1100 FOR M = 15 TO 1 STEP -1
1110 PLAY "V=M;06A","V=M;04B"
1120 NEXT M
1130 LOCATE 1,16
1140 PRINT"NOG VEEL PLEZIER MET DE REST VAN DE"
1150 PRINT" PROGRAMMA'S."
1160 END
1170 REM *** DAGEN VAN DE WEEK ***
1180 DATA Zondag, Maandag, Dinsdag, Woensdag, Donderdag
1190 DATA Vrijdag, zaterdag
1200 REM *** INTRODUCTIE SCHERM ***
1210 SCREEN 2:KM = 14:COLOR 15,4,4:CLS
1220 SC=14:X=140-10*SC:Y=90+SC*4:KO=KM:GOSUB1320
1230 PLAY"V12T180","V10T180","V9T180"
1240 PLAY"04C8E8G8R8GO5C04A2B-8A8G05C04AF2","03CB-8R8B-8R8FO
4C8R8C8R803CB-8R8B-8R8FO4C8R8C8R8","R03E8R8E8R8RA8R8A8R8RE8R
8E8R8RA8R8A8R8"
1250 PLAY"04C8E8G8R8GO5C04A2B-8A8G05C04AF2","03CB-8R8B-8R8FO
4C8R8C8R803CB-8R8B-8R8FO4C2","R03E8R8E8R8RA8R8A8R8RE8R8E8R8R
A2"
1260 PAINT(0,0),KM
1270 PLAY"05F8R8C2F8R8C2E8D8C2E8D8C04","03FO4C8R8C8R803FO4C8
R8C8R803GO4C8R8C8R803C04C8R8C8R8","R03A8R8A8R8RA8R8A8R8RB-8R
8B-8R8RB-8R8B-8R8"
1280 PLAY"A8R8AF8R8F8E8F8A8G8F8E8F8G8C8D8E8F8R8F8","03FO4C8R
8C8R803FO4C8R8C8R803CB-8R8B-8R8A8R8A8","FA8R8A8R8RA8R8A8R8RE
8R8E8R8F8R8F8"
1290 PAINT(0,180),KM
1300 PAINT(254,100),KM
1310 IF PLAY(0) = 0 THEN CLOSE#1:SCREEN 0:RETURN
1320 DRAW"S=SC;C=KO;BM=X; ,=Y;"
1330 DRAW"M+7,-26R6M+3,+12M+3,-12R6M+5,+20R13E1U2H1L7M-3,-1H
2M-1,-3U5M+1,-3E2M+3,-1R22M+4,+7M+4,-7R8M-8,+14M+8,+14L8M-4,
-7M-4,+7L8M+8,-14M-4,-7L17G1D2F1R7"

```

```
1340 DRAW"M+3,+1F2M+1,+3D5M-1,+3G2M-3,1L19M-3,-12M-3,+12L6M-  
3,-12M-3,+12L7U1"  
1350 RETURN
```

Voorbeeld



2.2 31 fiches

Het programma '31 fiches' is één van de kleinere programma's. Het is een leuk spel, zeker wanneer men niet doorheeft hoe het werkt. De werking van het programma is overigens vrij simpel.

De speler begint met fiches van de stapel te pakken, minimaal 1 en maximaal 4. De computer doet hetzelfde. Degene die de laatste fiche moet pakken, heeft verloren.

Het spel is zodanig in elkaar gezet dat de speler altijd verliest. Wanneer men het programma volgt, zal men er snel achter komen waarom.

Suggestie 1: Veel mensen zullen het niet leuk vinden dat de computer iedere keer wint. Daar is natuurlijk best wel wat aan te doen. Men zou bijvoorbeeld de speler óf de computer als eerste kunnen laten beginnen. De keuze kan men dan aan de speler overlaten of men kan via een willekeurig (random) getrokken getal bepalen wie er begint.

Suggestie 2: Men kan de basis van dit programma gebruiken door bijvoorbeeld twee muizen 31 stukjes kaas te laten opeten. Een muis van de computer en een muis van de speler. De muizen kunnen in dit geval door sprites worden uitgebeeld. De stukjes kaas zou u kunnen voorstellen door gebruik te maken van de grafische tekens met een gele achtergrondkleur.



Programma

```
10 REM *** 31 FICHES ***
20 CLEAR 5000
30 GR$=CHR$(1):R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR
$(89)
40 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86):R7$=GR
$+CHR$(81)
50 R8$=GR$+CHR$(82):R9$=CHR$(249)
60 S1$=CHR$(32)+R6$
70 S2$=R1$+R1$+R1$+R7$
80 S3$=R1$+R1$+R1$+R8$
90 S4$=R9$+R9$+R9$+R9$
100 SCREEN 0:COLOR 1,11,11:WIDTH 40:KEY OFF
110 GOTO 1200
120 :
130 REM *** SPELREGELS ***
140 DEFUSR=&H41:DEFUSR1=&H44:X=USR(0)
150 CLS
160 FOR X=1 TO 35:L$=L$+R1$:NEXT X
170 PRINT" ";R2$;L$;R3$
180 PRINT S1$;" *** 31 FICHES *** ";R6$
190 PRINT S1$;" - SPELREGELS - ";R6$
200 PRINT" ";R4$;L$;R5$
210 PRINT
220 PRINT" ";R2$;L$;R3$
230 PRINT S1$;SPACE$(35);R6$
240 PRINT S1$;" De speler en de computer mogen om ";R6$
250 PRINT S1$;" de beurt fiches van de stapel af ";R6$
260 PRINT S1$;" nemen. Maximaal zijn dit vier ";R6$
270 PRINT S1$;" fiches en minimaal is dat een ";R6$
280 PRINT S1$;" fiche. Degene die het laatste moet ";R6$
290 PRINT S1$;" pakken, heeft verloren. ";R6$
300 PRINT S1$;SPACE$(35);R6$
310 PRINT S1$;" De speler mag beginnen! ";R6$
320 PRINT S1$;SPACE$(35);R6$
330 PRINT" ";R4$;L$;R5$
340 PRINT:PRINT
350 X=USR1(0)
360 PRINT" ";R2$;L$;R3$
370 PRINT S1$;" Heeft u dit gelezen, druk dan ";R6$
380 PRINT S1$;" de <RETURN-TOETS> in. ";R6$
390 PRINT" ";R4$;L$;R5$
400 GOSUB 1450
410 I$=INKEY$
420 IF I$<>CHR$(13) GOTO 410
430 RETURN
440 :
450 REM *** INITIALIZEREN ***
460 A$=R2$+S3$+S3$+S3$+S3$+R1$+R1$+R1$+R3$
470 E$=R4$+S2$+S2$+S2$+S2$+R1$+R1$+R1$+R5$
480 C$=""
490 S=31 : 'AANTAL FICHES
500 H=3 : 'TEKST OVERSLAAN
510 S$ = R9$+" De speler "+R9$+R9$
520 GOSUB 570
530 H=0 : 'SPELER BEGINT
```

```

540 RETURN
550 :
560 REM *** UITVOER ***
570 COLOR 15,12,12:CLS :`SCHERM WISSEN
580 DEFUSR=&H41:DEFUSR1=&H44:X=USR(0)
590 LOCATE 1,0:PRINT STRING$(38,R9$)
600 LOCATE 1,1:PRINT S4$;" 31 FICHES ";S4$;S4$;R9$;R9$
610 LOCATE 25,1:PRINT S$
620 LOCATE 1,2:PRINT STRING$(38,R9$)
630 LOCATE 9,6:PRINT A$
640 FOR I=6 TO 0 STEP -1
650 LOCATE 9,13-I:PRINT R6$;
660 FOR J=1 TO 5
670 Y=I*5+J
680 B$=" "+R9$+" "
690 IF Y<=S GOTO 710
700 B$=" "
710 PRINT B$;R6$;
720 NEXT J
730 PRINT
740 NEXT I
750 LOCATE 9,14:PRINT E$
760 FOR I = 3 TO 21
770 LOCATE 1,I:PRINT R9$
780 LOCATE 38,I:PRINT R9$
790 NEXT I
800 LOCATE 1,22:PRINT STRING$(38,"")
810 X=USR1(0)
820 COLOR 15,4,4
830 IF H=3 GOTO 920
840 LOCATE 5,15:PRINT D$;"neemt ";K;"fiche(s). "
850 PLAY"V14T255A32B32"
860 IF H=2 GOTO 920
870 LOCATE 5,16:PRINT" Er zijn nog ";S;"fiches over."
880 IF H=0 GOTO 920
890 REM ** WACHTLUS **
900 TIME=0
910 IF TIME<150 GOTO 910
920 RETURN
930 :
940 REM *** ZET SPELER ***
950 LOCATE 3,20:PRINT"Hoeveel fiches gaan eraf 1-2-3-4:";
960 LINE INPUT K$:K$ = LEFT$(K$,1)
970 K = VAL(K$)
980 IF (K<1) OR (K>4) THEN LOCATE 34,21:PRINT" ":GOTO
950
990 K=INT(K)
1000 S=S-K
1010 H=1
1020 D$=" De speler "
1030 S$ = R9$+" De computer"+R9$
1040 RETURN
1050 :
1060 REM *** ZET COMPUTER ***
1070 K=5-K
1080 S=S-K
1090 H=0

```




```

1100 D$=" De computer "
1110 S$ = R9$+" De speler "+R9$+R9$
1120 IF S<>1 GOTO 1160
1130 H=2
1140 S$ = R9$+" DE COMPUTER"+R9$
1150 C$=" Ik heb gewonnen, er is nog een fiche!!"
1160 RETURN
1170 :
1180 REM **** STUURMODULE ****
1190 REM *** SPELREGELS ***
1200 GOSUB 130
1210 REM *** INITIALIZEREN ***
1220 GOSUB 460
1230 REM *** ZET SPELER ***
1240 GOSUB 950
1250 REM *** SPEELBORD ***
1260 GOSUB 570
1270 REM *** ZET COMPUTER ***
1280 GOSUB 1070
1290 REM *** SPEELBORD ***
1300 GOSUB 570
1310 IF H<>2 THEN 1240
1320 REM *** EIND ***
1330 COLOR 1,8,8
1340 FOR I=0 TO 10
1350 LOCATE 0,11:PRINT C$
1360 PLAY"V14T255A32B32V12C32D32V10E32F32V8G."
1370 GOSUB 1420
1380 LOCATE 1,11:PRINT STRING$(38,R9$)
1390 GOSUB 1420
1400 NEXT I
1410 END
1420 TIME=0
1430 IF TIME<20 GOTO 1430
1440 RETURN
1450 PLAY"V14T130","V10T130"
1460 PLAY"O5D8F+8ED8O4A8O5D8F+8ED8O4G8O5D8F+8ED8O4A8B8D8G2",
"O3RGO4D8R8D8R8O3GO4D8R8D8R8O3GO4DRO3G2","O4RRC8R8C8R8O3RB8R
8B8R8O4RC8R8C8R8R"
1470 RETURN

```


Voorbeeld



2.3 Dobbelsteen

Dit programma vormt een basis voor de ontwikkeling van andere bordspellen met de computer. Een voorbeeld daarvan is het programma 'Sul en Xantippe'. Het programma laat zien hoe met één tot maximaal vier dobbelstenen kan worden gegooid. Tevens geeft dit programma er een aardig beeld van hoe men met eenvoudige middelen een aardig plaatje kan opbouwen.

Suggestie: Probeer eens om de kleuren van de plaatjes te veranderen. Dit is namelijk een eenvoudige aanpassing waardoor het beeld al direct sterk verandert.

Programma

```
10 REM *** DOBBELSTEEN ***
20 CLEAR 5000
30 GR$=CHR$(1)
40 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
50 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
60 R9$=CHR$(249)
70 S4$=R9$+R9$+R9$+R9$
80 REM *** SCHERM INITIALIZATIE ***
90 SCREEN 3:OPEN"GRP:" AS£1:COLOR 4,5,5:CLS
100 PRESET(8,8):PRINT£1,"DOBBEL-"
110 PRESET(72,72):PRINT£1,"STEEN"
120 BEEP:GOSUB 1640
130 DIM A$(3),A(9,4)
140 KEY OFF:SCREEN 0:WIDTH 40:COLOR 1,2,2:CLS
150 GOTO 1290
160 :
170 REM *** INITIALIZEREN ***
180 H = 0
190 C$=R2$+R1$+R1$+R1$+R1$+R1$+R1$+R1$+R3$
200 C1$=R4$+R1$+R1$+R1$+R1$+R1$+R1$+R1$+R5$
210 C2$=R6$+"      "+R6$
220 RETURN
230 :
240 REM *** STRINGS WISSEN ***
250 FOR I=1 TO 3
260 A$(I) = ""
270 NEXT I
280 A$ = ""
290 A1$ = ""
300 A2$ = ""
310 S = 0           : 'SOM AANTAL OGEN
320 RETURN
330 :
340 REM *** INVOER ***
350 IF H > 0 THEN 480
360 H = 1
370 CLS
380 GOSUB 1450
390 LOCATE ,18
```

```

400 PRINT " Met hoeveel dobbelstenen wil je gooien"
410 PRINT " <maximaal 4> ";
420 INPUT X
430 IF (X <= 0) OR (X > 4) THEN BEEP:GOTO 370
440 IF X = 1 THEN SP = 15
450 IF X = 2 THEN SP = 10
460 IF X = 3 THEN SP = 6
470 IF X = 4 THEN SP = 1
480 LOCATE ,21
490 PRINT " Hoeveel keer moeten de dobbelstenen"
500 PRINT " rollen ";
510 PRINT " <STOP=0> ";
520 INPUT K
530 RETURN
540 :
550 REM *** ROLLEN VAN DOBBELSTEEN ***
560 FOR I = 1 TO ABS(K) :`BEVEILIGING
570 Y = RND(-TIME) :`ROLLEN
580 NEXT I
590 RETURN
600 :
610 REM *** UITVOER OPBOUWEN ***
620 FOR Y = 1 TO X
630 A$ = A$ + C$
640 A1$ = A1$ + C1$
650 A2$ = A2$ + C2$
660 FOR J = 1 TO 3
670 A$(J) = A$(J) + R6$
680 FOR I = (J - 1) * 3 + 1 TO (J - 1) * 3 + 3
690 B$ = " "
700 IF A(I,Y) <> 1 GOTO 710 ELSE B$ = " "+R9$
710 A$(J) = A$(J) + B$
720 NEXT I
730 A$(J) = A$(J) + " "+R6$
740 NEXT J
750 NEXT Y
760 RETURN
770 :
780 REM *** DOBBELSTENEN OPBOUWEN ***
790 FOR J = 1 TO X
800 FOR I = 1 TO 9
810 A(I,J) = 1
820 NEXT I
830 REM ** POSITIES WORDEN NOOIT GEBRUIKT **
840 A(4,J) = 0
850 A(6,J) = 0
860 REM ** DOBBELSTEEN ROLLEN
870 G = INT(RND(-TIME) * 6 + 1)
880 S = S + G :`SOMMEREN
AANTAL WOPREN
890 REM ** DOBBELSTEEN MAKEN **
900 IF G <> 6 GOTO 920
910 A(5,J) = 0
920 IF G > 5 GOTO 950
930 A(2,J) = 0
940 A(8,J) = 0
950 IF G <> 4 GOTO 970

```

```

960 A(5,J) = 0
970 IF G > 3 GOTO 1050
980 A(3,J) = 0
990 A(7,J) = 0
1000 IF G <> 2 GOTO 1020
1010 A(5,J) = 0
1020 IF G <> 1 GOTO 1050
1030 A(1,J) = 0
1040 A(9,J) = 0
1050 NEXT J
1060 GOSUB 1550
1070 RETURN
1080 :
1090 REM *** UITVOER ***
1100 CLS
1110 GOSUB 1450
1120 GOSUB 1510
1130 LOCATE SP,5
1140 PRINT A$
1150 FOR I = 1 TO 3
1160 PRINT SPACE$(SP);A$(I)
1170 IF I <= 2 THEN PRINT SPACE$(SP);A2$
1180 NEXT I
1190 PRINT SPACE$(SP);A1$
1200 LOCATE 1,14:
1210 PRINT"TOTAAL GEGOOID ";USING"££";S;
1220 PRINT" OGEN."
1230 IF K <> 0 GOTO 1250
1240 LOCATE 1,18:PRINT"DIT WAS LAATSTE WORP!!!"
1250 RETURN
1260 :
1270 REM **** STUURMODULE ****
1280 REM *** INITIALIZEREN ***
1290 GOSUB 160
1300 REM *** STRINGS WISSEN ***
1310 GOSUB 230
1320 REM *** INVOER ***
1330 GOSUB 330
1340 REM *** ROLLEN DOBBELSTEEN ***
1350 GOSUB 540
1360 REM *** DOBBELSTENEN OPBOUWEN ***
1370 GOSUB 770
1380 REM *** UITVOER OPBOUWEN ***
1390 GOSUB 600
1400 REM *** UITVOER ***
1410 IF H = 0 THEN GOSUB 1080 ELSE GOSUB 1120
1420 IF K <> 0 GOTO 1310
1430 END
1440 REM *** OPSCHRIFT ***
1450 FOR X=1 TO 36:S$=S$+R1$:NEXT X
1460 LOCATE 0,0:PRINT R2$+S$+R3$
1470 LOCATE 0,1:PRINT R6$;" ";S4$;R9$;R9$;" DOBBELSTEEN ";S4
$;S4$;S4$;R9$;R9$;R9$;" ";R6$
1480 LOCATE 0,2:PRINT R4$+S$+R5$
1490 RETURN
1500 REM *** GEDEELTE SCHERM SCHOON WISSEN ***
1510 FOR WI = 18 TO 22

```




```

1520 LOCATE 0,WI:PRINT SPACE$(39)
1530 NEXT WI
1540 RETURN
1550 REM *** GELUID VAN ROLLENDE DOBBELSTENEN ***
1560 SOUND 1,11
1570 SOUND 7,&B00111110
1580 FOR TE = 1 TO K
1590 FOR GE = 15 TO 0 STEP -2
1600 SOUND 8,GE
1610 NEXT GE
1620 NEXT TE
1630 RETURN
1640 PLAY"V14T150","V10T150","V10T150"
1650 T1$="O3RE8R8E8R8":T2$="O3RF8R8F8R8"
1660 PLAY"O4G2O5CE2DC2O4BO5CO4BA","O3CG8R8G8R8O2GO3G8R8G8R8C
A8R8A8R8CA8R8A8R8","XT1$;XT1$;XT2$;XT2$;"
1670 PLAY"O4G.R8GE2FG2.G2R","O3CG8R8G8R8O2GO3G8R8G8R8CG8R8G8
R8O2GO3G8R8G8R8","XT1$;XT1$;XT1$;XT1$;"
1680 PLAY"O4F.R8FD2EF2.F2R","O3DG8R8G8R8O2GO3G8R8G8R8DG8R8G8
R8O2GO3G8R8G8R8","XT2$;XT2$;XT2$;XT2$;"
1690 PLAY"O4G.R8GO5E2DC2.CRR","O3CG8R8G8R8O2GO3G8R8G8R8CEGCR
R","XT1$;XT2$;"
1700 RETURN

```

Voorbeeld



2.4 Sul en Xantippe

Sul en Xantippe is een oud dobbelspelletje. Het is de bedoeling om in een zo klein mogelijk aantal worpen een Sul en een Xantippe te gooien. Een Sul is een doublet; de twee dobbelstenen hebben een gelijk aantal ogen. Een Xantippe is een worp waarbij het totaal aantal ogen zeven bedraagt.

Het programma is gebaseerd op de random-generator die in de MSX-computer aanwezig is. Wanneer men het programma bekijkt, zal het opvallen dat bepaalde modules uit het programma 'Dobbelsteen' zijn gebruikt. Het blijkt dat wanneer men de modules goed ontwerpt, ze ook in andere programma's zijn te gebruiken. In dit boek zijn meer van dit soort voorbeelden te vinden.

Suggestie 1: Het programma gaat ervan uit dat men tegen de computer speelt. Het is echter eenvoudig aan te passen, zodat het met meer personen kan worden gespeeld.



Programma

```
10 REM *** SUL EN XANTIPPE ***
20 CLEAR 1000
30 GR$=CHR$(1)
40 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
50 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
60 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82):R9$=CHR$(249)
70 S1$=CHR$(32)+R6$:S2$=R1$+R1$+R1$+R7$
80 S3$=R1$+R1$+R1$+R8$
90 S4$=R9$+R9$+R9$+R9$
100 DIM A$(3),A(9,2),B(2),C(2,2),D(2)
110 REM *** SCHERM INITIALIZATIE ***
120 SCREEN 0:WIDTH 40:KEY OFF:COLOR 1,14,14:CLS
```

```

130 GOTO 1920
140 :
150 REM *** INITIALIZEREN ***
160 H = 1 : 'STUURVAR.
SPELER/COMP
170 FOR I=1 TO 2
180 B(I) = 0 : 'RESULTAAT
EEN WORP
190 C(I,1) = 0 : 'SUL
SCHOON
200 C(I,2) = 0 : 'XANTIPPE
SCHOON
210 D(I) = 0 : 'AANTAL
WORPEN
220 NEXT I
230 X = 2 : 'AANTAL
DOBBELSTENEN
240 C$ = " "+R2$+R1$+R1$+R1$+R1$+R1$+R1$+R1$+R3$
250 C1$= " "+R4$+R1$+R1$+R1$+R1$+R1$+R1$+R1$+R5$
260 C2$= S1$+" "+S1$
270 RETURN
280 :
290 REM *** SPELREGELS ***
300 CLS
310 LOCATE 0,0
320 GOSUB 1790
330 LOCATE 0,6
340 PRINT" Om de beurt gooit men met twee dobbel-"
350 PRINT" stenen. De speler begint en daarna"
360 PRINT" gooit de computer. Wie het eerst een"
370 PRINT" SUL en een XANTIPPE gooit, of omge-"
380 PRINT" keerd, heeft gewonnen.
390 PRINT
400 PRINT" Een SUL is een doublet d.w.z. twee"
410 PRINT" gelijke aantallen ogen."
420 PRINT
430 PRINT" Een XANTIPPE is een worp waarbij het"
440 PRINT" totaal aantal ogen gelijk aan 7 is."
450 LOCATE 0,20
460 PRINT R2$;:FOR RS=1 TO 37:PRINT R1$;:NEXT RS:PRINT R3$
470 PRINT R6$;" GELEZEN, DRUK DAN DE <RETURN> IN ";R6$
480 PRINT R4$;:FOR RS=1 TO 37:PRINT R1$;:NEXT RS:PRINT R5$
490 GOSUB 2240 : 'INTRO
MUZIEK
500 T$=INKEY$
510 IF T$=CHR$(13) GOTO 530
520 GOTO 500
530 COLOR 1,7,7:CLS
540 RETURN
550 :
560 REM *** STRINGS WISSEN ***
570 FOR I=1 TO 3
580 A$(I) = ""
590 NEXT I
600 A$ = ""
610 A1$ = ""
620 A2$ = ""

```



```

630 E$ = ""
640 F$ = ""
650 D(Q) = D(Q) + 1
: `WORPEN-TELLER
660 S = 0
AANTAL OGEN
670 RETURN
680 :
690 REM *** INVOER-SPELER ***
700 IF FLAG = 0 THEN GOSUB 1790
710 LOCATE 1,21
720 PRINT"Hoeveel keer moeten de dobbelstenen"
730 PRINT" rollen "
740 LOCATE 10,22:PRINT SPACE$(10)
750 PLAY"V1407LBL8D":LOCATE 9,22:INPUT K
760 RETURN
770 REM *** ROLLEN VAN DOBBELSTEEN ***
780 FOR I=1 TO ABS(K)
790 Y = RND(-TIME)
800 NEXT I
810 RETURN
820 :
830 REM *** UITVOER OPBOUWEN ***
840 REM ** MATRIX A$ OPBOUWEN **
850 FOR Y=1 TO X:
860 A$ = A$ + C$
870 A1$ = A1$ + C1$
880 A2$ = A2$ + C2$
890 FOR J=1 TO 3
900 A$(J) = A$(J) + S1$
910 FOR I=(J - 1) * 3 + 1 TO (J - 1) * 3 + 3
920 B$ = " "
930 IF A(I,Y) <> 1 GOTO 950 ELSE B$ = " " + R9$
940 :
950 A$(J) = A$(J) + B$
960 NEXT I
970 A$(J) = A$(J) + S1$
980 NEXT J
990 NEXT Y
1000 A$(2) = A$(2) + E$
1010 RETURN
1020 :
1030 REM *** DOBBELSTENEN OPBOUWEN ***
1040 FOR I=1 TO 9
1050 A(I,J) = 1
1060 NEXT I
1070 REM POSITIES WORDEN NOOIT GEBRUIKT
1080 A(4,J) = 0
1090 A(6,J) = 0
1100 REM ** DOBBELSTEEN ROLLEN **
1110 G = INT(RND(-TIME)*6 + 1)
1120 S = S + G
AANTAL WOPREN
1130 B(J) = G
OGEN ONTHOUDEN
1140 REM ** DOBBELSTEEN MAKEN **
1150 IF G <> 6 GOTO 1160 ELSE A(5,J) = 0

```

: `SOM

`LUS=AANTAL DOBBELST

: `SOMMEREN

: `AANTAL


```

1160 IF G > 5 GOTO 1170 ELSE A(2,J) = 0:A(8,J) = 0
1170 IF G <> 4 GOTO 1180 ELSE A(5,J) = 0
1180 IF G > 3 GOTO 1210 ELSE A(3,J) = 0:A(7,J) = 0
1190 IF G <> 2 GOTO 1200 ELSE A(5,J) = 0
1200 IF G <> 1 GOTO 1210 ELSE A(1,J) = 0:A(9,J) = 0
1210 RETURN
1220 :
1230 REM *** RESULTAAT ***
1240 REM ** AANWIJZING = E$ & F$ **
1250 IF (B(1) <> B(2)) AND (S <> 7) GOTO 1370
1260 E$ = " - SUL -"
1270 P = 1
1280 IF S <> 7 GOTO 1310
1290 E$ = " - XANTIPPE -"
1300 P = 2
1310 C(Q,P) = 1 : 'OPSLAAN RESULTAAT
1320 IF (C(Q,1) <> 1) OR (C(Q,2) <> 1) GOTO 1370
1330 F$ = " JIJ HEBT GEWONNEN, "
1340 IF Q <> 2 GOTO 1360
1350 F$ = " IK HEB GEWONNEN, "
1360 H = 2
1370 RETURN
1380 :
1390 REM *** UITVOER ***
1400 GOSUB 1790:VL = 1
1410 LOCATE 0,3:PRINT R2$;S5$;R3$
1420 LOCATE 0,4:PRINT R6$;" Aantal worpen speler      :";
R6$
1430 LOCATE 0,5:PRINT R6$;" Aantal worpen MSX-computer :";R
6$
1440 LOCATE 0,6:PRINT R4$;S5$;R5$
1450 LOCATE 31,4:PRINT SPACE$(3)
1460 LOCATE 31,5:PRINT SPACE$(3)
1470 LOCATE 31,4:PRINT USING"###";D(1)
1480 LOCATE 31,5:PRINT USING"###";D(2)
1490 GOSUB 1870
1500 LOCATE 3,8:PRINT A$
1510 FOR I = 1 TO 3
1520 PRINT SPACE$(3);A$(I)
1530 PRINT SPACE$(3);A2$
1540 NEXT I
1550 LOCATE 3,14:PRINT A1$
1560 LOCATE 8,16:PRINT SPACE$(12)
1570 LOCATE 1,16:PRINT"WORP : ";D$
1580 LOCATE 1,17:PRINT"Totaal gegooid";S;"ogen."
1590 IF E$ <> "" THEN GOSUB 2280
1600 LOCATE 1,19:PRINT F$;
1610 IF H <> 2 GOTO 1660
1620 PRINT"MET";D(Q);"WORPEN."
1630 LOCATE 0,21:PRINT SPACE$(39)
1640 LOCATE 0,22:PRINT SPACE$(37)
1650 LOCATE 37,20
1660 RETURN
1670 :
1680 REM *** BEURT ***
1690 REM ** AANWIJZING = D$ **
1700 H = H * (-1)

```

```

1710 Q = 1
1720 D$ = "speler"
1730 IF H < 0 GOTO 1760
1740 D$ = "computer"
1750 Q = 2
1760 RETURN
1770 :
1780 REM ** OPSCHRIFT **
1790 S5$=CHR$(0):FOR RS=1 TO 37:S5$=S5$+R1$:NEXT RS
1800 S$=LEFT$(S5$,33)
1810 LOCATE 0,0:PRINTR2$;S5$;R3$
1820 LOCATE 0,1:PRINT R6$;" --- SUL EN XANTIPPE ---SPELREGEL
S ";R6$
1830 IF VL = 1 THEN LOCATE 0,1:PRINT R6$;" --- SUL EN XANTIP
PE ----- ";R6$
1840 LOCATE 0,2:PRINTR4$;S5$;R5$
1850 RETURN
1860 REM ** GEDEELTE VAN SCHERM SCHOONWISSEN **
1870 FOR I = 9 TO 18
1880 LOCATE 0,I:PRINT SPACE$(39)
1890 NEXT I
1900 RETURN
1910 REM **** STUURMODULE ****
1920 REM *** SPELREGELS ***
1930 GOSUB 300
1940 REM *** INITIALIZEREN ***
1950 GOSUB 160
1960 REM *** SPEL ***
1970 REM *** BEURT ***
1980 GOSUB 1690
1990 REM *** STRING$ WISSEN ***
2000 GOSUB 570
2010 K = INT(RND(-TIME) * 50 + 1)           :`WILLEKEURIG GETAL
2020 IF Q <> 1 GOTO 2060
2030 REM *** INVOER SPELR ***
2040 GOSUB 700
2050 REM *** ROLLEN DOBBELSTENEN ***
2060 FOR J = 1 TO 2
2070 GOSUB 780
2080 REM *** DOBBELSTEEN OPBOUWEN ***
2090 GOSUB 1040
2100 NEXT J
2110 REM *** RESULTAAT ***
2120 GOSUB 1240
2130 REM *** UITVOER OPBOUWEN ***
2140 GOSUB 850
2150 REM *** UITVOER ***
2160 IF VL = 1 THEN GOSUB 1450 ELSE GOSUB 1400
2170 IF Q <> 1 GOTO 2210
2180 REM *** WACHTLUS ***
2190 T = TIME
2200 IF TIME < T + 200 GOTO 2200
2210 IF H <> 2 GOTO 1980
2220 END
2230 REM *** MUZIEK ***
2240 PLAY"V12T160","V10T160","V9T160"
2250 PLAY"O4G8B805D804B8G8B805D804B805D8C804A8F+8E2","O3G04D


```

```

03D04D03A04D03D04D", "RO3BRBRO4CRC"
2260 PLAY"05D8C804A8F+8E8F+8A805C8E8D804B8G8D2", "03A04D03D04D
03G04D03D04D", "RO4CRCRO3BRB"
2270 PLAY"V904G8B805D804B8G8B805D804B805E8D8C804B8A.A+8", "V70
3G04D03DGCGCG8C+8", "V603RBRFRERE-8R8"
2280 PLAY"04B8G8D8G8B8G8D8G8A8F+8D8F+8G2", "03DBDBD04CB2", "03R
GRGRAG2"
2290 RETURN

```

Voorbeeld



2.5 Hoog-laag

Bij dit spel speelt de computer tegen de speler en de speler tegen de computer. De speler wordt gevraagd een getal te kiezen tussen 0 en 100. De computer doet hetzelfde. De computer laat een getal zien en vraagt of dit het getal is dat de speler heeft gekozen. Klopt het dan geeft men 'juist'. Is het getal dat de speler heeft hoger, respectievelijk lager, dan geeft men 'hoger' respectievelijk 'lager'. Maakt de speler een fout dan heeft de computer dat uiteindelijk door; hij maakt de speler er dan op attent. Omgekeerd werkt het precies hetzelfde: de computer geeft de speler aanwijzingen of hij hoger of lager moet kiezen. Het programma maakt gebruik van de binaire zoekmethode.

Suggestie 1: Er wordt uitsluitend gespeeld met getallen van 0 tot en met 100. Door het omzetten van een variabele kan men ook met getallen van 0 tot en met 1000 of van 0 tot en met 10 spelen.

Suggestie 2: Dit spel is uitsluitend opgezet om met getallen te spelen. Het zou ook heel aantrekkelijk zijn om het spel met letters te spelen (1 tot en met 26). Men moet dan gebruik maken van de numerieke waarde van de letters. De wijzigingen die men moet aanbrengen, zijn heel beperkt. Men vraagt een karakter op via het toetsenbord, converteert het karakter naar een numerieke waarde en vergelijkt deze met de numerieke waarde van het karakter dat de computer 'in gedachten heeft'. Al naar gelang de uitslag geeft men 'hoger', 'lager' of 'juist'. De speler doet hetzelfde bij de suggesties van de computer.

Programma

```
10 REM *** HOOG - LAAG ***
20 REM *** SCHERM INITIALIZATIE ***
30 SCREEN 0:WIDTH 40:KEY OFF:COLOR 15,8,8:CLS:CLEAR 1000
40 GR$=CHR$(1)
50 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
60 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
70 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82):S1$=CHR$(32)+R6$
80 GOTO 1380
90 REM *** SPELREGELS ***
100 :
110 LOCATE 0,0
120 LY$="":FOR RS=1 TO 37:LY$=LY$+R1$:NEXT RS
130 PRINTR2$;LY$;R3$
140 PRINT R6$;"          --- HOOG - LAAG ---          ";R6$
150 PRINTR4$;LY$;R5$
160 LOCATE 0,6
170 PRINT" De computer en de speler nemen beide"
180 PRINT" een geheel getal tussen 1 en 100 in"
190 PRINT" gedachten. De speler begint met raden."
200 PRINT" Al naar gelang het resultaat geeft de"
210 PRINT" computer 'HOGER', 'LAGER' of 'JUIST'."
220 PRINT" De speler doet hetzelfde bij de"
230 PRINT" suggestie van de computer."
240 PRINT
```



```

250 PRINT" Degene, die als eerste het getal"
260 PRINT" raadt, heeft gewonnen."
270 LOCATE 0,20
280 PRINTR2$;LY$;R3$
290 PRINT R6$;" GEDAAN! DRUK DAN DE <RETURN> IN ";R6$
300 PRINTR4$;LY$;R5$
310 GOSUB 1620
320 A$=INKEY$
330 IF A$=CHR$(13) GOTO 350
340 GOTO 320
350 CLS :`SCHERM WISSEN
360 RETURN
370 :
380 REM *** INITIALIZEREN ***
390 N = 100 :`AANTAL GETALLEN
400 X1 = 1 :`ONDERGRENS
410 X2 = N + 1 :`BOVENGRENS
420 X3 = X1 + INT((X2-X1)/2):`EERSTE GOK
430 G = INT(RND(-TIME) * 100 + 1):`GETAL COMPUTER
440 S = 0 :`BEURTENTELLER SPELER
450 C = 0 :`BEURTENTELLER COMPUTER
460 E = 0 :`VORIGE GETAL COMPUTER
470 GOSUB 970
480 RETURN
490 :
500 REM *** INVOER SPELER GETAL ***
510 GOSUB 1280
520 GOSUB 1230
530 LOCATE 1,21
540 PRINT"Welke getal denk je dat ik heb";
550 BEEP:INPUT K
560 K = INT(K)
570 REM *** CONTROLE SPELER ***
580 REM ** AANWIJZING = B$ **
590 H = 0
600 B$ = "Jij moet hoger!"
610 IF K < G GOTO 630
620 B$ = "Jij moet lager!"
630 IF K <> G GOTO 660
640 B$ = "INDERDAAD DAT WAS HET GETAL!"
650 H = 2
660 RETURN
670 :
680 REM *** INVOER SPELER ANTWOORD ***
690 REM ** AANWIJZING = C$ **
700 H = 0
710 PRINT " <HOGER, LAGER of JUIST>? "
720 LOCATE 26,14:PRINT SPACE$(6)
730 LOCATE 26,14:PLAY"O8CO7C":LINE INPUT Q$
740 Q$ = LEFT$(Q$,1) :`ALLEEN EERSTE KAR.
750 IF Q$ <> "H" AND Q$ <> "h" GOTO 780
760 X1 = X3
770 H = 1
780 IF Q$ <> "L" AND Q$ <> "l" GOTO 810
790 X2 = X3
800 H = 1
810 IF Q$ <> "J" AND Q$ <> "j" GOTO 840

```

```

820 H = 2
830 C$ = " GOED HE!!"
840 IF H <> 0 GOTO 860
850 PRINT:C$ = " Ik begrijp je niet, het moet: <HOGER, LAGER
  of JUIST> zijn."
860 IF H = 2 GOTO 940
870 REM ** BEREKEN GETAL **
880 REM ** AANWIJZING = D$ **
890 X3 = X1 + INT((X2-X1)/2)
900 IF E <> X3 GOTO 930
910 D$ = "Jij weet niet wat hoger en lager is!!"
920 H = 2
930 E = X3
940 RETURN
950 :
960 REM *** STRINGS WISSEN ***
970 B$ = ""
980 C$ = ""
990 D$ = ""
1000 E$ = ""
1010 S = S + 1
1020 RETURN
1030 :
1040 REM *** UITVOER COMPUTER ***
1050 GOSUB 1330
1060 LOCATE 0,0
1070 GOSUB 1280
1080 PRINT
1090 PRINT" Jij koos";K
1100 LOCATE 1,5
1110 PRINT B$
1120 IF H = 2 GOTO 1160
1130 LOCATE 1,12
1140 PRINT"Is jouw getal soms:";X3
1150 PRINT
1160 RETURN
1170 :
1180 REM *** UITVOER SPELER ***
1190 LOCATE 5,17:PRINT C$
1200 LOCATE 1,12:PRINT D$
1210 RETURN
1220 REM *** INPUT SPELER ***
1230 LOCATE 1,20
1240 PRINT LY$
1250 LOCATE 1,22
1260 PRINT LY$
1270 RETURN
1280 LOCATE 0,0:PRINT"X";LY$;"Y"
1290 PRINT R6$;" --- HOOG - LAAG ---          ";R6$
1300 PRINT"Z";LY$;"["
1310 LOCATE 26,1:PRINT"BEURT:";S
1320 RETURN
1330 FOR SS=3 TO 19
1340 LOCATE,SS:PRINT SPACE$(39)
1350 NEXT SS
1360 LOCATE 32,21:PRINT SPACE$(3)
1370 RETURN


```

```

1380 REM **** STUURMODULE ****
1390 REM *** SPELREGELS ***
1400 GOSUB 100
1410 REM *** INITIALIZEREN ***
1420 GOSUB 390
1430 REM *** INVOER SPELER GETAL ***
1440 GOSUB 510
1450 REM *** UITVOER ***
1460 GOSUB 1050
1470 IF H = 2 GOTO 1560
1480 REM *** INVOER SPELER ANTWOORD ***
1490 GOSUB 700
1500 REM *** UITVOER SPELER ***
1510 GOSUB 1190
1520 REM *** STRINGS WISSEN ***
1530 GOSUB 970
1540 IF H < 2 GOTO 1440
1550 REM *** EINDE ***
1560 LOCATE 1,10:PRINT"MIJN GETAL WAS :";G
1570 GOSUB 1620
1580 COLOR 1,5,5
1590 LOCATE 0,22
1600 END
1610 REM *** MUZIEK ***
1620 T1$="S9M7000T130"
1630 PLAY "XT1$;O4F8G8A8B-8O5CO4A05CO4AA8A8G8G8FR", "XT1$;R2O
3F8G8A8B-8O4C8D8E8F8CO3B-AR"
1640 RETURN

```

Voorbeeld



2.6 Gokken

In dit spel geeft het programma een getal tussen 0 en 100. De speler wordt gevraagd of het volgende getal hoger of lager is dan het vorige. Na beantwoording kan men inzetten op zijn gok.

Is de gok juist dan wordt de inzet verdubbeld, was het fout dan is men zijn inzet kwijt. Het is alleen jammer dat de machine de winst niet in geld uitkeert.

Suggestie: Dit programma is zó gemaakt dat het slechts door één persoon kan worden gespeeld. Vaak is het veel leuker wanneer meer mensen mee kunnen spelen. Men zou het programma zodanig kunnen wijzigen dat dit kan. Men moet dan wel een vastgesteld aantal ronden spelen. Aan het eind van het spel kan de winnaar worden bepaald.



Programma

```
10 REM *** GOKKEN ***
20 SCREEN 0:WIDTH 40:KEY OFF:COLOR 15,12,12:CLS:CLEAR 1000
30 GR$=CHR$(1)
40 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
50 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
60 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
70 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
80 GOTO 1280
90 :
100 REM *** SPELREGELS ***
110 CLS : 'SCHERM WISSEN
120 LOCATE 0,0
130 PRINT R2$;LEFT$(S1$,44);R8$;LEFT$(S1$,28);R3$
140 PRINT R6$;" <<< GOKKEN >>> " ;R6$;" -SPELREGELS- ";
R6$
150 PRINT R4$;LEFT$(S1$,44);R7$;LEFT$(S1$,28);R5$
```



```

160 PRINT:PRINT
170 PRINT" De bedoeling van dit spel is te raden"
180 PRINT" of het volgende te trekken getal HOGER"
190 PRINT" of LAGER is dan het getrokken getal."
200 PRINT" Heeft men goed gegokt dan wordt de in-"
210 PRINT" zet verdubbeld. In het andere geval is"
220 PRINT" men zijn inzet kwijt."
230 PRINT" Blijkt het volgende getal echter ge-"
240 PRINT" lijk te zijn dan is men eveneens zijn"
250 PRINT" inzet kwijt."
260 PRINT
270 PRINT" De getallen, die getrokken worden,"
280 PRINT" gen tussen 1 en 100."
290 PRINT" Inzetten kan men doen tot maximaal:"
300 PRINT" Honderd gulden."
310 PRINT
320 PRINT R2$;LEFT$(S1$,74);R3$
330 PRINT R6$;" GELEZEN DRUK DAN <RETURN-TOETS> IN ";R6$
340 PRINTR4$;LEFT$(S1$,74);R5$
350 A$=INKEY$
360 IF A$=CHR$(13) THEN RETURN
370 GOTO 350
380 :
390 REM *** INITIALIZEREN ***
400 S = 0
410 G = INT(RND(-TIME) * 100 + 1)
420 P = 0
430 T = 0
440 SG = 3
450 RETURN
460 :
470 REM *** UITVOER ***
480 GOSUB 1130 :`SCHERM SCHOON
490 GOSUB 1210 :`TITEL
500 LOCATE 1,4:PRINT"TOTALE WINST:";S
510 LOCATE 20,4:PRINT"TOTAAL INGEZET:";T
520 PRINT
530 IF T = 0 GOTO 620
540 LOCATE ,6
550 PRINT" HET EERSTE GETAL WAS :";X
560 PRINT" HET TWEEDE GETAL WAS :";Q
570 PRINT
580 PRINT" Jij gokte ";A$;" en het is ";C$
590 PRINT D$;Z
600 PRINT
610 PRINT STRING$(39,"")
620 PRINT" Het eerste getal is :";G
630 PRINT
640 RETURN
650 :
660 REM *** INVOER ***
670 PRINT" Is het volgende getal HOGER of LAGER?"
680 LOCATE 3,21
690 PRINT" <HOGER, LAGER, STOP>";
700 PLAY"V12L6407C06G07C":INPUT A$
710 P = 0
720 A$ = LEFT$(A$,1)

```


```

730 IF A$ <> "S" AND A$ <> "s" GOTO 750
740 P = 3
750 IF A$ <> "H" AND A$ <> "h" GOTO 780
760 A$ = "HOGER"
770 P = 1
780 IF A$ <> "L" AND A$ <> "l" GOTO 810
790 A$ = "LAGER"
800 P = 2
810 IF P = 0 GOTO 680
820 IF P = 3 GOTO 890
830 REM ** INZET **
840 LOCATE 0,18
850 PRINT" Hoeveel zet U in";
860 BEEP:INPUT Z
870 IF ABS(Z) > 100 GOTO 840
880 Z = ABS(Z)
890 RETURN
900 :
910 REM *** BEREKENING ***
920 H = 0
930 Q = INT(RND(-TIME) * 100 + 1)
940 X = G
950 C$ = "GELIJK"
960 IF G < Q GOTO 1010
970 REM ** TWEEDE WAARDE IS KLEINER **
980 C$ = "LAGER"
990 IF P <> 2 GOTO 1060
1000 H = 1
1010 IF G > Q GOTO 1060
1020 REM ** TWEEDE WAARDE IS GROTER **
1030 C$ = "HOGER"
1040 IF P <> 1 GOTO 1060
1050 H = 1
1060 D$ = " Je bent je inzet kwijt van : "
1070 IF H <> 1 GOTO 1100
1080 D$ = " Je verdubbelde je inzet van : "
1090 S = S + 2 * Z
1100 T = T + Z
1110 G = INT(RND(-TIME) * 100 + 1)
1120 RETURN
1130 REM *** SCHERMSCHOON ***
1140 FOR SS = SG TO 19
1150 LOCATE 0,SS:PRINT SPACE$(39)
1160 NEXT SS
1170 LOCATE 1,21
1180 PRINT SPACE$(37)
1190 RETURN
1200 REM *** OPBOUW TITEL ***
1210 LOCATE 0,0
1220 PRINT R2$;LEFT$(S1$,74);R3$
1230 PRINT R6$;" LAGER <<< GOKKEN >>> HOGER ";R6$
1240 PRINT R4$;LEFT$(S1$,74);R5$
1250 RETURN
1260 REM **** STUURMODULE ****
1270 REM *** SPELREGELS ***
1280 GOSUB 110
1290 REM *** INITIALIZEREN ***

```

```
1300 GOSUB 400
1310 REM *** BEREKENING
1320 GOSUB 910
1330 REM *** UITVOER ***
1340 GOSUB 470
1350 REM *** INVOER ***
1360 GOSUB 660
1370 IF P <> 3 GOTO 1320
1380 SG = 13:GOSUB 1130
1390 LOCATE 2,21
1400 PRINT"BEDANKT VOOR HET SPEL, TOT ZIENS."
1410 GOTO 1410
```

Voorbeeld



2.7 Letters

Letters is een programma dat men onder de leerprogramma's zou kunnen rangschikken. De werking van het programma is vrij eenvoudig. Met behulp van een subroutine worden één of meer willekeurige letters op het beeldscherm geplaatst. Afhankelijk van de tijd blijven deze letters op het beeldscherm staan. Na het verstrijken van de tijd vraagt het programma welke letters op het beeldscherm stonden. Men moet dan de letters die men heeft gezien in de juiste volgorde weer intikken. Het programma geeft aan of men het goed of fout heeft gedaan en wat het had moeten zijn. Heeft men het goed gedaan dan zal het aantal letters met één worden uitgebreid, maar de tijd wordt met een derde verkort. Was het antwoord fout dan wordt het aantal letters niet met één vermeerderd en de tijd wordt met een derde gedeelte verlengd. Het record is op dit moment twaalf letters. Het is opmerkelijk dat kinderen bij dit spelletje beter scoren dan volwassenen.

Suggestie 1: Dit spel wordt uitsluitend met letters gespeeld. Het is een bekend gegeven dat getallen gemakkelijker worden onthouden. Met dit gegeven in het achterhoofd kan men spelniveaus in het programma aanbrengen. Het laagste spelniveau werkt alleen met getallen. Het tweede spelniveau uitsluitend met karakters. Het hoogste spelniveau is een combinatie van letters en karakters.

Suggestie 2: In suggestie 1 kan de speler zelf het spelniveau kiezen. Het is best leuk dit spelniveau door het programma te laten bepalen. Het programma begint bij het laagste niveau, wanneer blijkt dat de speler weinig fouten maakt wordt overgeschakeld naar spelniveau 2. Gaat dit ook weer goed dan kan worden overgeschakeld naar spelniveau 3.

Suggestie 3: Bij die ontwikkeling van dit spel is ervan uitgegaan dat men aléén speelt. Men zou ook een zeker competitie-element kunnen toevoegen door te vragen met hoeveel mensen men speelt en de scores bij te houden. Er moet dan door het programma wel een maximaal aantal beurten worden vastgesteld. Bijvoorbeeld tien beurten per speler. Is de competitie voorbij dan bepaalt het programma de winnaar.

Programma

```
10 REM *** LETTERS ***
20 SCREEN 0:WIDTH 40:KEY OFF:COLOR 5,1,1:CLS:CLEAR 1000
30 GR$=CHR$(1)
40 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
50 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
60 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
70 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
80 GOTO 1350
90 :
100 REM *** SPELREGELS ***
110 CLS : 'SCHERM WISSEN
120 LOCATE 0,0
```



```

130 PRINT R2$;LEFT$(S1$,38);R8$;LEFT$(S1$,34);R3$
140 PRINT R6$;" XXX LETTERS XXX ";R6$;" -SPELREGELS- ";R
6$
150 PRINT R4$;LEFT$(S1$,38);R7$;LEFT$(S1$,34);R5$
160 LOCATE 0,6
170 PRINT" De computer geeft een of meer letters"
180 PRINT" op. Deze letters blijven gedurende"
190 PRINT" een korte tijd op de scherm staan;"
200 PRINT" daarna verdwijnen ze plotseling."
210 PRINT
220 PRINT" Jij moet dan proberen de letters in de"
230 PRINT" goede volgorde weer in te tikken."
240 PRINT
250 PRINT" Vergeet na de letters niet <RETURN> in"
260 PRINT" te drukken."
270 LOCATE 0,20
280 PRINT R2$;LEFT$(S1$,74);R3$
290 PRINT R6$;" GELEZEN, DRUK DAN <RETURN-TOETS> IN ";R6$
300 PRINT R4$;LEFT$(S1$,74);R5$
310 GOSUB 1540
320 K$ = INKEY$
330 IF K$ = CHR$(13) THEN RETURN
340 GOTO 320
350 :
360 REM *** INITIALIZEREN ***
370 T = 200 :`TIJDSVERTRAGING
380 Q = 1 :`AANTAL LETTERS
390 V = 0 :`BEURT
400 G = 0 :`AANTAL GOEDEN
410 A$ = "ABCDEFGHIJKLMNOPQRSTUVWXYZ"
420 RETURN
430 :
440 REM *** LETTERS TREKKEN ***
450 V$ = "" :`VRAAGSTRING SCHOON
460 V = V + 1 :`BEURT + 1
470 FOR I=1 TO Q
480 L = INT(RND(-TIME) * 26 + 1)
490 B$ = LEFT$(A$,L)
500 B$ = RIGHT$(B$,1)
510 V$ = V$ + B$ :`CONCATENATIE
520 NEXT I
530 RETURN
540 :
550 REM *** RESULTAAT ***
560 H = 1
570 T1 = INT(T/4) :`TIJD + OF -
580 R$ = " JAMMER!! HET IS FOUT."
590 IF V$ <> Q$ GOTO 640
600 H = -1
610 Q = Q + 1 :`EEN LETTER ERBIJ
620 G = G + 1 :`AANTAL GOEDEN + 1
630 R$ = " JE ANTWOORD WAS GOED!!"
640 T = T + T1 * H
650 RETURN
660 :
670 REM *** UITVOER-KOP ***

```

```

680 GOSUB 1160
690 :
700 RETURN
710 :
720 REM *** UITVOER-RESULTAAT ***
730 GOSUB 680
740 LOCATE 0,10
750 PRINT" De vraag was      :";V$
760 PRINT" Het antwoord was  :";Q$
770 PRINT
780 PRINT R$
790 LOCATE 4,21
800 PRINT"WIL JE DOORGAAN JA/NEE :";
810 PLAY"V15L32O2GO4CO2G":INPUT Q$
820 RETURN
830 :
840 REM *** UITVOER-VRAAG ***
850 GOSUB 680
860 LOCATE 2,4
870 PRINT"De letters zijn:"
880 LOCATE 2,6:PRINT V$
890 T2 = TIME
900 IF TIME < T2 + T GOTO 900
910 LOCATE 2,6:PRINT SPACE$(16)
920 RETURN
930 :
940 REM *** INVOER ***
950 LOCATE 1,10
960 PRINT"Welke letters waren het :";
970 PLAY"V12LO5CL64O6CO4DO6C":INPUT Q$
980 :
990 RETURN
1000 :
1010 REM *** STOP ***
1020 GOSUB 680
1030 X = INT(G/V * 1000)/10
1040 LOCATE 0,10
1050 PRINT" JE BENT TOT";Q;"LETTERS GEKOMEN"
1060 PRINT" IN";V;"BEURTEN, WAARVAN";X;"% GOED."
1070 LOCATE 7,21:RE$ = "RESULTAAT: "
1080 IF 0 < X AND X <= 25 THEN PRINT RE$;"SLECHT"
1090 IF 25 < X AND X <= 50 THEN PRINT RE$;"ONVOLDOENDE"
1100 IF 50 < X AND X <= 60 THEN PRINT RE$;"MATTIG"
1110 IF 60 < X AND X <= 75 THEN PRINT RE$;"BEVREDIGEND"
1120 IF 75 < X AND X <= 85 THEN PRINT RE$;"GOED!"
1130 IF 85 < X AND X <= 100 THEN PRINT RE$;"UITSTEKEND!!"
1140 RETURN
1150 :
1160 LOCATE 0,0
1170 PRINT R2$;LEFT$(S1$,34);R8$;LEFT$(S1$,38);R3$
1180 PRINT R6$;" XXX LETTERS XXX ";R6$;"      RESULTAAT ";R6$
1190 PRINT GR$;CHR$(84);LEFT$(S1$,34);GR$;CHR$(85);LEFT$(S1$
,38);GR$;CHR$(86)
1200 FOR I=1 TO 5
1210 PRINT R6$;"          ";R6$;"          "
;R6$
1220 NEXT I

```


Voorbeeld

XXX LETTERS XXX	RESULTAAT
	AANTAL BEURTEN: 2
	AANTAL GOEDEN: 1
	AANTAL LETTERS: 2

De vraag was :V
Het antwoord was :V
JE ANTWOORD WAS GOED!!

WIL JE DOORGAAN JA/NEE :? ■

2.8 Galgje

Tijdens de Romeinse tijd werd in Gallië ook al galgje gespeeld. Men speelde het toen wel iets anders dan tegenwoordig. Uit verschillende geschriften blijkt dat het lot iemand aanwees. De ongelukkige werd voorzien van een strop en aan een staak gehangen. Hij kreeg een zwaard in zijn handen waarmee hij moest proberen om de strop door te snijden. Of dit vaak lukte, verhaalt de overlevering niet.

Dit spel is gelukkig niet zo bloeddorstig...

De computer (de beul) kiest een woord. Men moet proberen om dit woord te raden. Wanneer men van een letter denkt dat deze in het woord voorkomt, geeft men dit op. Is de letter juist dan zal hij op de goede plaats op het scherm worden gezet. Is de letter fout dan begint de beul met zijn werk. Heeft men meer dan tien keer fout geraden dan hangt men.

Suggestie 1: Heeft men een paar keer gespeeld dan kent men alle woorden die in de interne data-file zitten uit het hoofd. Het eerste wat men zou kunnen doen, is de data-file drastisch uitbreiden met andere woorden.

Suggestie 2: U zult hebben gemerkt dat de woorden die u gebruikt niet al te groot zijn. Het is best leuk om moeilijke woorden te nemen. Men moet dan bijvoorbeeld wel de eerste en laatste letter geven. Dit is slechts een kleine aanpassing in de module WOORD OPHALEN.



Programma

```
10 REM *** GALGJE ***
20 CLEAR 2000      :`STRINGRUIMTE
30 DEFSTR Z
40 DIM A$(10), B$(20), D$(20), G$(14)
50 GR$=CHR$(1)
60 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
70 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
80 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
90 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
100 DATA COMPUTER, CLOWN, VUUR, MOLEN, AARDBEI, DEUR
110 DATA BOOMGAARD, SPEELGOED, ZOMER, HERFST
120 DATA POP, AUTO, MOTOR, VLIEGTUIG, BOOT
130 SCREEN 0:WIDTH 40:KEY OFF:COLOR 4,7,7:CLS
140 GOTO 2610
150 :
160 REM *** SPELREGELS ***
170 CLS          :`SCHERM WISSEN
180 LOCATE 0,0
190 PRINT R2$;LEFT$(S1$,36);R8$;LEFT$(S1$,36);R3$
200 PRINT R6$;" *** GALGJE *** ";R6$;" -SPELREGELS-";R6$
210 PRINT R4$;LEFT$(S1$,36);R7$;LEFT$(S1$,36);R5$
220 PRINT
R2$;LEFT$(S1$,24);R3$;SPACE$(11);R2$;LEFT$(S1$,24);R3$
230 PRINT
R6$;R2$;LEFT$(S1$,22);R5$;SPACE$(11);R4$;LEFT$(S1$,22);R3$;R
6$
240 FOR I=1 TO 15
250 PRINT R6$;R6$;SPACE$(35);R6$;R6$
260 NEXT I
270 PRINT R6$;R4$;LEFT$(S1$,70);R5$;R6$
280 PRINT R6$;SPACE$(37);R6$
290 PRINT R4$;LEFT$(S1$,74);R5$
300 Y = 0: LOCATE 2,5:GOTO 340
310 Y = Y + 1
320 LOCATE 2,(5 + Y)
330 RETURN
340 PRINT"De computer trekt een woord. U moet":GOSUB 310
350 PRINT"dat woord zien te raden. Een letter":GOSUB 310
360 PRINT"waarvan U denkt dat hij in het":GOSUB 310
370 PRINT"woord voorkomt, geeft U op. Komt de":GOSUB 310
380 PRINT"letter inderdaad voor, dan ver-":GOSUB 310
390 PRINT"schijs de letter boven in het":GOSUB 310
400 PRINT"beeldscherm op de juiste plaats. Is":GOSUB 310
410 PRINT"hij fout, dan begint de beul met de":GOSUB 310
420 PRINT"terechtstelling. U kunt maximaal 10":GOSUB 310
430 PRINT"fouten maken, anders hangt U! Komt":GOSUB 310
440 PRINT"een letter meerdere malen voor, dan":GOSUB 310
450 PRINT"moet U hem ook meerdere keren vra-":GOSUB 310
460 PRINT"gen. Denkt U het woord te weten,":GOSUB 310
470 PRINT"toets het dan in."
480 LOCATE 2,21
490 PRINT"GELEZEN, DRUK DAN <RETURN-TOETS> IN."
500 K$ = INKEY$
510 IF K$ = CHR$(13) THEN RETURN
```

```

520 GOTO 500
530 :
540 REM *** INITAILIZEREN ***
550 Z1 = " :deze letter komt niet voor!"
560 Z2 = " :deze letter is al gebruikt!"
570 Z3 = " :deze letter is correct!"
580 ZV = "Geef een letter of geef een woord"
590 ZF = "De foute letters zijn:"
600 A$(1) = "HET PLATFORM"
610 A$(2) = "DE STAAK"
620 A$(3) = "HET DWARSHOUT"
630 A$(4) = "DE STROP"
640 A$(5) = "HET HOOFD"
650 A$(6) = "DE ROMP"
660 A$(7) = "DE LINKERARM"
670 A$(8) = "DE RECHTERARM"
680 A$(9) = "HET LINKERBEEN"
690 A$(10) = "HET RECHTERBEEN"
700 G$ = "" : 'GEBRUIKTE KARAKTERS
710 C$ = ""
720 H = 0
730 F = 0 : 'FOUTTELLER
740 VE = 0
750 RETURN
760 :
770 REM *** WOORD OPHALEN UIT INTERNE FILE ***
780 X = INT(RND(-TIME) * 15 + 1) : 'WOORDNUMMER
790 RESTORE 40
800 FOR I = 1 TO X
810 READ AW$
820 NEXT I
830 L = LEN(AW$) : 'LENGTE WOORD
840 FOR I=1 TO L
850 B$(I) = "."
860 Q$ = LEFT$(AW$,I)
870 D$(I) = RIGHT$(Q$,1)
880 NEXT I
890 RETURN
900 :
910 REM *** KOMT LETTER VOOR? ***
920 H = 0
930 E$ = "" : 'STRING WISSEN
940 FOR I=1 TO L
950 IF D$(I) <> AA$ GOTO 1010
960 H = 1
970 IF B$(I) = AA$ GOTO 1010
980 H = 2
990 B$(I) = AA$
1000 I = L
1010 NEXT I
1020 IF H <> 0 GOTO 1090
1030 C$ = AA$ + Z1
1040 G$ = G$ + AA$
1050 AA$=G$
1060 F = F + 1
1070 IF F<11 THEN E$ = A$(F)
1080 GOSUB 2950

```



```

1090 IF H <> 1 GOTO 1110
1100 C$ = AA$ + Z2
1110 IF H <> 2 THEN AA$="":GOTO 1300
1120 C$ = AA$ + Z3
1130 AA$ = ""
1140 RETURN
1150 :
1160 REM *** INVOER ***
1170 X=4:Y=172:C=14
1180 T$ = ZV
1190 GOSUB 2360
1200 X=4:Y=182
1210 DRAW"BM=X; ,=Y;":PRINT#1,CHR$(95):BEEP
1220 A$=INKEY$
1230 IF A$ = CHR$(13) GOTO 1270
1240 IF A$ > CHR$(64) AND A$ <= CHR$(90) THEN AA$=AA$+A$:GOS
UB 2430
1250 IF A$ = CHR$(8) THEN PSET(X,Y):GOSUB 2490
1260 GOTO 1220
1270 X = LEN(AA$) : `AANTAL KARAKTERS
1280 IF X = 1 GOTO 1300
1290 X = 2
1300 LINE(4,182)-(156,190),4,BF
1310 COLOR 15
1320 RETURN
1330 :
1340 REM *** HELE WOORD ***
1350 H = 0
1360 F = F + 1
1370 C$ = "NEE, HOOR HET WOORD IS ANDERS!"
1380 IF Q$ <> AA$ THEN AA$="":GOTO 1460
1390 REM ** WOORD GERADEN **
1400 C$ = "JA, HOOR DAT WAS HET WOORD!"
1410 H = 3
1420 F = F - 1
1430 FOR I=1 TO L
1440 B$(I) = D$(I)
1450 NEXT I
1460 RETURN
1470 :
1480 REM *** UITVOER ***
1490 SCREEN 2 : `HIGH RESOLUTION
1500 OPEN "GRP:" AS#1 : `TOGANG TOT SCHERM 2
1510 COLOR 15,5,1:CLS:VE=1
1520 LINE(0,0)-(255,24),4,BF
1530 LINE(113,25)-(255,192),15,BF
1540 LINE(0,170)-(255,192),4,BF
1550 X = 9:Y = 9:C = 14
: `COORDINATEN + KLEUR
1560 T$ = "... GALGJE ... BEURT:" : `LENGTE VAN
STRING T$
1570 GOSUB 2360
1580 DRAW"BM202,9":COLOR 4:PRINT#1,STRING$(4,219)
1590 DRAW"BM202,9":COLOR 15:PRINT#1,USING"###";B
1600 LINE(0,158)-(255,170),14,BF
1610 LINE(26,26)-(92,34),5,BF
1620 X = 34:Y = 26:C = 1 : `COORDINATEN + KLEUR

```



```

1630 DRAW"BM30,26C1D7"
1640 T$ = ""
1650 FOR I = 1 TO L
1660 T$ = T$ + B$(I)
1670 NEXT I
1680 GOSUB 2360
1690 X = X + 2:DRAW"BM=X; ,=Y;D7"
1700 T$ = "AANTAL LETTERS:"
1710 C = 1:X = 120:Y = 26
1720 GOSUB 2360
1730 T$ = STR$(L):GOSUB 2360
1740 T$ = "De foute letters zijn:"
1750 C = 4:X = 120:Y = 38
1760 GOSUB 2360
1770 IF F < 10 THEN GOSUB 1790
1780 RETURN
1790 GOSUB 1860
1800 X = 8:Y = 160
1810 GOSUB 2360
1820 RETURN
1830 LINE(132,148)-(254,157),15,BF
1840 DRAW"BM132,148":COLOR C:PRINT#1,A$(F)
1850 RETURN
1860 X = 120:Y = 54:C = 1
1870 T$ = G$
1880 GOSUB 2360:T$ = C$
1890 RETURN
1900 REM *** HET PLATFORM ***
1910 LINE(4,148)-(90,156),1,BF
1920 GOSUB 1830
1930 RETURN
1940 REM *** DE STAAK ***
1950 LINE(7,148)-(15,43),1,B
1960 GOSUB 1830
1970 RETURN
1980 REM *** HET DWARSHOUT ***
1990 LINE(4,50)-(80,56),1,BF
2000 DRAW"BM15,66E15"
2010 DRAW"BM16,74E20"
2020 GOSUB 1830
2030 RETURN
2040 REM *** DE STROP ***
2050 LINE(72,56)-(73,64),1,B
2060 GOSUB 1830
2070 RETURN
2080 REM *** HET HOOFD ***
2090 DRAW"BM69,65":PRINT#1,CHR$(1);CHR$(65)
2100 GOSUB 1830
2110 RETURN
2120 REM *** DE ROMP ***
2130 LINE(72,72)-(73,74),1,B
2140 DRAW"BM72,74G5":DRAW"BM73,74F5"
2150 LINE(67,79)-(78,95),1,B
2160 GOSUB 1830
2170 RETURN
2180 REM *** LINKER ARM ***
2190 DRAW"BM67,79G7":DRAW"BM67,82G7"

```

```

2200 DRAW"BM60,86D2"
2210 GOSUB 1830
2220 RETURN
2230 REM *** RECHTER ARM ***
2240 DRAW"BM78,79F7":DRAW"BM78,82F7"
2250 DRAW"BM85,86D2"
2260 GOSUB 1830
2270 RETURN
2280 REM *** LINKER BEEN ***
2290 LINE(67,95)-(71,112),1,B
2300 GOSUB 1830
2310 RETURN
2320 REM *** RECHTER BEEN ***
2330 LINE(74,95)-(78,112),1,B
2340 GOSUB 1830
2350 RETURN
2360 REM *** INPUT & PRINT ***
2370 FOR I = 1 TO LEN(T$)
2380 COLOR C:DRAW"BM=X; ,=Y;":X = X + 6
2390 PRINT#1,MID$(T$,I,1)
2400 NEXT I
2410 RETURN
2420 REM ** INVOER CURSOR **
2430 FOR T=1 TO LEN(A$)
2440 DRAW"BM=X; ,=Y;":COLOR 4:PRINT#1,CHR$(200)
2450 DRAW"BM=X; ,=Y;":COLOR 15:PRINT#1,MID$(A$,T,1):X=X+6
2460 NEXT T
2470 DRAW"BM=X; ,=Y;":PRINT#1,CHR$(95)
2480 RETURN
2490 Z$ = CHR$(200)
2500 K = LEN(AA$)
2510 COLOR 4:PRINT#1,Z$
2520 :
2530 X=X-6:DRAW"BM=X; ,=Y;":PRINT#1,Z$
2540 IF X < 0 THEN X = 4
2550 COLOR 15:DRAW"BM=X; ,=Y;":PRINT#1,CHR$(95)
2560 IF K = 0 THEN K = 1
2570 AA$=LEFT$(AA$,K-1)
2580 RETURN
2590 REM **** STUURMODULE ****
2600 REM *** SPELREGELS ***
2610 GOSUB 150
2620 REM *** INITIALIZEREN ***
2630 GOSUB 550
2640 REM *** WOORD OPHALEN ***
2650 GOSUB 780
2660 REM *** UITVOER ***
2670 IF VE=0 THEN GOSUB 1490 ELSE GOSUB 1580
2680 IF F = 0 GOTO 2710
2690 ON F GOSUB
1900 1950,1980,2040,2090,2120,2180,2240,2290,2330
2700 REM *** INVOER ***
2710 GOSUB 1170
2720 B = B + 1
2730 REM *** BEWREKING ***
2740 ON X GOSUB 920,1350
2750 IF (H <> 3) AND (F <= 10) GOTO 2670


```

```

2760 REM *** UITVOER ***
2770 GOSUB 1580
2780 C$ = "GOED HOOR, HET WOORD WAS INDERDAAD:"
2790 IF H = 3 GOTO 2820
2800 C$ = "JE HANGT, HET WOORD WAS:"
2810 REM *** EIND VAN HET PROGRAMMA ***
2820 LINE(0,170)-(255,192),4,BF
2830 X = 4:Y = 172:C = 15
2840 T$ = C$
2850 GOSUB 2370
2860 DRAW"BM100,182":PRINT#1,AW$
2870 COLOR 7
2880 DRAW"BM101,182":PRINT#1,AW$
2890 FOR I = 1 TO 15
2900 LINE(113,25)-(255,157),I,BF
2910 FOR T = 1 TO 100:NEXT T
2920 NEXT I
2930 GOTO 2930
2940 REM *** MUZIEK ***
2950 PLAY"V15T90","V10T90","V10T90"
2960 PLAY"O5E.D+16E16F16R16E16R16C16R16C16O4B16A.R8","O4C8E8
C8E8C8E8C8E8C.R8","O3A8R8A8R8A8R8A8R8A.R8"
2970 IF PLAY(0) <> 0 GOTO 2970
2980 RETURN

```


Voorbeeld



2.9 Mastermind

Dit spel, dat overigens is afgeleid van het in Engeland bekende spel 'cows and bulls' is vrij algemeen bekend. Vandaar dat hier slechts een vrij summiere uitleg wordt gegeven. Het programma ontwikkelt een geheime kleurcode van vier verschillende kleuren. Het is de bedoeling dat de speler deze code breekt. Om dit te bereiken, geeft de computer u aanwijzingen: is een kleur juist maar de plaats foutief dan verschijnt op het scherm een asterisk (*). Is zowel de kleur als de plaats juist dan wordt dit aangegeven met een hekje ('#'). Om het wat gemakkelijker te maken, worden de laatste vijf pogingen met de resultaten op het scherm gezet.

Suggestie: Men speelt dit spel tegen de computer. Het is natuurlijk interessant wanneer de computer ook tegen de speler speelt. Op het eerste gezicht lijkt dit een vrij moeilijke programma-ontwikkeling. Een aanzet kan zijn om bij uzelf eens na te gaan hoe u bepaalt uit welke kleuren de code bestaat. Aan de hand hiervan moet het mogelijk zijn om een dergelijke routine te ontwerpen. Het is wel aan te bevelen om uit te gaan van een code van vier verschillende kleuren.



Programma

```
10 REM *** MASTERMIND ***
20 VL = 0: CLEAR 1000
30 GR$=CHR$(1)
40 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
50 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
60 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
70 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
80 CU$=CHR$(200)
90 SCREEN 3:OPEN"GRP:" AS#1:COLOR 9,6,6:CLS
100 PRESET(8,8):PRINT#1,"MASTER-"
110 PRESET(72,72):PRINT#1,"MIND"
120 GOSUB 2260:CLOSE
130 SCREEN 0:WIDTH 40:KEY OFF:COLOR 15,5,5:CLS
140 DIM A(5,8), B$(2), C(4), G(4)
150 GOTO 2010
160 :
170 REM *** SPELREGELS ***
180 LOCATE 0,0
190 PRINT R2$;LEFT$(S1$,40);R8$;LEFT$(S1$,32);R3$
200 PRINT R6$;" *** MASTERMIND *** ";R6$;" - SPELREGELS - ";
R6$
210 PRINT R4$;LEFT$(S1$,40);R7$;LEFT$(S1$,32);R5$
220 PRINT
230 PRINT" De computer maakt een geheime code van"
240 PRINT" vier verschillende kleuren."
250 PRINT" U moet proberen om die code te raden."
260 PRINT" De kleuren liggen tussen 0 en 9."
270 PRINT" Heeft men een kleur en de plaats van"
280 PRINT" de kleur goed dan geeft de computer dit"
290 PRINT" aan met #. Heeft men wel een kleur"
300 PRINT" goed, maar niet zijn juiste plaats dan"
310 PRINT" wordt dit aangegeven met *."
320 PRINT" Voor het gemak van de speler worden de"
330 PRINT" vijf kleuren die het laatst zijn ge-
340 PRINT" probeerd, getoond."
350 PRINT
360 PRINT" Na een kleurnr. dient geen komma te"
370 PRINT" staan; dus bijv: 1234 of 3784!!"
380 LOCATE 0,20
390 PRINT R2$;LEFT$(S1$,74);R3$
400 PRINT R6$;" GELEZEN, DRUK <RETURNTOETS> IN. ";R6$
410 PRINT R4$;LEFT$(S1$,74);R5$
420 A$ = INKEY$
430 IF A$ = CHR$(13) THEN RETURN
440 GOTO 420
450 :
460 REM *** INITIALIZEREN ***
470 Y1 = 62
480 REM ** SCHOONVEGEN MAT A **
490 FOR I = 1 TO 5
500 FOR J = 1 TO 8
510 A(I,J) = 0
520 NEXT J
530 NEXT I
540 REM *** GEHEIME CODE AANMAKEN ***
```

```

550 FOR I = 1 TO 4
560 X = INT(RND(-TIME) * 10)
570 IF X = 0 GOTO 560
580 IF I = 1 GOTO 680
590 REM ** CIJFER AL GEBRUIKT **
600 H = 0
610 FOR J = 1 TO I-1
620 IF X <> C(J) GOTO 660
630 REM ** CIJFER KOMT VOOR **
640 H = 1
650 J = I
660 NEXT J
670 IF H <> 0 GOTO 560
680 C(I) = X
690 NEXT I
700 REM ** STRINGMATRIX VULLEN **
710 B$(1) = " * "
720 B$(2) = " # "
730 REM ** STARTVARIABLEN **
740 B = 0
750 T = 0
760 A$ = ""
770 RETURN
780 :
790 H = 0
800 REM *** UITVOER ***
810 OPEN "GRP:" AS#1
820 COLOR 1,15,1:SCREEN 2
830 GOSUB 1770
840 DRAW"BM176,8":COLOR 10:PRINT#1,CU$;CU$
850 DRAW"BM176,8":COLOR 1:PRINT#1,B
860 DRAW"BM24,32"
870 PRINT#1,"Plaats & kleur OK : - # -"
880 DRAW"BM24,40"
890 PRINT#1,"Alleen kleur is OK: - * -"
900 IF T = 0 GOTO 1030
910 FOR I = 1 TO T
920 DRAW"BM24,=Y1;":COLOR A(I,1):PRINT#1,CU$;
930 COLOR A(I,2):PRINT#1,CU$;
940 COLOR A(I,3):PRINT#1,CU$;
950 COLOR A(I,4):PRINT#1,CU$;
960 COLOR 1:PRINT#1,"-- ";
970 FOR J = 5 TO 8
980 IF A(I,J) = 0 GOTO 1010
990 X = A(I,J)
1000 PRINT#1,B$(X);
1010 NEXT J
1020 NEXT I
1030 DRAW"BM20,136":COLOR 1:PRINT#1,A$
1040 LINE(0,162)-(255,192),5,BF
1050 TE = 0
1060 IF Y1=112 THEN LINE(0,70)-(255,120),15,BF
1070 RETURN
1080 :
1090 REM *** INVOER ***
1100 FOR I=8 TO 9
1110 DRAW"BM=I;,168"

```



```

1120 PRINT#1,"WAT IS DE CODE (GEEN KOMMA'S):"
1130 NEXT I
1140 X = 8:Y = 182
1150 DRAW"BM=X; ,=Y;":PRINT#1,CHR$(192):BEEP
1160 A$ = INKEY$
1170 A = VAL(A$)
1180 IF A$ = CHR$(13) GOTO 1240
1190 IF A >= 1 AND A <= 14 THEN TE = TE + 1:GOSUB 1840
1200 IF A$ = CHR$(8) THEN PSET(X,Y):GOSUB 1920
1210 GOTO 1160
1220 RETURN
1230 REM ** CONTROLE OP CIJFERS **
1240 LINE(8,182)-(256,192),5,BF
1250 H = 0
1260 FOR I = 1 TO 4
1270 IF (G(I) > 0) AND (G(I) <= 9) GOTO 1300
1280 A$ = "CIJFERS INVOEREN! GEEN GETALLEN"
1290 H = 1
1300 NEXT I
1310 IF H = 1 GOTO 1410
1320 FOR I = 1 TO 4
1330 FOR J = 1 TO 4
1340 IF (G(I) <> G(J)) OR (I=J) GOTO 1390
1350 H = 1
1360 I = 4
1370 J = 4
1380 A$ = "DE CIJFERS MOETEN VERSCHILLEN!"
1390 NEXT J
1400 NEXT I
1410 RETURN
1420 :
1430 REM *** BEREKENING ***
1440 T = T + 1
1450 IF T < 6 GOTO 1540
1460 REM ** GETALLEN OPSCHUIVEN **
1470 FOR I = 1 TO 4
1480 FOR J = 1 TO 8
1490 A(I,J) = A(I+1,J)
1500 NEXT J
1510 NEXT I
1520 T = T - 1
1530 REM ** GOK OPNEMEN IN RIJ **
1540 FOR J = 1 TO 8
1550 A(T,J) = 0
1560 IF J > 4 GOTO 1580
1570 A(T,J) = G(J) :`OVERNEMEN GOK
1580 NEXT J
1590 REM ** CONTROLE OP GETALLEN **
1600 L = 4 :`HULPVARIABELE
1610 S = 0
1620 FOR I = 1 TO 4
1630 FOR J = 1 TO 4
1640 IF G(I) <> C(J) GOTO 1710
1650 P = 1
1660 IF I <> J GOTO 1690
1670 P = 2
1680 S = S + 2

```

```

1690 L = L + 1
1700 A(T,L) = P : `PLAATS/KLEUR
1710 NEXT J
1720 NEXT I
1730 IF S <> 8 GOTO 1760
1740 A$ = "U heeft de code geraden."
1750 H = 3
1760 RETURN
1770 LINE(0,0)-(255,24),10,BF
1780 DRAW"BM0,0"
1790 PRINT#1,R2$;LEFT$(S1$,24);R8$;LEFT$(S1$,34);R3$
1800 DRAW"BM0,8":PRINT#1,R6$;" MISTERMIND ";R6$;" BEURT: "
;R6$
1810 DRAW"BM1,8":PRINT#1," MISTERMIND BEURT:"
1820 DRAW"BM0,16":PRINT#1,R4$;LEFT$(S1$,24);R7$;LEFT$(S1$,34
);R5$
1830 RETURN
1840 FOR T=1 TO LEN(A$)
1850 DRAW"BM=X; ,=Y; ":COLOR 5:PRINT#1,CU$
1860 DRAW"BM=X; ,=Y; ":COLOR 1:PRINT#1,MID$(A$,T,1):X = X + 6
1870 NEXT T
1880 DRAW"BM=X; ,=Y; ":PRINT#1,CHR$(192)
1890 G(TE)=A
1900 LINE(20,136)-(255,146),15,BF
1910 RETURN
1920 Z$=CU$
1930 L=LEN(AA$)
1940 COLOR 5:PRINT#1,Z$
1950 X=X-6:DRAW"BM=X; ,=Y; ":PRINT#1,Z$
1960 COLOR 1:DRAW"BM=X; ,=Y; ":PRINT#1,CHR$(192)
1970 TE = TE - 1
1980 RETURN
1990 REM **** STUURMODULE ****
2000 REM *** SPELREGELS ***
2010 GOSUB 180
2020 REM *** INITIALIZEREN ***
2030 GOSUB 470
2040 B = B + 1 : `BEURTELLER + 1
2050 REM *** UITVOER ***
2060 IF VL=0 THEN GOSUB 810:GOTO 2080
2070 IF VL = 1 AND Y1 <= 102 THEN Y1 = Y1 + 10:GOSUB 840 ELS
E Y1 = 72:GOSUB 840
2080 A$ = ""
2090 VL = 1
2100 REM *** INVOER ***
2110 GOSUB 1100
2120 IF H = 1 GOTO 2040
2130 REM *** BEREKENING ***
2140 GOSUB 1440
2150 IF H < 3 GOTO 2040
2160 REM *** UITVOER ***
2170 Y1 = Y1 + 10:GOSUB 840
2180 REM *** RESULTAAT ***
2190 FOR X=32 TO 33
2200 DRAW"bm=x; ,175"
2210 PRINT#1,"U HAD";B;"BEURTEN NODIG."
2220 NEXT X


```

```

2230 GOSUB 2260
2240 GOTO 2240
2250 END
2260 PLAY"V13T120","V10T120"
2270
T1$="O3A1604E16C16E1603A1604E16C16E16":T2$="O3G+1604E1603B16
O4E1603G+1604E1603B1604E16":T3$="O4C16G16E16G16C16G16E16G16"
2280 PLAY"R2R2O4A16R16A8O5C8D8EO4A16R16A8O5FE8D8ER8E8",T1$+T
1$+T1$+T1$+"O3A1604F16D16F1603A1604F16D16E16"+T1$
2290 PLAY"O5E8.O4B1605E8O4B8O5CO4A8O5C8EO4B16R16B8O5C.E8G8.F
16E8.D16CR8C8",T2$+T1$+T2$+T1$+T3$+T3$
2300 PLAY"O5D8R16D16O4B8G8O5E.R8E8.F16E8.D+16EA8R8F8.D16O4B8
O5E8O4AO5A8R8","O3B1604G16D16G16O3B16G16D16G16"+T3$+T2$+T1$+
"O3D16F16A16O4D16O3E16G+16B16O4E16O3A16O4C16E16C16O3A8R8"
2310 RETURN

```

Voorbeeld



2.10 Vier op een rij

De spelregels van 'vier op een rij' zijn vrij eenvoudig. Het is de bedoeling om vier dezelfde schijven op één rij te krijgen. Dit geldt zowel horizontaal, verticaal als diagonaal.

Het spel wordt tegen de computer gespeeld. De computer heeft groene schijven, de speler rode. Aan de randkleur is steeds te zien wie er aan de beurt is.

Het programma neemt alleen beslissingen op basis van bepaalde spelsituaties. Zo wordt de positie van drie eigen symbolen op een rij anders gewaardeerd dan drie symbolen van de tegenstander. De werking van het programma kan men het beste volgen aan de hand van de listing. Alleen de waardering verdient een nadere uitleg.

Het programma bekijkt steeds de situatie van vier opeenvolgende vakken. Dit gebeurt horizontaal, verticaal, diagonaal van rechts naar links en diagonaal van links naar rechts. Dit is overigens meteen de verklaring waarom het zo lang duurt voordat het programma met een zet komt. Het hele speelbord moet per zet worden afgewerkt.

Kan het programma een zet doen waarbij hij vier eigen symbolen op een rij kan krijgen dan wordt dit met de hoogste waardering gehonoreerd. Kan door een zet van het programma worden voorkomen dat de tegenstander vier op een rij maakt dan krijgt die zet de op één na hoogste waardering. Dit gaat zo door voor drie en twee op een rij. Begint de computer dan wordt met een willekeurige zet begonnen.

Veel strategische bordspellen die als programma verkrijgbaar zijn, werken volgens dit principe. Een aardig spel dat in het verlengde ligt van 'vier op een rij' is Gomoko. Gomoko wordt gespeeld op een veld van 19 bij 19. Men moet proberen vijf op een rij te krijgen. In bijna elk boek over bordspellen staat een beschrijving van Gomoko.

Suggestie 1: Wanneer men dit spel speelt, zal het opvallen dat het programma vrij lang moet nadenken. Dit is te optimaliseren door de scan-module te verbeteren.

Suggestie 2: Men kan experimenteren met de waardering van de zetten. Op deze manier krijgt men een aardig inzicht in wat er gebeurt. Het programma is in het begin van het spel niet zo sterk; dit komt omdat alleen de volgende zet wordt bepaald. Dit laatste is te verhelpen door niet één zet diep door te rekenen, maar bijvoorbeeld twee zetten. Er zal dan wel wat moeten worden gedaan aan de lange rekentijd.

Programma

```
10 REM *** VIER OP EEN RIJ ***
20 CLEAR 1000
30 GR$=CHR$(1)
40 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
50 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
60 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
70 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
```

```

80 BO$=CHR$(188):BD$=CHR$(249)
90 SCREEN 0:WIDTH 40:KEY OFF:COLOR 1,12,12
100 DIM A(6,7), B(7), C(7), D(2), W(6,2)
110 DATA 1,0,1,7,1,3           :`SCAN RECHTOMHOOG
120 DATA 0,1,1,4,1,6           :`SCAN HORIZONTAAL
130 DATA 1,1,1,4,1,3           :`SCAN DIAGONAAL L-R
140 DATA 1,-1,4,7,1,3          :`SCAN DIAGONAAL R-L
150 DATA 9999,3,-6666,3,100,2
160 DATA -50,2,3,1,-1,1
170 GOTO 2380
180 :
190 REM *** SPELREGELS ***
200 CLS                           :`SCHERM WISSEN
210 GOSUB 2340
220 LOCATE 0,0
230 PRINT L$
240 PRINT " VIER OP EEN RIJ      -SPELREGELS-  "
250 PRINT L$
260 LOCATE,5
270 PRINT " Het speelbord bestaat uit zeven kolom-"
280 PRINT " men. De speler en de computer gooien"
290 PRINT " om beurten hun symbool in een van de"
300 PRINT " kolommen."
310 PRINT " Het is de bedoeling vier eigen symbo-"
320 PRINT " len op een rij te krijgen. Dit geldt"
330 PRINT " zowel horizontaal, verticaal als dia-"
340 PRINT " gonaal. Degene, die dit het eerst lukt"
350 PRINT " heeft gewonnen."
360 LOCATE,16
370 PRINT " Het symbool van de speler is : (";BO$;")."
380 PRINT " Het symbool van de computer is: (";BD$;")."
390 LOCATE,20
400 PRINT L$
410 PRINT " WIL JIJ BEGINNEN      JA/NEE : "
420 PRINT L$
430 GOSUB 2630
440 RETURN
450 REM *** WIE BEGINT ? ***
460 LOCATE32,21:LINE INPUT Q$
470 Q$ = LEFT$(Q$,1)
480 M = 2                           :`SPELER BEGINT
490 IF Q$ <> "J" AND Q$ <> "j" GOTO 510
500 M = 1                             :`COMPUTER
BEGINT
510 RETURN
520 :
530 REM *** INITIALIZEREN ***
540 REM ** DUMMY LEZEN **
550 RESTORE                           :`DATAPOINTER V
OORAAN
560 FOR I=1 TO 24
570 READ X
580 NEXT I
590 REM ** VARIABELEN STARTWAARDE GEVEN **
600 FOR I=1 TO 6
610 B(I) = 0
620 C(I) = 0                           :`MATRIX MET KO

```

```

LOMVULLING
630 REM ** WAARDERINGSMATRIX W VULLEN **
640 READ W(I,1), W(I,2) : `WAARDERING
INLEZEN
650 REM ** REGEL SPEELVELD WISSEN **
660 FOR J=1 TO 7
670 A(I,J) = 0
680 NEXT J
690 NEXT I
700 VL = 0
710 Z = 1 : `ZETTENTELLER
720 A$=R2$:FOR RS=1 TO 6:A$=A$+R1$+R1$+R1$+R8$:NEXT RS:A$=A$
+R1$+R1$+R1$+R3$
730 A1$=R4$:FOR RS=1 TO 6:A1$=A1$+R1$+R1$+R1$+R7$:NEXT RS:A1
$=A1$+R1$+R1$+R1$+R5$
740 RETURN
750 :
760 REM *** ZET SPELER ***
770 SS$ = LEFT$(S1$,38)
780 LOCATE ,20:PRINT SS$
790 PRINT" WELKE KOLOM : "
800 PRINT SS$
810 LOCATE 15,21:PLAY"O4A32":LINE INPUT K$
820 LOCATE 14,21:PRINT" "
830 K = VAL(K$)
840 IF (K <= 0) OR (K > 7) THEN PLAY"O2G.":GOTO 770
850 IF VL = 1 THEN LOCATE 26,5,0:PRINT"? "
860 PLAY"V15S13T255O6L63M1D"
870 PLAY"V14S13T255O5L63M1D"
880 VL = 1
890 K = INT(K) : `ZEKER EEN
GEHEEL GETAL
900 IF C(K) > 6 GOTO 770 : `KAN ER NOG
SYMBOOL BIJ?
910 C(K) = C(K) + 1 : `HOOGTE
BIJWERKEN
920 P = C(K)
930 A(P,K) = 9 : `ZET PLAATSEN
SPELER
940 Z = Z + 1 : `ZETTENTELLER
VERHOGEN
950 RETURN
960 :
970 REM *** UITVOER SPEELVELD ***
980 CLS: : `SCHERM
WISSEN
990 LOCATE 0,0
1000 PRINT R2$;LEFT$(S1$,34);R8$;LEFT$(S1$,38);R3$
1010 PRINT R6$;" VIER OP EEN RIJ ";R6$;" ZETNR.:
";R6$
1020 PRINT R4$;LEFT$(S1$,34);R7$;LEFT$(S1$,38);R5$
1030 LOCATE 30,1:PRINT Z
1040 LOCATE 0,5
1050 IF M < 2 GOTO 1070
1060 PRINT"De computer zet in kolom:";Y
1070 LOCATE 5,8:PRINT A$ : `SCHEIDINGSREGEL
1080 FOR I = 6 TO 1 STEP -1

```



```

1090 FOR J = 1 TO 7
1100 K = (J - 1) * 4 + 5
1110 LOCATE K:PRINT"V";
1120 B$ = " "
1130 IF A(I,J) = 0 GOTO 1170
1140 B$ = " "+BD$+" "
1150 IF A(I,J) <> 9 GOTO 1170
1160 B$ = " "+BO$+" "
1170 PRINT B$;
1180 NEXT J
1190 LOCATE 33:PRINT R6$
1200 NEXT I
1210 LOCATE 5:PRINT A1$
1220 PRINT"KOLOM 1 2 3 4 5 6 7"
1230 IF C$ <> "" THEN GOSUB 2300
1240 LOCATE ,20,0:PRINT C$
1250 IF C$ = "JE HEBT GEWONNEN. GEFELICITEERD!" THEN GOSUB 2
630
1260 RETURN
1270 :
1280 REM *** ZETBESLISSING ***
1290 IF H > 2 GOTO 1400 :`SPELER GEWONNEN
?
1300 Y = 1
1310 FOR J = 2 TO 7
1320 IF B(Y) => B(J) GOTO 1340
1330 Y = J
1340 NEXT J
1350 REM ** ZET **
1360 C(Y) = C(Y) + 1
1370 P = C(Y)
1380 A(P,Y) = 1 :`COMPUTER ZET
1390 Z = Z + 1 :`ZETTENTELLER
VERHOGEN
1400 RETURN
1410 :
1420 REM *** AKTIE ***
1430 REM ** AANWIJZING = C$ **
1440 I = R :`RELATIEVE
START-REGEL
1450 J = K :`RELATIEVE
START-KOLOM
1460 FOR N = 1 TO 4
1470 IF A(I,J) <> 0 GOTO 1580
1480 REM ** KAN PLAATS GEBRUIKT WORDEN **
1490 IF C(J) + 1 <> I GOTO 1580
1500 REM ** PLAATS KAN GEBRUIKT WORDEN **
1510 B(J) = B(J) + ABS(W(G,1)) :`WAARDERING
1520 T = 1
1530 H = 1
1540 IF W(G,1) <> 9999 GOTO 1580
1550 C$ = "IK BEN JE TE SLIM AF GEWEEST, GOED HE!"
1560 N = 4
1570 H = 2
1580 I = I + V1
1590 J = J + V2
1600 NEXT N

```

```

1610 RETURN
1620 :
1630 REM *** SCAN ***
1640 REM ** AANWIJZING = C$ **
1650 FOR X = 1 TO 4
1660 IF A(I,J) = 0 GOTO 1720
1670 Q = 1 : `COMPUTER -
AANDUIDING
1680 IF A(I,J) <> 9 GOTO 1700
1690 Q = 2 : `SPELER -
AANDUIDING
1700 S = S + 1 : `AANTAL
BEZETTE PLAATSEN
1710 D(Q) = D(Q) + 1 : `AANTAL
SPELER OF COMP.
1720 I = I + V1
: `REGELAANPASSING
1730 J = J + V2
: `KOLOMAANPASSING
1740 NEXT X
1750 IF S = 0 GOTO 1910
1760 REM ** VOORBEWERKING **
1770 H = 3
1780 C$ = "JE HEBT GEWONNEN. GEFELICITEERD!"
1790 IF D(2) = 4 GOTO 1910
1800 C$ = ""
1810 H = 0
1820 FOR G = 1 TO 6
1830 P = 1 : `AANDUIDING
SPELER
1840 IF W(G,1) > 0 GOTO 1860
1850 P = 2 : `AANDUIDING
COMPUTER
1860 IF D(P) <> W(G,2) GOTO 1900
1870 REM ** AKTIE **
1880 GOSUB 1440
1890 G = 6
1900 NEXT G
1910 RETURN
1920 :
1930 REM *** SCAN-STURING ***
1940 T = 0
1950 RESTORE
1960 H = 0
1970 FOR I = 1 TO 7
1980 B(I) = 0
1990 NEXT I
2000 FOR E = 1 TO 4
2010 READ V1,V2,Y1,Y2,L1,L2 : `STUURWAARDEN
2020 FOR K = Y1 TO Y2
2030 FOR R = L1 TO L2
2040 I = R : `REGEL
2050 J = K : `KOLOM
2060 D(1) = 0 : `MAT D
SCHOONVEGEN
2070 D(2) = 0
2080 S = 0

```

```

: `PLAATSENTELLER = 0
2090 REM ** SCAN **
2100 GOSUB 1650
2110 IF H < 2 GOTO 2160
2120 REM ** LUSSEN BEEINDIGEN **
2130 R = L2
2140 K = Y2
2150 E = 4
2160 NEXT R
2170 NEXT K
2180 NEXT E
2190 IF T <> 0 GOTO 2220
2200 C$ = "GELIJK SPEL!"
2210 H = 3
2220 RETURN
2230 :
2240 REM *** RANDOM-ZET PROGRAMMA ***
2250 Y = INT(RND(-TIME) * 3 + 3) : `RANDOM
KOLOM
2260 C(Y) = C(Y) + 1 : `VERWERKEN IN
VULLING
2270 P = C(Y)
2280 A(P,Y) = 1 : `ZET
PLAATSEN
2290 RETURN
2300 FOR I = 18 TO 22
2310 LOCATE 0,I,0:PRINT SPACE$(39)
2320 NEXT I
2330 :
2340 L$ = LEFT$(S1$,78)
2350 RETURN
2360 REM **** STUURMODULE ****
2370 REM *** SPELREGELS ***
2380 GOSUB 200
2390 REM *** INITIALISEREN ***
2400 GOSUB 550
2410 REM *** WIE BEGINT ***
2420 GOSUB 460
2430 IF M <> 2 GOTO 2470
2440 REM *** RANDOM ZET ***
2450 GOSUB 2250
2460 REM *** UITVOER SPEELVELD ***
2470 IF VL = 0 THEN GOSUB 980 ELSE GOSUB 1030
2480 REM *** ZET SPELER ***
2490 GOSUB 770
2500 M = 1 : `SPELER
UITVOER
2510 REM *** UITVOER SPEELVELD ***
2520 IF VL = 0 THEN GOSUB 980 ELSE GOSUB 1030
2530 M = 2 : `PROGRAMMA
UITVOER
2540 REM *** SCAN - STURING ***
2550 GOSUB 1940
2560 REM *** ZET-BESLISSING ***
2570 GOSUB 1290
2580 IF H < 2 GOTO 2460
2590 REM *** UITVOER SPEELVELD ***

```




```

2600 IF VL = 0 THEN GOSUB 980 ELSE GOSUB 1030
2610 END
2620 REM *** MUZIEK ***
2630 PLAY"V14T120","V14T120","V10T120"
2640 T1$="O3F804C803A804C8":T2$=" O3C804C803C804C8"
2650
PLAY"O4F8G8A16R16A16R16A16R16A16R16A16R16A16R16A16R16A16R16A16R16A16R16
AO5C2O4B-8A8GAB-O5CO4A2R","O4RF16R16F16R16F16R16F16R16F16R16F16R1
6F16R16F16R16F16R16FA2G8F8EFGEF2R","R"+T1$+T1$+T1$+T1$+T2$+
T2$+"O3F804C803A804C803FR"
2660 RETURN

```

Voorbeeld



3 Science fiction-spellen

3.1 Maanlander

Dit programma is de voorloper van het programma 'Ruimtevaartuig'. De meeste suggesties die bij deze maanlander zouden kunnen worden gegeven, zijn in 'Ruimtevaartuig' verwezenlijkt. Als u beide programma's naast elkaar legt, zal de overeenkomst blijken. Overigens kan men wel goed zien hoe men van het kleine naar het grote programma overgaat, zonder de structuur van het programma aan te tasten.

Het programma 'Maanlander' stelt de gebruiker in staat een Lunar Module op de maan te zetten. Op dit moment is dat niet zo iets nieuws meer, maar een tiental jaren geleden was het bijzonder spectaculair.

Wanneer men zelf een dergelijk programma zou willen maken, kan men uit dit programma alle formules halen. Er wordt overigens bij deze formules van uitgegaan dat er geen luchtweerstand is.

Men moet in dit programma een bepaald tijdsinterval kiezen, alsmede de stuwdruk en de richting daarvan. Deze gegevens moeten in één keer worden ingevoerd, waarbij de komma's tussen de gegevens dan ook niet mogen worden vergeten.

Om het geheel nog aantrekkelijker te maken, wandelt er zo nu en dan toch een ruimtevaartuigje door het beeld.

Programma

```
10 REM *** MAANLANDER ***
20 SCREEN 0:WIDTH 40:KEY OFF:COLOR 14,1,1
30 GOTO 2150
40 :
50 REM *** INSTRUCTIE ***
60 CLS                                : 'SCHERM WISSEN
70 LOCATE 0,4
80 R1$=CHR$(1)+CHR$(87)
90 L$ = "":FOR RS=1 TO 39:L$=L$+R1$:NEXT RS
100 PRINT L$
110 PRINT"*****      MAANLANDER      *****"
120 PRINT L$
130 LOCATE ,9
140 PRINT" De bedoeling is dit ruimtevaartuig be-"
150 PRINT" houden op een vastgestelde plaats te"
160 PRINT" laten landen."
170 PRINT" Het ruimtevaartuig bevindt zich al"
180 PRINT" volledig binnen de aantrekkingskracht"
190 PRINT" van de maan."
200 LOCATE ,17:PRINT L$
210 PRINT" GELEZEN, DRUK <RETURN-TOETS> IN."
220 PRINT L$
230 GOSUB 2330
```

```

240 A$ = INKEY$
250 IF A$ = CHR$(13) GOTO 270
260 GOTO 240
270 CLS
280 DEFUSR = &H41:DEFUSR1 = &H44:X = USR(0)
290 LOCATE 0,0
300 PRINT L$
310 PRINT" * RUIMTEVAARTUIG *           -BESTURING- "
320 PRINT L$
330 LOCATE 0,4
340 PRINT" 1. De stuwdruk is te regelen van 1 tot"
350 PRINT"     en met 100% van het motorvermogen."
360 PRINT
370 PRINT" 2. De richting waarin de stuwdruk ge-"
380 PRINT"     geven kan worden, is te geven van"
390 PRINT"     -180 t/m 180 graden. Recht omhoog"
400 PRINT"     is 0 graden, 90 graden is naar"
410 PRINT"     rechts en -90 graden is naar links."
420 PRINT"     180 en -180 graden is recht omlaag."
430 PRINT"     Alle tussenliggende waarden zijn"
440 PRINT"     ook mogelijk."
450 PRINT
460 PRINT" 3. Het tijdsinterval moet bij iedere"
470 PRINT"     besturingsopdracht gegeven worden."
480 PRINT"     De tijdseenheid is de seconde."
490 LOCATE ,20
500 PRINT L$
510 PRINT" NU GAAN WE LANDEN, DRUK <RETURN> IN."
520 PRINT L$
530 A = USR1(0)
540 A$ = INKEY$
550 IF A$ = CHR$(13) THEN RETURN
560 GOTO 540
570 :
580 REM *** INITIALIZEREN ***
590 REM ** VARIABELEN EEN STARTWAARDE GEVEN **
600 C = 3.14159 / 180           : `GRADEN NAAR
RADIAAL
610 G = 1.62                   : `GRAVITATIE
MAAN
620 S = 0                       : `VERSTREKEN
TIJD
630 T = 1                       : `TIJDSINTERVAL =
1 sec.
640 K = 0                       : `HULPTELLER
650 Z = INT(RND(-TIME) * 50/G + 5) : `WILLEKEURIGE
TIJD
660 M = INT(RND(-TIME) * 500 + 1000) : `TOTALE MASSA
670 M2 = INT(.6 * M)           : `HOEVEELHEID
BRANDSTOF
680 M1 = M - M2                : `LEEGGEWICHT
LANDER
690 F = INT(M * G * (RND(-TIME) * 3 + 5)) : `TOTALE
STUWDruk
700 A = (F/M - G) * (RND(-TIME) + 1) * (-1) : `INIT.
ACCELERATIE
710 Y = (INT(.5 * ABS(A) * Z^2/10)) * 10 : `STARThOOGTE

```



```

720 K = K + 1                                : 'HULPTELLER
730 IF K < 10 GOTO 750
740 Y = (RND(-TIME) + 1) * 300
750 IF (Y > 3000) OR (Y < 300) GOTO 650
760 X = (INT ((RND(-TIME) * Y/2 + Y/2)/10)) * 10
: 'LANDINGSPL.
770 B1 = (M2/90) * (RND(-TIME) + 1)
: 'BRANDSTOFFACTOR
780 R = INT(RND(-TIME) * 180 + 90)           : 'RICHTING
RUIIMTEVAARTUIG
790 IF R => 180 GOTO 810
800 X = X * (-1)
810 U = RND(-TIME) * 100                     : 'PERCENTAGE
STUWDRUK
820 V1 = 0                                   : 'STARTSNELHEID
HOR.
830 V2 = -50                                 : 'STARTSNELHEID
VERT.
840 RETURN
850 :
860 REM *** BEREKENING ***
870 REM ** STUWDRUK **
880 U = U / 100                              : 'PERC. WORDT
BREUK
890 F1 = U * F                               : 'ACTIEVE
STUWDRUK
900 R1 = SIN(R * C) * F1                     : 'HORIZONTALE
DEEL
910 R2 = COS(R * C) * F1                     : 'VERTICALE DEEL
920 V1 = V1 + T * (R1/M) / 2
: 'HORIZ.GEM.SNELH.
930 V2 = V2 + T * (R2/M - G) / 2
: 'VERT.GEM.SNELH.
940 V = (SQR(V1^2 + V2^2)) * 3.6             : 'SNELHEID KM/H
950 X = X + V1 * T                           : 'AFSTAND IN
METERS
960 Y = Y + V2 * T                           : 'HOOGTE IN
METERS
970 IF Y > 5 GOTO 990                        : 'GELAND HOE?
980 Y = 0
990 S = S + T                                 : 'AANTAL
VLUCHTSEC.
1000 B = (U^2*.75+U*.25)*T*B1
: 'BRANDSTOFVERBR.
1010 M2 = INT(M2 - B)                        : 'RESTANT
BRANDSTOF
1020 IF M2 > 0 GOTO 1070
1030 REM ** BRANDSTOF UITGEPUT **
1040 M2 = 0
1050 F = 0
1060 T = 1                                    : 'TIJDSINT. = 1
sec.
1070 M = M1 + M2                             : 'MASSA
RUIIMTEV.
1080 RETURN
1090 :
1100 REM *** STRINGS WISSEN ***

```

```

1110 H = 0
1120 B$ = ""
1130 C$ = ""
1140 D$ = ""
1150 RETURN
1160 :
1170 REM *** UITVOER ***
1180 OPEN "GRP:" AS#1:SCREEN 2:COLOR 14,1,1:CLS:VL = 1
1190 LINE(0,0)-(238,11),14,BF:COLOR 1
1200 LINE(239,0)-(255,11),15,B
1210 DRAW"BM4,3":PRINT#1,"MAANLANDER - VERSTREKEN TIJD:"
1220 DRAW"BM5,3":PRINT#1,"MAANLANDER - VERSTREKEN TIJD:"
1230 LINE(0,12)-(255,192),15,B
1240 COLOR 14:CIRCLE(128,312),181:CIRCLE(128,312),182:COLOR
15
1250 CIRCLE(64,158),20,,,,.4:CIRCLE(132,140),25,,,,.3:CIRCLE
(178,156),21,,,,.25
1260 DRAW"BM4,16":PRINT#1,"HOOGTE           :"
1270 DRAW"BM4,26":PRINT#1,"AFSTAND           :"
1280 DRAW"BM4,36":PRINT#1,"SNELHEID KM/H       :"
1290 DRAW"BM4,46":PRINT#1,"HOR.SNELHEID M/S     :"
1300 DRAW"BM4,56":PRINT#1,"VERT.SNELHEID M/S    :"
1310 DRAW"BM4,66":PRINT#1,"BRANDSTOF          :"
1320 DRAW"BM4,78":PRINT#1,B$
1330 LINE(240,1)-(254,10),1,BF
1340 DRAW"bm233,3":PRINT#1,USING"###";S
1350 LINE(1,168)-(254,190),4,BF
1360 DRAW"BM72,170":PRINT#1,C$
1370 DRAW"BM73,170":PRINT#1,C$
1380 DRAW"BM32,180":PRINT#1,D$
1390 LINE(175,16)-(230,76),1,BF
1400 DRAW"BM175,16":PRINT#1,USING"#####";INT(Y);:PRINT#1,"
M"
1410 DRAW"BM175,26":PRINT#1,USING"#####";INT(X);:PRINT#1,"
M"
1420 DRAW"BM175,36":PRINT#1,USING"#####";INT(V)
1430 DRAW"BM175,46":PRINT#1,USING"#####";INT(V1)
1440 DRAW"BM175,56":PRINT#1,USING"#####";INT(V2)
1450 DRAW"BM175,66":PRINT#1,USING"#####";INT(M2)
1460 RETURN
1470 :
1480 REM *** INVOER ***
1490 DRAW"BM24,170":PRINT#1,"TIJD  STUWDR.  RICHT."
1500 DRAW"BM25,170":PRINT#1,"TIJD  STUWDR.  RICHT."
1510 LINE(20,180)-(240,189),4,BF
1520 REM *** INVOER VOOR SCHERM 2 ***
1530 S$ = CHR$(132):Z$ = CHR$(252):N = 0
1540 SPRITE$(1) = STRING$(5,CHR$(0))+Z$+S$+Z$
1550 X=32:Y=180:X1=32
1560 PSET(X,Y),4
1570 PUT SPRITE 1, (X,Y-1),15,1
1580 A$ = INPUT$(1)
1590 IF A$ = CHR$(13) THEN GOTO 1830
1600 IF A$ = CHR$(8) THEN GOSUB 1720
1610 IF A$ <> "" THEN GOSUB 1660
1620 AA$ = AA$ + A$
1630 GOTO 1580

```

```

1640 :
1650 REM *** PLAATSEN VAN CURSOR ***
1660 PSET(X,Y),4
1670 PRINT#1,A$:X=X+6
1680 PUT SPRITE 1,(X,Y-1),15,1
1690 RETURN
1700 :
1710 REM *** BACKSPACE ***
1720 X = X - 6
1730 IF X <= X1 THEN X = X1
1740 PSET(X,Y),4
1750 PUT SPRITE 1,(X,Y-1),15,1
1760 LINE(X,Y)-(X+6,Y+8),4,BF
1770 IF AA$ = "" THEN GOTO 1800
1780 AA$ = LEFT$(AA$,LEN(AA$)-1)
1790 A$ = ""
1800 RETURN
1810 :
1820 REM *** EIND INVOER ***
1830 PUT SPRITE 1,(X,Y-1),0,1
1840 N = N + 1
1850 IF N = 1 THEN A$ = "":A1 = VAL(AA$):AA$ = "":X = 86:X1
= 86:GOTO 1570
1860 IF N = 2 THEN A$ = "":A2 = VAL(AA$):AA$ = "":X =
158:X1 = 158:GOTO 1570
1870 IF N = 3 THEN A$ = "":A3 = VAL(AA$):AA$ = "":GOTO
1900
1880 GOTO 1880
1890 REM *** CONTROLE ***
1900 IF A1 < 0 GOTO 1920
1910 T = A1
1920 IF (A2 < 0) OR (A2 > 100) GOTO 1940
1930 U = INT(A2)
1940 IF (A3 < -180) OR (A3 > 180) GOTO 1960
1950 R = A3
1960 RETURN
1970 :
1980 REM *** AANWIJZINGEN ***
1990 IF M2 > 0 GOTO 2010
2000 B$ = "BRANDSTOF UITGEPUT!"
2010 IF Y < 10000 GOTO 2040
2020 D$ = "COMPUTER UITGEVALLEN!"
2030 H = 1
2040 IF Y > 0 GOTO 2120
2050 C$ = " *** CRASH ***"
2060 IF V > 20 GOTO 2080
2070 C$ = "VEILIG GELAND "
2080 D$ = "IN HET LANDINGSGBIED"
2090 IF ABS(X) < 10 GOTO 2110
2100 D$ = "BUITEN HET LANDINGSGBIED"
2110 H = 1
2120 RETURN
2130 REM *** STUURMODULE ***
2140 REM *** INSTRUCTIE ***
2150 GOSUB 60
2160 REM *** INITIALIZEREN ***
2170 GOSUB 580

```



```

2180 REM *** STRINGS WISSEN ***
2190 GOSUB 1110
2200 REM *** BEREKENING ***
2210 GOSUB 880
2220 REM *** AANWIJZINGEN ***
2230 GOSUB 1990
2240 REM *** UITVOER ***
2250 IF VL = 0 THEN GOSUB 1180 ELSE GOSUB 1320
2260 IF H > 0 GOTO 2310
2270 IF M2 = 0 GOTO 2190
2280 REM *** INVOER ***
2290 GOSUB 1490
2300 IF (A1 <> -9) OR (A2 <> -9) GOTO 2190
2310 GOTO 2310
2320 REM *** MUZIEK ***
2330 PLAY"V13T120", "V12T120"
2340 PLAY
"04F8G8A8B-805CO4AO5CO4A8R8A16R16A16R16G16R16G16R16FR", "R20
3F8G8A8B-804C8D8E8F8CO3B-AR"
2350
PLAY"04A16R16A16R16G16R16G16R16A16R16A16R16B-A16R16A16R16G1
6R16G16R16A16R16A16R16B-A16R16A16R16G16R16G16R16FR", "03F8C8
E8C8F8C8E8C8F8C8E8C8F8C8E8C8F8C8E8C8F8C8E8C8FR"
2360 RETURN

```

Voorbeeld



3.2 Ruimtevaartuig

Het moederschip van de grote ruimtevaartexpeditie is uitgerust met een aantal ruimtesloepen, van het type HEX 32. Deze ruimtevaartuigen zijn uitermate geschikt voor landingen op de diverse planeten. De expeditie heeft ons eigen zonnestelsel als doelgebied.

Afhankelijk van welke planeet wordt gekozen, kan het moederschip zich al binnen een tiental seconden in een lage parkeerbaan om de gekozen planeet bevinden. De astronauten maken zich gereed om aan boord te gaan van de HEX 32. Ondertussen worden de boordcomputers van de sloep met nieuwe programma's geladen en worden alle systemen voor de laatste keer getest. Nadat een HEX 32 is gelanceerd, start volautomatisch de raketmotor die ervoor zorgt dat het ruimtevaartuig binnen de aantrekkingskracht van de planeet komt. De hele procedure staat onder controle van de eigen boordcomputers. Alhoewel het laatste gedeelte van de afdaling naar de planeet ook wel automatisch kan geschieden, wordt toch vaak op handbesturing overgegaan. De voornaamste reden is wel dat men vaak op locaties landt waar nog niet alle gegevens van bekend zijn. Het zal niet de eerste keer zijn dat een HEX 32 in een pulserend magnetisch veld komt en daardoor onbestuurbaar wordt.

Het ruimtevaartuig wordt voor een bepaald tijdsinterval vooruit geprogrammeerd. Dit gebeurt altijd in seconden waarbij het tijdsinterval kan liggen tussen 0,1 en 100 seconden. Tegelijkertijd dient de astronaut de richting van de stuwdruk en de hoeveelheid stuwdruk op te geven. De stuwdruk wordt opgegeven als percentage van de maximale stuwdruk en de richting daarvan wordt opgegeven in graden. De boordcomputers houden bij wat de effecten zijn op het ruimtevaartuig. Hiertoe wordt een beeldscherm opgebouwd met daarin alle noodzakelijke gegevens.

Links op het beeldscherm wordt het vluchtbeeld weergegeven met de afstand tot de landingsplaats en de hoogte boven het planeetoppervlak. Rechts staat een aantal vluchtgegevens met daarin o.a. een gradenverdeling als geheugensteun voor de bepaling van de richting van de stuwdruk. Tevens wordt de snelheid van het ruimtevaartuig, de verticale en de horizontale snelheid in m/s, de brandstofvoorraad, de hoogte en de afstand gegeven.

Het zal de eerste keer niet meevallen om het ruimtevaartuig veilig op de grond te zetten. Mocht de brandstof opraken dan slaat men zeker te pletter op het planeetoppervlak.

Suggestie: De planeten waaruit kan worden gekozen, zijn allemaal planeten uit ons eigen zonnestelsel. Het is best wel eens aardig om een planeet toe te voegen met een extreem hoge zwaartekracht. Bijvoorbeeld de planeet Silico (zie programma 3.5).

Programma

```
10 REM *** RUIMTEVAARTUIG ***
20 SCREEN 0,0:WIDTH 40:COLOR 4,10,7:KEY OFF:CLS:BEEP
30 R1$=CHR$(1)+CHR$(87):R2$=CHR$(1)+CHR$(85)
40 DATA PLUTO,4.89, SATURNUS,8.77, MARS,3.84, AARDE,9.81
```



```

50 DATA MAAN,1.62, IO,2.26, EUROPA,1.98, CALLISTO,3.20
60 DATA PALAS,0.54, JUNO,0.21
70 GOTO 3180
80 :
90 REM *** INSTRUCTIE ***
100 CLS                                :REM SCHERM
WISSEN
110 L$ = "":FOR RS=1 TO 39:L$=L$+R1$:NEXT RS
120 LOCATE 0,5
130 PRINT L$
140 PRINT"***** RUIMTEVAARTUIG *****"
150 PRINT L$
160 PRINT
170 PRINT" De bedoeling is dit ruimtevaartuig"
180 PRINT" veilig op een vastgestelde plaats te"
190 PRINT" laten landen (0,0). Het ruimtevaartuig"
200 PRINT" bevindt zich al volledig binnen de"
210 PRINT" aantrekkingskracht van de gekozen"
220 PRINT" planeet."
230 PRINT
240 PRINT L$
250 PRINT" GELEZEN, DRUK DAN <RETURN-TOETS> IN."
260 PRINT L$
270 GOSUB 3910
280 A$ = INKEY$
290 IF A$ <> CHR$(13) GOTO 280
300 CLS
310 PRINT L$
320 PRINT" RUIMTEVAARTUIG          - BESTURING -
330 PRINT L$
340 PRINT
350 PRINT"1. De stuwdruk is te regelen van 1 tot"
360 PRINT"   en met 100% van het motorvermogen."
370 PRINT
380 PRINT"2. De richting waarin de stuwdruk ge-"
390 PRINT"   geven kan worden, is aan te geven"
400 PRINT"   van -180 t/m 180 graden."
410 PRINT"   Recht omhoog is 0 graden, 90 graden"
420 PRINT"   is naar rechts, -90 graden is naar"
430 PRINT"   links en 180 en -180 graden is recht"
440 PRINT"   omlaag. Alle tussenliggende waarden"
450 PRINT"   zijn ook mogelijk."
460 PRINT
470 PRINT"3. Het tijdsinterval moet bij iedere"
480 PRINT"   besturingsopdracht gegeven worden."
490 PRINT"   De tijdseenheid is de seconde."
500 PRINT
510 PRINT L$
520 PRINT" GELEZEN, DRUK DAN <RETURN-TOETS> IN."
530 PRINT L$
540 A$ = INKEY$
550 IF A$ <> CHR$(13) GOTO 540
560 RETURN
570 :
580 REM *** INVOER 1 ***
590 CLS                                :SCHERM
WISSEN

```



```

600 PRINT L$
610 PRINT"***** RUIMTEVAARTUIG *****"
620 PRINT L$
630 PRINT
640 PRINT" Op de volgende hemellichamen zijn"
650 PRINT" landingsplaatsen:"
660 LOCATE ,10
670 REM ** INLEZEN NAMEN HEMELLIJCHAMEN **
680 RESTORE
690 FOR I = 1 TO 9 STEP 2
700 READ H$, G, A$, G
710 PRINT I;" ...";TAB(9);H$;
720 PRINT TAB(20) USING"##";I+1;
730 PRINT TAB(23);" ...";TAB(29);A$
740 NEXT I
750 REM ** KEUZE **
760 LOCATE ,19
770 PRINT L$
780 PRINT" WELKE HEMELLIJCHAAM >KEUZENR<"
790 PRINT L$
800 LOCATE 32,20:INPUT K
810 IF (K < 1) OR (K > 10) THEN LOCATE 32,20:PRINT"
":GOTO 800
820 RETURN
830 :
840 REM *** INITIALIZEREN ***
850 REM ** OPNIEUW LEZEN INTERNE DATAFILE **
860 RESTORE :`DATAPOINTER VOORAAN
870 FOR I = 1 TO K
880 READ H$,G :`INLEZEN NAAM +
GRAVITATIE
890 NEXT I
900 REM ** VARIABELEN EEN STARTWAARDE GEVEN **
910 C = 3.14159/180 :`GRADEN NAAR RADIAAL
920 S = 0 :`VERSTREKEN TIJD
930 T = 1 :`TIJDSINTERVAL = 1
sec
940 K = 0 :`HULPTELLER
950 Z = INT(RND(-TIME) * 50/G + 5) :`WILLEKEURIGE TIJD
960 M = INT(RND(-TIME) * 500 + 1000) :`TOTALE MASSA
970 M2 = INT(.6 * M) :`HOEVEELHEID
BRANDSTOF
980 M1 = M - M2 :`LEEGGEWICHT
RUIMTEVAART.
990 F = INT(M * G * (RND(-TIME) * 3 + 5)) :`TOTALE
STUWDRIJK
1000 A = (F/M - G) * (RND(-TIME) + 1) * (-1) :`INIT.
ACCELERATIE
1010 Y = (INT(.5 * ABS(A) * Z^2/10)) * 10 :`STARTHOOGTE
1020 K = K + 1 :`HULPTELLER
1030 IF K < 10 GOTO 1050
1040 Y = (RND(-TIME) + 1) * 300
1050 IF (Y > 3000) OR (Y < 300) GOTO 950
1060 X = (INT((RND(-TIME) * Y/2 + Y/2)/10)) * 10
: `LANDINGSPL.
1070 B1 = (M2/90) * (RND(-TIME) + 1) :`BRANDSTOFFACTOR
1080 R = INT(RND(-TIME) * 180 + 90) :`RICHTING

```

```

RUIMTEVAART.
1090 IF R >= 180 GOTO 1110
1100 X = X * (-1)
1110 U = RND(-TIME) * 100           : 'PERCENTAGE
STUWDRUK
1120 V1 = 0                        : 'STARTSNELHEID HOR.
1130 V2 = -50                      : 'STARTSNELHEID
VERT.
1140 VLAG = 0
1150 RETURN
1160 :
1170 REM *** BEREKENING ***
1180 REM ** STUWDRUK **
1190 U = U/100                      : 'PERC. WORDT BREUK
1200 F1 = U * F                     : 'ACTIEVE STUWDRUK
1210 R1 = SIN(R * C) * F1           : 'HORIZONTALALE DEEL
1220 R2 = COS(R * C) * F1           : 'VERTICALE DEEL
1230 V1 = V1 + T * (R1/M)/2         : 'HORIZ.GEM.SNELH.
1240 V2 = V2 + T * (R2/M - G)/2     : 'VERT.GEM.SNELH.
1250 V = (SQR(V1^2 + V2^2)) * 3.6   : 'SNELHEID KM/H
1260 X = X + V1 * T                 : 'AFSTAND IN METERS
1270 Y = Y + V2 * T
1280 IF Y > 5 GOTO 1300             : 'GELAND HOE?
1290 Y = 0
1300 S = S + T
1310 B = (U^2 * .75 + U * .25) * T * B1 : 'BRANDSTOFVERBR.
1320 M2 = INT(M2 - B)               : 'RESTANT BRANDSTOF
1330 IF M2 > 0 GOTO 1370
1340 M2 = 0
1350 F = 0
1360 T = 1                          : 'TIJDSINT. = 1 sec
1370 M = M1 + M2                    : 'MASSA RUIMTEV.
1380 RETURN
1390 :
1400 REM *** STRINGS WISSEN ***
1410 H = 0
1420 B$ = ""
1430 C$ = ""
1440 D$ = ""
1450 E$ = ""
1460 F$ = ""
1470 RETURN
1480 :
1490 REM *** OPBOUWEN PLAATJE ***
1500 REM ** Y-WAARDEN **
1510 FOR I = 1 TO 5
1520 N = 10^I                       : 'BOVENGRENS GRAFIEK
1530 Y1 = INT(Y/N)
1540 IF Y1 > 0 GOTO 1560
1550 I = 5
1560 NEXT I
1570 Y2 = INT(Y * 10/N + .5)        : 'Y-COORDINAAT
1580 Y3 = N/10                      : 'STAPGROOTTE
1590 REM ** X-WAARDEN **
1600 REM ** RUIMTEVAARTUIG DICHT BIJ LANDINGSPLAATS **
1610 K = ABS(X)
1620 X1 = -50                       : 'ONDERGRENS X

```

```

1630 X3 = 4 : 'STAPGROOTTE
1640 X4 = INT((X + 50)/4 + .5)
1650 REM ** LANDINGSPLAATS VERDERWEG **
1660 FOR I = 2 TO 5
1670 K = ABS(X)
1680 X2 = 10^I : 'BOVENGRENS X
1690 K = INT(K/X2)
1700 IF K > 0 GOTO 1820
1710 X1 = 0
1720 IF X > 0 GOTO 1770
1730 REM ** LANDINGSPLAATS LIGT LINKS **
1740 REM * GRENZEN WISSELEN *
1750 X1 = X2 * (-1)
1760 X2 = 0
1770 I = 5
1780 X3 = ABS(X2 - X1)/25 : 'STAPGROOTTE
1790 X4 = INT(X/X3 + .5) : 'X-COORDINAAT
1800 IF X > 0 GOTO 1820
1810 X4 = 25 + X4 : 'VERSCHUIVEN
1820 NEXT I
1830 REM ** PLAATS IN PLAATJE **
1840 P = 100 - Y2 * 8 : 'Y-COORDINAAT
1850 P1 = 36 + X4 * 8 : 'X-COORDINAAT
1860 REM ** TEKSTEN IN PLAATJE **
1870 E$ = "(Y*1 M)"
1880 IF Y < 1000 GOTO 1930
1890 E$ = "(Y*1.000M)"
1900 N = N/1000
1910 Y3 = Y3/1000
1920 REM ** X-AS **
1930 F$ = "(X*1 M)"
1940 IF ABS(X) < 1000 GOTO 2010
1950 F$ = "(X*1.000M)"
1960 X1 = X1/1000
1970 IF ABS(X1) > .9 GOTO 1990
1980 X1 = 0
1990 X2 = X2/1000
2000 X3 = X3/1000
2010 RETURN
2020 :
2030 :
2040 REM *** AFDRUK RUIMTEVAARTUIG ***
2050 PUT SPRITE 1, (P1,P), 5, 1
2060 RETURN
2070 :
2080 REM *** UITVOER *
2090 OPEN "GRP:" AS #1
2100 SCREEN 2:COLOR 1,4,4:CLS
2110 SPRITE$(1)=CHR$(&B01111110)+CHR$(&B11000011)+CHR$(&B001
11100)+CHR$(&B01111110)+CHR$(&B11111111)+CHR$(&B00100100)+CH
R$(&B01000010)+CHR$(&B10000001)
2120 LINE(0,0)-(255,10),5,B
2130 FOR I =8 TO 9
2140 DRAW"BM=I;,2"
2150 PRINT#1,"RUIMTEVAARTUIG -TIJD-"
2160 NEXT I
2170 REM ** PLAATJE **

```



```

2180 Y5 = 22: X5 = 32
2190 FOR I = 0 TO 10
2200
TE$=R2$+R1$+R1$+R1$+R1$: A$=TE$+TE$+TE$+TE$+TE$+CHR$(1)+CHR$(
(83)
2210 GOSUB 3080
2220 NEXT I
2230 LINE(0,120)-(255,191),5,BF
2240 LINE(2,159)-(253,189),4,BF
2250 FOR I = 4 TO 5
2260 I1 = 131 + I
2270 DRAW"BM=I;,122":PRINT#1,"HOOGTE      ":"
2280 DRAW"BM=I1;,122":PRINT#1,"AFSTAND  ":"
2290 DRAW"BM=I;,130":PRINT#1,"SNELH.KM/H:"
2300 DRAW"BM=I1;,130":PRINT#1,"BRANSTOF:"
2310 DRAW"BM=I;,142":PRINT#1,"HOR.SNELHEID M/S:"
2320 DRAW"BM=I;,150":PRINT#1,"VER.SNELHEID M/S:"
2330 NEXT I
2340 BEEP
2350 SOUND 7,&B00111100
2360 SOUND 0,100:SOUND 3,100
2370 SOUND 8,26:SOUND 9,26
2380 SOUND 12,35
2390 SOUND 13,14
2400 SOUND 11,0
2410 :
2420 Q = -1
2430 LINE(215,1)-(248,9),3,BF
2440 DRAW"BM218,2":PRINT#1,USING"####";S
2450 LINE(0,11)-(255,20),5,BF
2460 DRAW"BM4,11":PRINT#1,E$
2470 DRAW"BM96,11":PRINT#1,H$
2480 DRAW"BM170,11":PRINT#1,F$
2490 LINE(0,22)-(31,119),4,BF
2500 Y5 = 22
2510 FOR I = 0 TO 10
2520 K = N - Y3 * I
2530 DRAW"BM0,=Y5;":PRINT#1,USING"####";INT(K)
2540 Y5 = Y5 + 8
2550 NEXT I
2560 K = 28
2570 LINE(0,112)-(255,120),4,BF
2580 FOR I = 0 TO 5
2590 Q = X1 + I * X3 * 5
2600 IF Q = 1000 THEN K = 218
2610 DRAW"BM=K;,112":PRINT#1,Q
2620 K = K + 40
2630 NEXT I
2640 LINE(94,122)-(130,140),5,BF
2650 LINE(210,122)-(255,140),5,BF
2660 LINE(150,142)-(255,158),5,BF
2670 DRAW"BM94,122":PRINT#1,USING"#####";INT(Y)
2680 DRAW"BM210,122":PRINT#1,USING"#####";INT(X)
2690 DRAW"BM94,130":PRINT#1,USING"#####";INT(V)
2700 DRAW"BM210,130":PRINT#1,USING"#####";INT(M2)
2710 DRAW"BM150,142":PRINT#1,USING"#####";INT(V1)
2720 DRAW"BM150,150":PRINT#1,USING"#####";INT(V2)

```

```

2730 GOSUB 2050
2740 VLAG = 1
2750 RETURN
2760 REM *** INVOER ***
2770 DRAW"BM16,173":NI = 0
2780 PRINT#1,"TIJD STUWDR. RICHT."
2790 DRAW"BM17,173"
2800 PRINT#1,"TIJD STUWDR. RICHT."
2810 LINE(15,181)-(240,189),4,BF
2820 GOTO 3480
2830 REM ** CONTROLE **
2840 IF A1 < 0 GOTO 2860
2850 T = A1
2860 IF (A2 < 0) OR (A2 > 100) GOTO 2880
2870 U = INT(A2)
2880 IF (A3 < -180) OR (A3 > 180) GOTO 2900
2890 R = A3
2900 RETURN
2910 :
2920 REM *** AANIJZINGEN ***
2930 IF M2 > 0 GOTO 2950
2940 B$ = "BRANDSTOF UITGEPUT"
2950 IF Y < 10000 GOTO 2980
2960 D$ = "COMPUTER UITGEVALLEN"
2970 H = 1
2980 IF Y > 0 GOTO 3060
2990 C$ = "*** CRASH ***"
3000 IF V > 20 GOTO 3020
3010 C$ = "VEILIG GELAND"
3020 D$ = "IN HET LANDINGSGEBIED"
3030 IF ABS(X) < 10 GOTO 3050
3040 D$ = "BUITEN HET LANDINGSGEBIED"
3050 H = 1
3060 RETURN
3070 :
3080 REM *** PRINTEN OP SCHERM 2 ***
3090 FOR A = 1 TO 52
3100 DRAW"BM=X5; ,=Y5;"
3110 PRINT#1,MID$(A$,A,1);:X5 = X5 + 4
3120 NEXT A
3130 Y5 = Y5 + 8:X5 = 32
3140 RETURN
3150 :
3160 REM **** STUURMODULE ****
3170 REM *** INSTRUCTIE ***
3180 GOSUB 100
3190 REM *** INVOER 1 ***
3200 GOSUB 590
3210 REM *** INITIALIZEREN ***
3220 GOSUB 860
3230 REM *** STRIING WISSEN ***
3240 GOSUB 1410
3250 REM *** BEREKENING ***
3260 GOSUB 1190
3270 REM *** OPBOUWEN PLAATJE ***
3280 GOSUB 1510
3290 REM *** AANWIJZINGEN ***

```

```

3300 GOSUB 2930
3310 REM *** UITVOER ***
3320 IF VLAG = 0 THEN GOSUB 2090 ELSE GOSUB 2420
3330 IF H > 0 GOTO 3380
3340 IF M2 = 0 GOTO 3240
3350 REM *** INVOER ***
3360 GOSUB 2770
3370 IF (A1 <> -9) OR (A2 <> -9) GOTO 3240
3380 LINE(3,160)-(253,189),6,BF
3390 DRAW"BM24,161":PRINT#1,B$
3400 DRAW"BM25,161":PRINT#1,B$
3410 DRAW"BM72,171":PRINT#1,C$
3420 DRAW"BM73,171":PRINT#1,C$
3430 DRAW"BM24,181":PRINT#1,D$
3440 DRAW"BM25,181":PRINT#1,D$
3450 IF C$ = "*** CRASH ***" THEN GOSUB 3830
3460 GOTO 3460
3470 REM *** INVOER VOOR SCHERM 2 ***
3480 S$ = CHR$(132):Z$ = CHR$(252)
3490 SPRITE$(3) = STRING$(5,CHR$(0))+Z$+S$+Z$
3500 X=22:Y=181:X1=22
3510 PUT SPRITE 3,(X,Y-1),1,3
3520 A$ = INPUT$(1)
3530 IF A$ = CHR$(13) THEN GOTO 3770
3540 IF A$ = CHR$(8) THEN GOSUB 3660
3550 IF A$ <> "" THEN GOSUB 3600
3560 AA$ = AA$ + A$
3570 GOTO 3520
3580 :
3590 REM *** PLAATSEN VAN CURSOR ***
3600 PSET(X,Y),4
3610 PRINT#1,A$:X=X+6
3620 PUT SPRITE 3,(X,Y-1),1,3
3630 RETURN
3640 :
3650 REM *** BACKSPACE ***
3660 X = X - 6
3670 IF X <= X1 THEN X = X1
3680 PSET(X,Y),4
3690 PUT SPRITE 3,(X,Y-1),1,3
3700 LINE(X,Y)-(X+6,Y+8),4,BF
3710 IF AA$ = "" THEN GOTO 3740
3720 AA$ = LEFT$(AA$,LEN(AA$)-1)
3730 A$ = ""
3740 RETURN
3750 :
3760 REM *** EIND INVOER ***
3770 PUT SPRITE 3,(X,Y-1),0,3
3780 NI = NI + 1
3790 IF NI = 1 THEN A1 = VAL(AA$):A$ = "":AA$ = "":X =
81:X1 = 81:GOTO 3510
3800 IF NI = 2 THEN A2 = VAL(AA$):A$ = "":AA$ = "":X =
161:X1 = 161:GOTO 3510
3810 IF NI = 3 THEN A3 = VAL(AA$):A$ = "":AA$ = "":GOTO
2840
3820 REM *** EXPLOSIE ***
3830 SOUND 6,20

```




```

3840 SOUND 7,55
3850 SOUND 13,0
3860 FOR CR = 15 TO 0 STEP -.02
3870 SOUND 8,CR
3880 NEXT CR
3890 RETURN
3900 REM *** INTRODUCTIE MUZIEK ***
3910 PLAY"V13T90","V9T90","V9T90"
3920 PLAY"04F+8E8F+8G8B8AR8O5D16R16D16R16C+16R16O4B16R16B8AR
8","03D8A16R16A16R16A16R16D8A16R16A16R16A16R16G8O4D16R16D16R
16D16R16O3D8A16R16A16R16A16R16","03R8F+16R16F+16R16F+16R16R8
F+16R16F+16R16F+16R16R8B16R16B16R16B16R16R8F16R16F+16R16F+16
R16"
3930 PLAY"05D16R16D16R16C+16R16O4B16R16B8AR8G8BG8F+8AF+8","O
3G8O4D16R16D16R16D16R16D8A16R16A16R16A16R16C+8A16R16A16R16A1
6R16D8A16R16A16R16A16R16","03R8B16R16B16R16B16R16R8F+16R16F+
16R16F+16R16R8E16R16E16R16E16R16"
3940 PLAY"04E8F+8G8B16R16B8AR8G8BG8F+8AF+8","02A8O3A16R16A16
R16A16R16D8A16R16A16R16A16R16C8A16R16A16R16A16R16D8A16R16A16
R16A16R16","03R8E16R16E16R16E16R16R8F+16R16F+16R16F+16R16R8E
16R16E16R16E16R16R8F+16R16F+16R16F+16R16"
3950 PLAY"04E8B8A8C+8E8DR8","02A8O3G8C+8G8F+16R16F+R8","03R2
D16R16DR8"
3960 RETURN

```

Voorbeeld



3.3 De blijde


De blijde werd niet alleen op de planeet Aarde als wapentuig gehanteerd, maar zoals uit het Galactisch jaarboek blijkt was het op de planeet Dzin van het stelsel 3c236 ook al een geliefd wapen om stamvetes mee uit te vechten.

Een blijde is een soort katapult waarmee stenen van enkele tonnen naar de tegenstander kunnen worden geslingerd. Op Dzin heeft men de blijde ver- gaand geperfectioneerd. Men is nu in staat om de werphoek en ook de werp- kracht exact in te stellen. Het gevolg is dat de trefkans zeer hoog wordt wan- neer men zo'n blijde laat instellen door een 'Dzjinball'.

Direct na Djinwende (zoiets als zonnewende) worden de blijdewedstrijden ge- houden, vete of geen vete. Deze wedstrijden worden gehouden in het Kooi- land. Een vlak gebied met in het midden een inslagkratertje. Met het kratertje als middelpunt wordt een roos aangebracht, net zoals op aarde bij het dart- spel. Kan men zijn steen in het kratertje werpen dan krijgt men 100 punten. Komt men in één van de buitenste ringen dan krijgt men minder punten. Degene met het hoogste aantal punten heeft gewonnen en is daarmee Dzjin- ball (een soort schutterskoning).

Om een eerlijk spelverloop te krijgen, staat de blijde bij elke speler op een an- dere afstand.

Suggestie 1: In het programma wordt op het einde een scorebord getoond. Het zou leuk zijn als de scores van de vorige spelers in het beeldscherm wor- den getoond als de volgende speler aan de beurt is.



Het scorebord kan dan na afloop van het spel worden uitgebreid met een stukje programma om de winnaar aan te wijzen.

Suggestie 2: Het programma geeft alleen een plaatje 'van bovenaf'. Met een kleine uitbreiding en enkele formules kan men de baan van de steen op het beeldscherm zetten. Op zo'n manier krijgt men een plaatje van opzij. De sprites van de blijde zijn al aanwezig.

Suggestie 3: Het programma vraagt om de hoek en kracht waarmee de steen de blijde moet verlaten. Moeilijker wordt het als ook de richting van de worp in het programma wordt verwerkt. Als informatie voor de speler moet het programma de afstand van de inslag tot de roos geven en de richting van de inslag ten opzichte van de roos.

Deze extra mogelijkheid geeft meer dimensie aan het spel.

Programma

```
10 REM *** DE BLIJDE ***
20 SCREEN 0:KEY OFF:WIDTH 40:COLOR 15,12,12:CLS
30 DIM A(10)
40 GOTO 1990
50 :
60 REM *** INSTRUCTIES ***
70 L$ = "":FOR RS=1 TO 39:L$=L$+CHR$(1)+CHR$(87):NEXT RS
80 LOCATE 0,2:PRINT L$
90 PRINT"***** DE BLIJDE *****"
100 PRINT L$
110 LOCATE 0,7
120 PRINT"        Jaarlijks na Dzinwende worden de"
130 PRINT"Blijde-wedstrijden gehouden tussen ver-"
140 PRINT"schillende stammen. U bent aanvoerder"
150 PRINT"van een van de ploegen.
160 PRINT"De blijde is een soort katapult,waarmee"
170 PRINT"stenen naar de vijand worden geschoten."
180 PRINT"De wedstrijden zijn wat vriendelijker."
190 PRINT"Er wordt alleen om de eer gespeeld."
200 PRINT"Voor deze wedstrijden heeft men op een"
210 PRINT"vlak terrein een grote roos aangebracht"
220 LOCATE 0,19:PRINT L$
230 PRINT" GEEF <RETURN> OM INSTRUCTIES TE LEZEN."
240 PRINT L$
250 GOSUB 2400
260 A$ = INKEY$
270 IF A$ <> CHR$(13) GOTO 260
280 CLS
290 LOCATE 0,0:PRINT L$
300 PRINT" * DE LEPELBIJDE *          -INSTRUCTIES-"
310 PRINT L$
320 LOCATE 0,4
330 PRINT"De bedoeling is een steen van vast ge-"
340 PRINT"wicht in een klein kratertje te schiet-"
350 PRINT"en. Het kratertje wordt aangegeven met"
360 PRINT"de kleur blauw."
```



```

370 PRINT"Om de blijde te kunnen richten, moet U"
380 PRINT"twee opdrachten geven:"
390 PRINT
400 PRINT"1. De hoek, waaronder de steen de blij-"
410 PRINT"    de verlaat."
420 PRINT"2. Een percentage van de maximale werp-"
430 PRINT"    kracht."
440 PRINT
450 PRINT"De partij met de meeste punten is"
460 PRINT"winnaar. Elke beurt bestaat uit 5 worp-"
470 PRINT"en. Dit spel is met max. 10 personen te"
480 PRINT"spelen."
490 LOCATE 0,21:PRINT L$
500 LOCATE 1,22:INPUT"MET HOEVEEL SPELERS SPEELT U ";Q
510 IF (Q <= 0) OR (Q > 10) GOTO 500
520 RETURN
530 :
540 REM *** INITIALISEREN ***
550 FL = 0
560 REM ** SCHOONVEGEN MAT A **
570 FOR I1 = 1 TO 10
580 A(I1) = 0
590 NEXT I1
600 REM ** VARIABELEN STARTWAARDE GEVEN **
610 C = 3.14159277#/180                : `GRADEN -
RADIALEN
620 OPEN"GRP:" AS#1:COLOR 4,5,1:CLS:SCREEN 2,1
630 RETURN
640 :
650 REM *** UITVOER ***
660 CLS
670 :
680 TS$="** De Blijde **"
690 FL = 1
700 COLOR 1
710 FOR A=0 TO 1 STEP .5
720   CIRCLE(188,36),65-38*A,1,,.4
730 NEXT A
740 LINE(4,66)-(251,146),4,BF
750 LINE(150,70)-(246,100),3,BF
760 CIRCLE(188,36),3,,.4:PAINT(188,36),1
770 HO=4:VE=2:GOSUB 1040
780 HO=5:VE=2:GOSUB 1040
790 COLOR 1:HO=120:VE=14:TS$="+05":GOSUB 1030
800 HO=145:VE=17:TS$="+10":GOSUB 1030
810 HO=160:VE=24:TS$="+20":GOSUB 1040
820 FOR I = 8 TO 9
830 HO=150+I:VE=76:TS$="SPELER :":GOSUB 1040
840 HO=150+I:VE=88:TS$="BEURT :":GOSUB 1040
850 HO=I:VE=154:TS$="WAT IS DE HOEK VAN DE WORP <1-89>
:" :GOSUB 1040
860 HO=I:VE=166:TS$="WAT IS HET PERCENTAGE WERPKRACHT
:" :GOSUB 1040
870 NEXT I
880 HO=I:VE=108:TS$="Aantal gescoorde punten :":GOSUB
1040
890 HO=I:VE=116:TS$="Achter het DOEL          :":GOSUB

```

```

1040
900 HO=I:VE=124:TS$="Hoek,waaronder afgevuurd ":"GOSUB
1040
910 HO=I:VE=132:TS$="Percentage werpkracht ":"GOSUB
1040
920 SPRITE$(0) =
CHR$(&H80)+CHR$(&HC0)+CHR$(&H20)+CHR$(&H10)+CHR$(&HE)+CHR$(
&HFF)+CHR$(&HA5)+CHR$(&H42)
930 PUT SPRITE 0,(16,28),1,0
940 LINE(210,76)-(240,97),3,BF
950 LINE(169,108)-(250,141),4,BF
960 DRAW"BM210,76":PRINT#1,USING"##";I2
970 DRAW"BM210,88":PRINT#1,USING"##";B
980 DRAW"BM170,108":PRINT#1,USING"#####";A(I2)
990 DRAW"BM170,116":PRINT#1,USING"#####";ABS(L)
1000 DRAW"BM170,124":PRINT#1,USING"#####";H
1010 DRAW"BM170,132":PRINT#1,USING"#####";P
1020 RETURN
1030 `**SCHRIJFROUTINE VOOR SCHERM 2**
1040 FOR T=1 TO LEN(TS$)
1050 DRAW"BM=HO;,-VE;"
1060 PRINT#1,MID$(TS$,T,1);:HO=HO+6
1070 NEXT T
1080 RETURN
1090 GOTO 1170
1100 REM *** INVOER ***
1110 VLAG = 0
1120 IF FL = 0 THEN RETURN
1130 H10 = 220:V1E = 154:AA$ = "":GOTO 1150
1140 H10 = 220:V1E = 166:AA$ = "":VLAG = 1
1150 COLOR 1:DRAW"BM=H10;,-V1E;":PRINT#1,CHR$(192)
1160 A1$ = INKEY$
1170 IF A1$ = CHR$(13) AND VLAG = 0 THEN H = VAL(AA$):GOTO
1140
1180 IF A1$ = CHR$(13) AND VLAG = 1 THEN P = VAL(AA$):GOTO
1220
1190 IF A1$ >= CHR$(48) AND A$ <= CHR$(57) OR A1$ =
CHR$(45) THEN AA$ = AA$ + A1$:GOSUB 2250
1200 IF A1$ = CHR$(8) THEN DRAW"BM=H10;,-V1E;":GOSUB 2310
1210 GOTO 1160
1220 LINE(220,154)-(253,176),5,BF
1230 RESTORE 1260
1240 FOR SO = 0 TO 13:READ SS
1250 SOUND SO,SS:NEXT SO
1260 DATA 0,0,0,0,0,0,17,7,16,16,16,1,5,1
1270 SPRITE$(0) = CHR$(16)+CHR$(24)+CHR$(8)+CHR$(8)+CHR$(14)
+CHR$(255)+CHR$(165)+CHR$(66)
1280 FOR I = 16 TO 4 STEP -1
1290 PUT SPRITE 0,(I,28),1,0
1300 NEXT I
1310 RETURN
1320 REM *** VOLGENDE SPELER ***
1330 LINE(8,80)-(144,90),4,BF
1340 COLOR 14
1350 HO=8:VE=80:TS$="SPELER":GOSUB 1040
1360 DRAW"BM40,80":PRINT#1,USING"##";I2
1370 HO=60:VE=80:TS$="GEREED.":GOSUB 1040

```

```

1380 COLOR 1
1390 REM ** STARTWAARDEN **
1400 H1 = INT(RND(-TIME) * 40 + 30)           : `HOEK
1410 V = INT(RND(-TIME) * 10 + 10) * 10      : `SNELHEID
1420 P1 = INT(RND(-TIME) * 40 + 40)
1430 S = V * P1 * .01
1440 A = S^2 * SIN(2 * H1 * C)               : `AFSTAND
1450 RETURN
1460 :
1470 REM *** BEREKENING ***
1480 IF FL = 0 THEN RETURN
1490 LINE(10,116)-(164,123),4,BF
1500 S = V * P/100
1510 R = S^2 * SIN(2 * H * C)               : `AFSTAND
1520 HO = 10:VE = 116
1530 L = INT(A - R + .5)                    : `AFSTAND TOT
KRATER
1540 TS$ = "Achter het DOEL"               : "
1550 IF L < 0 GOTO 1570
1560 TS$ = "Voor het DOEL"                 : "
1570 IF ABS(L) > 75 GOTO 1600
1580 TS$ = "MIDDEN IN DE ROOS"            : "
1590 L = 0
1600 X = ABS(L)
1610 GOSUB 1040
1620 Y = 100
1630 IF X = 0 GOTO 1710
1640 Y = 20
1650 IF X <= 375 GOTO 1710
1660 Y = 10
1670 IF X <= 675 GOTO 1710
1680 Y = 5
1690 IF X <= 975 GOTO 1710
1700 Y = 0
1710 SPRITES$(1) = CHR$(0) + CHR$(224) + CHR$(224) + CHR$(224)
) +CHR$(0) + CHR$(0) + CHR$(0) + CHR$(0)
1720 A(I2) = A(I2) + Y                      : ` PUNTEN
1730 REM ** PLAATS IN DE ROOS **
1740 F = INT(L/75) * (-1) + 12
1750 PUT SPRITE 1, (F+174,30),6,1
1760 RESTORE 1790
1770 FOR SO = 0 TO 13:READ SS
1780 SOUND SO,SS:NEXT SO
1790 DATA 0,0,0,0,0,0,31,7,16,16,16,0,60,0
1800 RETURN
1810 :
1820 REM *** RESULTAAT ***
1830 LINE(4,66)-(251,189),10,BF
1840 DRAW"BM32,68":PRINT#1,"UITSLAG VAN DE WEDSTRIJD"
1850 DRAW"BM33,68":PRINT#1,"UITSLAG VAN DE WEDSTRIJD"
1860 DRAW"BM8,80":PRINT#1,"AANTAL DEELNEMERS:";Q
1870 AX = 1
1880 Q=Q-1
1890 FOR I = 92 TO (Q * 11) + 92 STEP 9
1900 DRAW"BM8,=I;":PRINT#1,"AANVOERDER :";AX;
1910 PRINT#1,USING"###";A(AX);
1920 PRINT#1," PUNTEN."

```




```

1930 AX = AX + 1
1940 NEXT I
1950 RETURN
1960 :
1970 REM **** STUURMODULE ****
1980 REM *** INSTRUCTIE ***
1990 GOSUB 50
2000 REM *** INITIALIZEREN ***
2010 GOSUB 570
2020 FOR I2 = 1 TO Q
2030 REM *** VOLGENDE SPELER ***
2040 IF FL = 0 THEN GOSUB 1400
2050 IF FL = 1 THEN GOSUB 1330
2060 FOR B = 1 TO 5
2070 REM *** INVOER ***
2080 GOSUB 1110
2090 REM *** BEREKENING ***
2100 GOSUB 1470
2110 REM *** UITVOER ***
2120 IF FL = 0 THEN GOSUB 660:GOSUB 1330
2130 IF FL = 1 THEN GOSUB 920
2140 NEXT B
2150 REM *** WACHTLUS ***
2160 FOR T = 1 TO 700
2170 NEXT T
2180 NEXT I2
2190 REM *** RESULTAAT ***
2200 GOSUB 1830
2210 BEEP:GOSUB 2400
2220 GOTO 2220
2230 :
2240 REM *** INVOER ROUTINE VOOR SCHERM 2 ***
2250 FOR T=1 TO LEN(A1$)
2260 DRAW"BM=H10; ,=V1E;":COLOR 5:PRINT#1,CHR$(200)
2270 DRAW"BM=H10; ,=V1E;":COLOR 1:PRINT#1,MID$(A1$,T,1):H10
= H10 + 6
2280 NEXT T
2290 DRAW"BM=H10; ,=V1E;":PRINT#1,CHR$(192)
2300 RETURN
2310 Z$ = CHR$(200)
2320 Z = LEN(AA$)
2330 DRAW"BM=H10; ,=V1E; "
2340 COLOR 5:PRINT#1,Z$;
2350 H10 = H10 - 6:DRAW"BM=H10; ,=V1E;":PRINT#1,Z$
2360 AA$ = LEFT$(AA$,Z - 1)
2370 COLOR 1:DRAW"BM=H10; ,=V1E;":PRINT#1,CHR$(192)
2380 RETURN
2390 GOTO 2390
2400 REM *** MUZIEK ***
2410 PLAY"V12T115","V9T115","V8T115"
2420 PLAY"O5CD8.C16O4A8.R16A8R8A8.G16A8.B-16A.R8","O4C8R8C8R
8C8R8C8R8C8R8C8R8C8R8C8R8","O3F8R8C8R8F8R8C8R8F8R8C8R8F8R8C8
R8"
2430 PLAY"O4B-G8.O5C16O4A.R8FD8.G16C.R8","O4D8R8C8R8C8R8C8R8
O3A8R8F8R8G8R8G8R8","O3G8R8C8R8F8R8E8R8D8R8O2G8R8O3C8R8C8R8"
2440 PLAY"O4G.R8G8O5D8C8O4B-8A8G8A8B-8A.R8","O4D8R8D8R8D8R8D
8R8D8R8D8R8D8R8D8R8","O3G8R8G8R8G8R8G8R8F8R8F8R8F8R8F8R8"

```

```
2450 PLAY"04EA8.R16AG+8B805E2", "04C8R8C8R8D8R8D8R8E2", "03E8R
8E8R8E8R8E8R8A2"
2460 RETURN
```

Voorbeeld



3.4 De Compusaurus

In het Dinosgebergte ten zuidwesten van de Voorburgen huist de Compusaurus. Ooit heeft iemand eens geprobeerd zo'n wezen te vangen, maar dat is fataal afgelopen; niet voor de Compusaurus, maar voor de 'vanger'...

De Compusaurus is een bijzonder vraatzuchtig monster dat de passen in het Dinosgebergte onveilig maakt voor reizigers die er doorheen moeten trekken. De passen zijn het jachtgebied van de Compusaurus maar het aantal passen in het Dinosgebergte is zó groot en ze lijken zoveel op elkaar dat de Compusaurus nooit weet bij welke pas hij is.

Is een reiziger een pas eenmaal gepasseerd dan is hij in veiligheid. Niemand weet hoe het komt, maar als een reiziger aan het begin van een pas staat, duikt altijd aan het andere einde de Compusaurus op.

Een pas lijkt wel wat op een speelbord. Zowel de Compusaurus als de reiziger kunnen per stap maar maximaal 8 vakken oversteken. Door een speling van de natuur zijn sommige vakken tevens de deksels van valkuilen. Trapt men op zo'n vak dan vliegt men een tunnel in en men komt op een andere plaats weer boven. Omdat na elke jacht de Compusaurus alle valkuilen weer vergeet, kan ook hij op zo'n valkuildeksel terechtkomen!

De Compusaurus staat de reiziger op vak 100 op te wachten. Kan de reiziger eerder op vak 100 komen dan de Compusaurus op vak 1 dan ligt de weg naar de vrijheid open. De Compusaurus is ondanks alles 'sportief', hij geeft iedereen een eerlijke kans, alleen is hij wel verduiveld slim. Wanneer de reiziger of hijzelf een stap doet, let hij donders goed op wat er gebeurt en onthoudt dit ook.

In de loop van het spel zal hij zich vaak afvragen of hij de reiziger te slim af kan zijn door hem te 'slaan'. De Compusaurus is vaak overmoedig. Juist op die momenten kan hij door de reiziger buiten gevecht worden gesteld.

Suggestie 1: De beslissingen van de Compusaurus zijn afhankelijk van het doorlopen van de beslissingsmodules. Wanneer men in de volgorde waarin de beslissingsmodules worden doorlopen veranderingen aanbrengt, zal ook het gedrag veranderen.

Suggestie 2: Het kan misschien erg leuk zijn om nog een Compusaurus in het spel te betrekken. Het speelveld zal dan wel drastisch moeten worden vergroot.

Programma

```
10 REM *** DE JAGENDE COMPUSAURUS ***
20 TIME = RND(-TIME)
30 SCREEN 0:WIDTH 40:KEY OFF:COLOR 1,14,14:CLS
40 DIM A(100), B(100)
50 REM ** INTERNE DATA-FILE MET JACHTGEBIED **
60 DATA 47, 190, 47, 36, 201, 36, 14, 223,14
70 DATA 14, 223, 14, 37, 201, 36, 47, 190, 47
80 GOTO 3010
90 REM *** SPELREGELS ***
100 CLS : 'SCHERM WISSEN
110 L$="":FOR X=1 TO 39:L$=L$+CHR$(1)+CHR$(87):NEXT X
```



```

120 LOCATE 0,3:PRINT L$
130 PRINT"***** DE COMPUSAURUS *****"
140 PRINT L$:PRINT
150 PRINT
160 PRINT"Men is terecht gekomen in het jachtge-"
170 PRINT"bied van de COMPUSAURUS. Een vraat-"
180 PRINT"zuchtig beest met uitstekende ogen en"
190 PRINT"een bijzonder goed geheugen. Juist dit"
200 PRINT"geheugen maakt hem levensgevaarlijk."
210 PRINT
220 PRINT
230 LOCATE 0,16:PRINT L$
240 PRINT" GELEZEN, DRUK DAN <RETURN-TOETS> IN."
250 PRINT L$
260 A$ = INKEY$:IF A$ <> CHR$(13) GOTO 260
270 CLS
280 LOCATE 0,0:PRINT L$
290 PRINT"***** DE COMPUSAURUS *****"
300 PRINT L$:PRINT
310 PRINT"Men zal moeten proberen van 1 naar 100"
320 PRINT"te komen. De compusaurus probeert het"
330 PRINT"tegenovergestelde te doen, dus van 100"
340 PRINT"naar 1. Per zet kan men maximaal acht"
350 PRINT"stappen voor- of achteruit. Achteruit"
360 PRINT"geeft men met een minteken aan."
370 PRINT"De compusaurus kan verslagen worden,"
380 PRINT"door met het aantal te nemen stappen"
390 PRINT"precies op zijn plaats uit te komen."
400 PRINT"De compusaurus doet dat bij jou ook."
410 PRINT"Let vooral op de tunnels! Het is maar"
420 PRINT"een hint."
430 PRINT:PRINT L$
440 PRINT"DRUK <RETURN> IN, WANNEER MEN KLAAR IS."
450 PRINT L$
460 A$ = INKEY$:IF A$ <> CHR$(13) GOTO 460
470 RETURN
480 REM *** INITIALIZEREN ***
490 OPEN "GRP:" AS#1
500 SCREEN 2,0
510 GOSUB 3560
520 FOR I = 1 TO 100
530 A(I) = I : 'TUNNELS
WISSEN
540 B(I) = 999 : 'GEHEUGEN
C-SAURUS WISSEN
550 NEXT I
560 SPRITE$(0) = CHR$(96) + CHR$(224) +CHR$(16) + CHR$(56)
+ CHR$(24) +CHR$(24) + CHR$(36) + CHR$(99)
570 SPRITE$(1) = CHR$(24) + CHR$(24) + CHR$(16) + CHR$(28)
+ CHR$(16) + CHR$(24) + CHR$(36) + CHR$(54)
580 REM ** TUNNELS PLAATSEN **
590 X = INT(RND(-TIME) * 60 + 25) : 'RANDOM AANTAL
TUNNELS
600 FOR Y = 1 TO X
610 I = INT(RND(-TIME) * 90 + 5) : 'PLAATS
TUNNELS
620 Z = INT(RND(-TIME) * 30 + 1) : 'RELATIEVE

```

```

VERPLAATSING
630 W = INT(RND(-TIME) * 1 + 1.5)           : `WAARDE VOOR
POS OF NEG
640 IF W > 1 GOTO 660
650 Z = Z * (-1)                           : `TERUG !!
660 A(I) = I + Z
670 REM ** CONTROLE OP REL.VERPLAATSING **
680 IF (A(I) < 2) OR (A(I) > 99) GOTO 620
690 NEXT Y
700 REM ** VARIABELEN STARTWAARDE GEVEN **
710 A1 = -1                                 : `ZETTENTELLER
SPELER
720 B = 0                                   : `ZETTENTELLER
COMPUSAURUS
730 D = 1                                   : `STARTPLAATS
SPELER
740 C = 100                                : `STARTPLAATS
COMPUSAURUS
750 E = 0                                   : `VORIGE ZET
COMPUSAURUS
760 REM ** TELLER **
770 GOSUB 810
780 RETURN
790 :
800 REM *** SPELER TELLER + STRING WISSEN ***
810 A1 = A1 + 1
: `ZETTENTELLER PLUS 1
820 F = 0                                   : `STUURVAR.
VOOR UITVOER
830 T$ = ""
840 RETURN
850 :
860 REM *** C-SAURUS TELLER + STRINGS WISSEN ***
870 B = B + 1                               : `OPHOGEN
TELLER
880 F = 1                                   : `STUURVAR.
UITVOER
890 T$ = ""                                 : `STRING
WISSEN
900 RETURN
910 :
920 REM *** UITVOER ***
930 SCREEN 2:COLOR 1,15,3:CLS
940 T = 0
950 LINE(0,0)-(255,10),6,BF
960 LINE(1,34)-(254,51),8,BF
970 LINE(1,53)-(254,125),14,BF
980 LINE(1,127)-(254,190),12,BF
990 FOR I = 1 TO 2
1000 DRAW"BM=I;,2":COLOR 15
1010 PRINT#1,"HET JACHT GEBIED VAN DE C-SAURUS"
1020 NEXT I:COLOR 1
1030 LINE(0,12)-(255,32),1,B
1040 DRAW"BM2,18"
1050 PRINT#1,"Aantal zetten<"
1060 DRAW"BM112,14"
1070 PRINT#1,"Compusaurus:"

```

```

1080 DRAW"BM112,24"
1090 PRINT#1,"Speler      :"
1100 RESTORE 1170
1110 FOR I = 1 TO 6
1120 READ X,X1,Y
1130 FOR J = X TO X1 STEP 11
1140 CIRCLE(J,Y),5,12
1150 NEXT J
1160 NEXT I
1170 DATA 51,194,60,40,205,71,18,237,82,18,237,93,40,205,104
,51,194,115
1180 T = 0
1190 LINE(8,140)-(254,148),12,BF
1200 LINE(112,40)-(128,47),8,BF
1210 DRAW"BM8,40"
1220 PRINT#1,"SAURUS HEEFT ";
1230 PRINT#1,USING"##";G;
1240 PRINT#1," STAPPEN GEDAAN"
1250 RESTORE 60
1260 FOR J = 1 TO 6
1270 JZ = J
1280 JX = 7 - J
1290 IF JZ = 6 THEN JZ = JZ - 1
1300 READ W,Z,DX
1310 FOR I = W TO Z STEP 11
1320 T = T + 1
: `CONTROLE-TELLER
1330 IF D <> T GOTO 1350
1340 PUT SPRITE 1,(I,(J * 11) + 45),4,1
1350 IF C <> T + J GOTO 1370
1360 PUT SPRITE 0,(I + (JZ * 11),122-(JX * 11)),1,0
1370 NEXT I
1380 NEXT J
1390 LINE(216,14)-(250,31),15,BF
1400 COLOR 1
1410 DRAW"BM216,14":PRINT#1,USING"##";B
1420 DRAW"BM216,24":PRINT#1,USING"##";A1
1430 Y1 = 140
1440 FOR I1 = 8 TO 9
1450 X1 = I1:GOSUB 3310
1460 NEXT I1
1470 IF LEFT$(T$,2) = "IK" THEN GOSUB 3560
1480 RETURN
1490 :
1500 REM *** INVOER-SPELER ***
1510 FOR I = 8 TO 9
1520 DRAW"BM=I;,180":PRINT#1,"Hoeveel stappen <max. 8> "
1530 NEXT I
1540 X2 = 210:Y2 = 180
1550 A$ = INKEY$
1560 FOR I = 1 TO 30:NEXT I
1570 DRAW"BM=X2; ,=Y2;":COLOR 1:PRINT#1,""
1580 IF A$ = CHR$(13) THEN GOSUB 3520:RETURN
1590 IF A$ >= CHR$(49) AND A$ < CHR$(57) OR A$ = CHR$(45)
THEN AA$ = AA$ + A$:GOSUB 3360
1600 IF A$ = CHR$(8) THEN DRAW"BM=X2; ,=Y2;":GOSUB 3430
1610 FOR I=1 TO 30:NEXT I

```



```

1620 DRAW"BM=X2; ,=Y2; ":COLOR 12:PRINT#1,CHR$(200)
1630 GOTO 1550
1640 :
1650 REM *** BEREKENING SPELER ***
1660 REM ** AANWIJZING = T$ **
1670 H = 0
1680 D = D + K                               : `GEWENSTE
LOCATIE
1690 IF D > 1 GOTO 1710
1700 D = 1
1710 IF D < 101 GOTO 1750
1720 T$ = "JE ZIT AAN HET BEGIN, BETER OPLETTEN !"
1730 D = D - 100
1740 REM ** COMPUSAURUS KIJKT NAAR SPELER **
1750 B(D) = A(D)
1760 IF A(D) = D GOTO 1800
1770 T$ = "ZO DAT SCHIET OP !!"
1780 IF A(D) > D GOTO 1800
1790 T$ = "HAASTIGE SPOED IS ZELDEN GOED."
1800 D = A(D)                                 : `UITEINDELIJKE
LOKATIE
1810 IF D < 100 GOTO 1840
1820 T$ = "GEFELICITEERD, EINDPUNT BEREIKT!"
1830 H = 2
: `STUURVARIABELE
1840 IF D <> C GOTO 1880
1850 T$ = "** JE HEBT DE COMPUSAURUS TE PAKKEN **"
1860 H = 1
: `STUURVARIABELE
1870 C = 0                                   : `COMPUSAURUS
UIT HET SPEL
1880 RETURN
1890 :
1900 REM *** GEWONNEN OP `1` ***
1910 IF C > 1 GOTO 1950
1920 T$ = "HOERA, IK ZIT OP `1` EN HEB GEWONNEN !"
1930 H = 2
: `STUURVARIABELE
1940 C = 1
1950 RETURN
1960 :
1970 REM *** AANVAL COMPUSAURUS ***
1980 REM ** AANWIJZING = T$ **
1990 REM ** DIRECTE - + INDIRECTE AANVAL **
2000 H = 0
: `STUURVARIABELE
2010 Y = C
2020 FOR I = Y-8 TO Y+8                       : `MOGELIJKE
STAPPEN C-SAURUS
2030 IF (I < 1) OR (I > 100) GOTO 2210
2040 IF (D <> I) OR (B(I) <> 999) GOTO 2120
2050 REM *** COMPUSAURUS GOKT ***
2060 T$ = "EEN TUNNEL, WAT JAMMER NOU !!"
2070 H = 1
: `STUURVARIABELE
2080 G = I - C                               : `AANTAL
STAPPEN C-SAURUS

```

```

2090 C = A(I) : 'PLAATS WAAR
SAURUS UITKOMT
2100 B(I) = A(I) : 'GEHEUGEN
VULLEN
2110 IF A(I) <> I GOTO 2190
2120 IF B(I) <> D GOTO 2210
2130 G = I - C : 'AANTAL
STAPPEN C-SAURUS
2140 D = 0 : 'SPELER UIT
SPEL
2150 H = 2
2160 T$ = "IK HEB JE TE PAKKEN !!"
2170 IF B(I) = I GOTO 2190
2180 T$ = "IK HEB JE TE PAKKEN VIA EEN TUNNEL !!"
2190 GOSUB 1910
2200 I = Y + 8
2210 NEXT I
2220 RETURN
2230 :
2240 REM *** COMPUSAURUS HINDERLAAG ***
2250 REM ** C-SAURUS PROBEERT HINDERLAAG **
2260 H = 0
2270 Y = C
2280 FOR I = Y-8 TO Y+8
2290 IF (I < 1) OR (I > 100) GOTO 2420
2300 IF (B(I) = 999) OR (B(I) = I) GOTO 2420
2310 REM ** LOCATIE IN GEHEUGEN BEKEND **
2320 X = B(I)
2330 Z = ABS(X - D) : 'AFSTAND
SPELER & C-SAURUS
2340 IF Z > 8 GOTO 2420
2350 REM ** COMPUSAURUS IN BEREIK SPELER **
2360 IF (B(X) = 999) OR (B(X) = X) GOTO 2420
2370 REM ** HINDERLAAG LEGGEN **
2380 G = I - C : 'AANTAL
STAPPEN
2390 C = A(I) : 'LOKATIE
C-SAURUS
2400 I = Y + 8
2410 H = 1
2420 NEXT I
2430 RETURN
2440 :
2450 REM *** COMPUSAURUS PROBEERT NAAR 1 TE KOMEN ***
2460 V = 999 : 'EXTREME
VERGELIJKINGSWAARDE
2470 Y = C
2480 FOR I = Y-8 TO Y+8
2490 IF (I < 1) OR (I > 100) GOTO 2590
2500 IF (B(I) = 999) GOTO 2590
2510 IF(B(I) = I) OR (B(I) => C) GOTO 2590
2520 IF ABS(B(I) - D) < 9 GOTO 2590
2530 IF E = I GOTO 2590
2540 REM ** WISSELEN **
2550 IF B(I) > V GOTO 2590
2560 H = 1
2570 V = B(I)

```

```

2580 X = I
2590 NEXT I
2600 IF H. = 0 GOTO 2650
2610 G = X - C                                : `AANTAL
STAPPEN
2620 C = V
2630 GOSUB 1910
2640 E = X                                    : `ZET
ONTHOUDEN
2650 RETURN
2660 :
2670 REM *** COMPUSAURUS GOKT NAAR 1 ***
2680 H = 0
2690 Y = C
: `VERGELIJKINGSWAARDE
2700 FOR I = Y-8 TO Y+8
2710 IF (I < 1) OR (I > 100) GOTO 2840
2720 IF (B(I) <> 999) AND (B(I) <> I) GOTO 2840
2730 IF I > 1 GOTO 2780
2740 G = I - C                                : `AANTAL
STAPPEN
2750 C = 1
2760 GOSUB 1910
2770 I = 8 + Y
2780 IF (ABS(I - D) < 9) OR (H = 2) GOTO 2840
2790 IF E = I GOTO 2840
2800 H = 1
2810 IF I > Y GOTO 2840
2820 Y = I
2830 Q = I - C
2840 NEXT I
2850 IF H = 2 GOTO 2980
2860 IF H = 1 GOTO 2920
2870 REM ** C-SAURUS WEET NIET WAT HIJ DOEN MOET ! **
2880 Q = INT(RND(-TIME) * (-8) - 1)
2890 IF ABS(C + Q - L) > 8 GOTO 2920
2900 Q = Q * (-1)
2910 IF (C + Q < 1) OR (C + Q > 100) GOTO 2880
2920 G = Q                                    : `AANTAL
STAPPEN
2930 C = C + Q
2940 E = C
2950 B(C) = A(C)
2960 C = B(C)
2970 GOSUB 1910
2980 RETURN
2990 :
3000 REM **** STUURMODULE COMPUSAURUS ****
3010 REM *** SPELREGELS ***
3020 GOSUB 90
3030 REM *** INITIALIZEREN ***
3040 GOSUB 490
3050 REM *** UITVOER ***
3060 IF VLAG = 0 THEN GOSUB 930 ELSE GOSUB 1180
3070 REM *** TELLER SPELER ***
3080 GOSUB 810:VLAG = 1
3090 REM *** INVOER SPELER ***

```




```

3100 GOSUB 1510
3110 REM *** BEREKENING SPELER ***
3120 GOSUB 1670
3130 REM *** UITVOER ***
3140 GOSUB 1180
3150 IF H = 2 GOTO 3270
3160 IF C = 0 GOTO 3060
3170 REM *** TELLER C-SAURUS ***
3180 GOSUB 870
3190 REM *** AGRESSIEF AANVALS-PATROON ***
3200 GOSUB 2000
3210 IF H <> 0 GOTO 3270
3220 GOSUB 2260
3230 IF H <> 0 GOTO 3270
3240 GOSUB 2460
3250 IF H <> 0 GOTO 3270
3260 GOSUB 2680
3270 IF H < 2 GOTO 3060
3280 GOSUB 1180
3290 GOTO 3290
3300 REM *** SCHRIJVEN OP SCHERM 2 ***
3310 FOR T = 1 TO LEN(T$)
3320 DRAW"BM=X1; ,=Y1; "
3330 PRINT#1,MID$(T$,T,1):X1 = X1 + 6
3340 NEXT T
3350 RETURN
3360 FOR T = 1 TO LEN(A$)
3370 REM *** INPUT CURSOR PLAATSEN ***
3380 DRAW"BM=X2; ,=Y2; ":COLOR 12:PRINT#1,CHR$(200)
3390 DRAW"BM=X2; ,=Y2; ":COLOR 1:PRINT#1,MID$(A$,T,1):X2 = X2
+ 6
3400 NEXT T
3410 DRAW"BM=X2; ,=Y2; ":PRINT#1,CHR$(192)
3420 RETURN
3430 Z$ = CHR$(200)
3440 Z = LEN(AA$)
3450 DRAW"BM=X2; ,=Y2; "
3460 COLOR 12:PRINT#1,Z$;
3470 X2 = X2 - 6:DRAW"BM=X2; ,=Y2; ":PRINT#1,Z$
3480 AA$ = LEFT$(AA$,Z - 1)
3490 COLOR 1:DRAW"BM=X2; ,=Y2; ":PRINT#1,CHR$(192)
3500 RETURN
3510 REM *** CURSOR WISSEN ***
3520 K = VAL(AA$)
3530 LINE(X2,Y2)-(X2+16,Y2+7),12,BF
3540 AA$ = ""
3550 RETURN
3560 REM *** EIND SPEL ***
3570 FOR I = 1 TO 90
3580 CIRCLE(125,95),I,6
3590 PLAY"S7M100002T255CD"
3600 NEXT I
3610 FOR I = 1 TO 2
3620 PRESET(88+I,92):COLOR 1
3630 PRINT#1,"COMPUSAURUS"
3640 NEXT I
3650 RETURN

```

Voorbeeld



3.5 De zandplaneet Silico

De zandplaneet Silico is de vijftigste planeet in de Schild-sector. Silico maakt een baan om een dubbelsterstelsel en uit waarnemingen is gebleken dat de massa van de planeet enorm is. Deze twee aspecten maken Silico uniek. Waarschijnlijk als gevolg van de baan heeft de planeet een zeer hoge rotatiesnelheid. De hoge rotatiesnelheid gecombineerd met de enorme zwaartekracht zorgen ervoor dat Silico meer weg heeft van een schijf dan van een bol. Hierdoor kan bij de navigatie op de planeet niet van boldriehoeksmeting worden uitgegaan. De planeet is ontzettend droog en ze bestaat voor het grootste gedeelte uit silicium. Daarom wordt zij ook wel de 'Zandplaneet' genoemd. Het stormt er verschrikkelijk en er ontstaan plotseling enorme zandstromen die soms dagen aanhouden. De zandstromen zijn ware rivieren die zich met verschillende snelheden in verschillende richtingen bewegen.

U zult begrijpen dat zo'n wereld vanuit wetenschappelijk oogpunt geweldig interessant is. Het Galactisch Genootschap (the Galactic Council) heeft daarom op de planeet een expeditiebasis ingericht. Deze basis staat op een zodanig punt dat hij niet door de zandstromen kan worden meegevoerd.

Zoals uit de instructie blijkt, is het uw taak een survey-robot, de XAM, naar de expeditiebasis te dirigeren. Vanwege de hoge zwaartekracht is het voor het ruimteschip onmogelijk om in een stationaire omloopbaan te komen. De enig mogelijke omloopbaan is die van precies 60 minuten. Het duurt dus altijd een uur voordat men kan zien hoe de XAM op de instructies heeft gereageerd.

Het beeldscherm laat behalve een kaart van het gebied ook nog een aantal belangrijke gegevens zien. Rechtsboven staan de omloopteller en de kompasroos. In het midden, onder de kompasroos, staan de coördinaten van de expeditiebasis. Links de X-coördinaat, rechts de Y-coördinaat. Rechtsmidden en rechtsonder staan twee kolommen met getallen. Links de getallen op basis van de ingevoerde gegevens zonder zandstromen en rechts de waarden zoals ze door de XAM zijn doorgeseind. Afwijkingen tussen beide waarden ontstaan door de zandstromen.

Suggestie: Men zou het programma 'Ruimtevaartuig' kunnen koppelen aan 'de Zandplaneet Silico'. Wanneer men deze suggestie zou verwezenlijken, krijgt men eerst de landing van XAM op Silico en daarna de tocht naar de expeditiebasis.

Niet bepaald een eenvoudige opgave!

Programma

```
10 REM *** DE ZANDPLANEET SILICO ***
20 DATA 4,1.8, 4,2.6, 3,1.4, 3,3.4, 2,4.8, 1,0.4, 1,5.8
30 SCREEN 0:KEY OFF:WIDTH 40:COLOR 6,9,9:CLS
40 DIM P(9,3), Z(10,2), A$(4)
50 GOTO 2920
60 :
70 REM *** INSTRUCTIES ***
80 L$="":FOR RS=1 TO 39:L$=L$+CHR$(1)+CHR$(87):NEXT RS
90 LOCATE 0,0
```



```

100 PRINT L$
110 PRINT"***** DE ZANDPLANEET SILICO *****"
120 PRINT L$
130 PRINT
140 PRINT"U bevindt zich met Uw ruimteschip in"
150 PRINT"een omloopbaan van 60 minuten om de"
160 PRINT"planeet SILICO, de vijftigste planeet"
170 PRINT"in de schildsector. XAM, een survey-"
180 PRINT"robot, is op dit moment geland. Uit de"
190 PRINT"eerste berichten blijkt dat dit vlakbij"
200 PRINT"de expeditie-basis is."
210 PRINT"Alleen de zandzee met haar zandstromen"
220 PRINT"een bergrug scheiden de Xam nog."
230 PRINT"Vanwege de omlooptijd van 60 min. kan"
240 PRINT"Xam elke uur van nieuwe instructies"
250 PRINT"worden voorzien. Deze instructies be-"
260 PRINT"staan uit een koers in graden en een"
270 PRINT"snelheid, die ligt tussen de 0 en 30"
280 PRINT"knots."
290 PRINT
300 PRINT L$
310 PRINT"DRUK <RETURN-TOETS> IN OM DOOR TE GAAN."
320 PRINT L$
330 A$ = INKEY$
340 IF A$ <> CHR$(13) GOTO 330
350 :
360 CLS
370 PRINT L$
380 PRINT"***** DE ZANDPLANEET SILICO *****"
390 PRINT L$
400 PRINT
410 PRINT"De computer zal aan de hand van de ge-"
420 PRINT"gevens de coördinaten van de berekende"
430 PRINT"lokatie en de werkelijke lokatie geven."
440 PRINT"Ook wordt in de kaart de werkelijke"
450 PRINT"plaats van de Xam aangegeven (*)."
460 PRINT"Afwijkingen tussen beide lokaties ont-"
470 PRINT"staan door de beruchte zandstormen."
480 PRINT
490 PRINT"- De expeditie-basis (#) -."
500 PRINT"- Zandbergen          (";CHR$(206);") -."
510 PRINT
520 PRINT"De bedoeling zal duidelijk zijn:"
530 PRINT"De Xam moet naar de expeditie-basis ge-"
540 PRINT"dirigeerd worden. Een moeilijke zware"
550 PRINT"tocht."
560 PRINT
570 PRINT L$
580 PRINT"HEEFT U DIT GELEZEN DRUK DAN <RETURN>IN"
590 PRINT L$
600 A$ = INKEY$
610 IF A$ <> CHR$(13) GOTO 600
620 RETURN
630 :
640 REM *** INITIALIZEREN ***
650 REM ** ZANDSTROMEN PER BREEDTE **
660 OPEN"GRP:" AS#1:SCREEN 2

```

```

670 FOR I = 1 TO 10
680 Z(I,1) = INT(RND(-TIME) * 360)           : 'RICHTING
ZANDSTR.
690 Z(I,2) = INT(RND(-TIME) * 15 + 1)/100    : 'SNELHEID
ZANDSTR.
700 NEXT I
710 REM ** PLAATSEN VAN BERGRUG
720 RESTORE 20
730 FOR I = 1 TO 7
740 READ P(I,1), P(I,2)                     : 'INLEZEN
LOKATIES
750 P(I,3) = 1
: 'AANDUIDING BERG
760 NEXT I
770 REM ** PLAATSEN VAN BASIS **
780 A = INT(RND(-TIME) * 2 + 6)
: 'Y-COORDINAAT
790 B = INT(RND(-TIME) * 20 + 45)/10
: 'X-COORDINAAT
800 P(8,3) = 2                               : 'BASIS
PLAATSEN
810 P(8,1) = A
820 P(8,2) = B
830 REM ** PLAATSEN SURVEY-ROBOT XAM **
840 Y = 0                                     : 'Y-COORD
XAM
850 X = INT(RND(-TIME) * 15 + 15)/10        : 'X-COORD
XAM
860 Y1 = Y                                   : 'BEREK
Y-COORD
870 X1 = X                                   : 'BEREK
X-COORD
880 P(9,1) = Y
890 P(9,2) = X
900 P(9,3) = 3
910 REM ** VARIABELEN EEN STARTWAARDE GEVEN **
920 REM * STRING$ *
930 SPRITE$(0) = CHR$(240)+CHR$(144)+CHR$(144)
+CHR$(144)+CHR$(255)+CHR$(249)+CHR$(249)+CHR$(255)
940 SPRITE$(1) = CHR$(24) + CHR$(60) + CHR$(126) +
CHR$(255) + CHR$(165) + CHR$(66) + CHR$(0)
950 S$ = CHR$(132):Z$ = CHR$(252)
960 SPRITE$(2) = STRING$(5,CHR$(0)) + Z$ + S$ + Z$
970 A$ = "0====1====2====3====4====5====6====7"
980 D$ = ""
990 C$(1) = CHR$(206)                        : 'SYMBOOL
BERG
1000 C$(2) = SPRITE$(0)                     : 'SYMBOOL
BASIS
1010 C$(3) = SPRITE$(1)                     : 'SYMBOOL
XAM
1020 C$(4) = " "                             : 'SYMBOOL
XAM
1030 REM * NUMERIEKE VARIABELEN *
1040 W = 1                                   : 'OMLOOP
1050 C = 3.1415926536#/180                  : 'GRADEN
-> RADIALEN

```

```

1060 D = 180/3.1415926536#
: `RADIALEN -> GRADEN
1070 F = 0
= 0
1080 H = 0
1090 RETURN
1100 :
1110 REM *** UITVOER ***
1120 SCREEN 2:COLOR 6,10,6:CLS
1130 LINE(2,2)-(253,12),15,BF
1140 YY = 4:C1L = 15
1150 FOR I = 0 TO 1
1160 XX = 8 + I
1170 T$ = "< DE ZANDPLANEET SILICO > OMLOOP>"
1180 GOSUB 3090
1190 NEXT I
1200 LINE(12,14)-(230,110),1,B
1210 LINE(20,22)-(222,102),1,B
1220 LINE(2,166)-(253,189),5,BF
1230 T$ = A$:C1L = 10
1240 FOR I = 1 TO 2
1250 XX = 14
1260 IF I = 1 THEN YY = 15 ELSE YY = 103
1270 GOSUB 3090
1280 NEXT I:YY = 14
1290 FOR I = 9 TO 0 STEP -1
1300 YY = YY + 8
1310 PSET(4,YY),C1L:PRINT#1,USING"#";I
1320 PSET(14,YY),C1L:PRINT#1,"="
1330 PSET(224,YY),C1L:PRINT#1,"="
1340 NEXT I
1350 CIRCLE(218,130),27,1,,.6
1360 PAINT(227,126),1,1:COLOR 15
1370 DRAW"BM210,118":PRINT#1,"360"
1380 DRAW"BM194,127":PRINT#1,"270";CHR$(188);"90"
1390 DRAW"BM210,136":PRINT#1,"180"
1400 GOSUB 1840
1410 COLOR 6
1420 LINE(0,111)-(180,164),10,BF
1430 LINE(225,2)-(253,12),14,BF
1440 DRAW"BM8,112"
1450 PRINT#1,USING"##.##";B;
1460 PRINT#1,"< X-B-Y >";
1470 PRINT#1,USING"##.##";A
1480 DRAW"BM0,120"
1490 PRINT#1,"<BEREIK>-----<WERK.>"
1500 DRAW"BM0,128"
1510 PRINT#1,USING"###";K1
1520 DRAW"BM48,128":PRINT#1,"< KOERS >";
1530 PRINT#1,USING"###";K
1540 DRAW"BM0,136":PRINT#1,USING"###";F1
1550 DRAW"BM48,136":PRINT#1,"< SNE LH >";
1560 PRINT#1,USING"###";F
1570 DRAW"BM0,148":PRINT#1,USING"###.##";X1
1580 DRAW"BM48,148":PRINT#1,"< X-COR >";
1590 PRINT#1,USING"###.##";X
1600 DRAW"BM0,156":PRINT#1,USING"###.##";Y1

```



```

1610 DRAW"BM48,156":PRINT#1,"< Y-COR >";
1620 PRINT#1,USING"###.#";Y
1630 PSET(225,4),15:PRINT#1,W
1640 LINE(21,23)-(221,101),10,BF
1650 M = 9
1660 L = P(M,3)
1670 X3 = INT(P(M,2) * 27 + 20 + .5)
1680 Y3 = 94 - P(M,1) * 8
1690 ON L GOTO 1700,1720,1740
1700 PSET(X3,Y3),C1L:PRINT#1,C$(1)
1710 GOTO 1760
1720 PUT SPRITE 0,(X3,Y3),1,0
1730 GOTO 1760
1740 Z = M
1750 PUT SPRITE 1,(X3,Y3),12,1
1760 M = M - 1:IF M <> 0 GOTO 1660
1770 IF D$ = "" GOTO 1810
1780 IF D$ <> "" THEN LINE(2,166)-(253,189),1,BF
1790 XX = 32:YY = 175:T$ = D$
1800 GOSUB 3090
1810 RETURN
1820 :
1830 REM *** SORTEREN MAT P ***
1840 FOR Q = 9 TO 1 STEP (-1)
1850 FOR R = 1 TO Q
1860 P = P(Q,1) * 1000 - P(Q,2)
1870 L = P(R,1) * 1000 - P(R,2)
1880 IF P > L THEN 1920
1890 FOR J = 1 TO 3
1900 SWAP P(R,J),P(Q,J)
1910 NEXT J
1920 NEXT R
1930 NEXT Q
1940 RETURN
1950 :
1960 REM *** GEGIST BESTEK ***
1970 P = K1 * C : 'KOERS IN
RADIALEN
1980 Y1 = INT((S * COS(P) + Y) * 10 + .5)/10 : 'NWE BR
1990 X1 = INT((S * SIN(P) + X) * 10 + .5)/10 : 'NWE L
2000 W = W + 1 : 'OMLOOP +
1
2010 RETURN
2020 :
2030 REM *** FEITELIJKE KOERS ***
2040 S = S / 10
2050 FOR I = 1 TO 10
2060 L = INT(Y) + 1
2070 K = Z(L,1)
2080 S1 = Z(L,2)
2090 IF S = 0 GOTO 2250
2100 REM ** KOERSCORRECTIE I.V.M. STROOM **
2110 R = Z(L,1) - K1
: 'STROOMRICHTING
2120 S1 = S^2 + Z(L,2)^2 - 2 * S * Z(L,2) * COS((180 - R) *
C)
2130 IF ABS(S1) > 1E-06 GOTO 2160

```

```

2140 S1 = 0
2150 K = -1                                : 'RELATIEVE
SNELHEID
2160 S1 = SQR(S1)
2170 IF S1 = 0 GOTO 2310
2180 P = SIN((180 - R) * C)/S1 * Z(L,2)
2190 P = ATN(P/SQR(-P * P + 1))           : 'C-KOERS
IN RAD
2200 K = INT(P * D + K1 + .5)             : 'FEITELIJKE
KOERS
2210 IF K <= 360 GOTO 2230
2220 K = K - 360
2230 IF K >= 0 GOTO 2250
2240 K = K + 360
2250 P = K * C
2260 F = INT(S1 * 30 * 10 + .5)
2270 REM ** GEGIST BESTEK OP BASIS FEITELIJKE KOERS **
2280 Y = S1 * COS(P) + Y                   : 'NIEUW
Y-COORD
2290 X = S1 * SIN(P) + X                   : 'NIEUW
X-COORD
2300 REM ** ERGENS IN DE BUURT **
2310 FOR J = 1 TO 9
2320 Y2 = P(J,1) - Y                       : 'AFSTAND
Y-RICHT.
2330 X2 = P(J,2) - X                       : 'AFSTAND
X-RICHT.
2340 IF ABS(Y2) > .15 GOTO 2500
2350 IF ABS(X2) > .15 GOTO 2500
2360 REM * WAAR DICHTBIJ ?*
2370 IF P(J,3) <> 1 GOTO 2420
2380 REM * TEGEN BERG *
2390 H = 2
: 'STUURVARIABELE
2400 L = J                                  : 'COORD
OVERNEMEN
2410 D$ = "> XAM UITGESCHAKELD <"
2420 IF P(J,3) <> 2 GOTO 2470
2430 REM * AANGEKOMEN BIJ BASIS *
2440 H = 3
2450 L = J
2460 D$ = "> * BASIS BEREIKT * <"
2470 IF H < 2 GOTO 2500
2480 J = 9                                  : 'UIT J-LUS
2490 I = 10                                 : 'UIT I-LUS
2500 NEXT J
2510 NEXT I
2520 Y = INT(Y * 10 + .5)/10                : 'AFGEROND
COORD
2530 X = INT(X * 10 + .5)/10                : 'AFGEROND
COORD
2540 RETURN
2550 :
2560 REM *** NIEUWE LOCATIE XAM ***
2570 P(Z,1) = INT(Y)
2580 P(Z,2) = X
2590 IF H < 2 GOTO 2640

```

```

2600 P = .1
2610 IF X2 >= 0 GOTO 2630
2620 P = -.1
2630 P(Z,2) = P(L,2) + P
2640 IF (P(Z,1) < 0) OR (P(Z,1) > 9) GOTO 2660
2650 IF (P(Z,2) >= 0) AND (P(Z,2) <= 7) GOTO 2710
2660 P(Z,1) = 0
2670 P(Z,2) = 0
2680 P(Z,3) = 4
2690 D$ = "> XAM BUITEN BEREIK <"
2700 H = 2
2710 RETURN
2720 :
2730 REM *** INVOER ***
2740 COLOR 14
2750 FOR I = 0 TO 1
2760 XX = 8 + I:YY = 170:T$ = "KOERS IN GRADEN 0 - 360 : "
2770 GOSUB 3090
2780 XX = 8 + I:YY = 180:T$ = "SNELHEID IN KNOTS 0 - 30 : "
2790 GOSUB 3090
2800 NEXT I
2810 XC = 170:YC = 170:XB = 170:VC = 0
2820 PUT SPRITE 2, (XC, YC), 1, 2
: 'PLAATSEN VAN CURSOR
2830 AC$ = INPUT$(1)
2840 IF AC$ = CHR$(13) THEN GOSUB 3330
2850 IF AC$ = CHR$(8) THEN GOSUB 3220
2860 IF AC$ <> "" THEN GOSUB 3160
2870 AA$ = AA$ + AC$
2880 GOTO 2820
2890 :
2900 REM **** STUURMODULE ****
2910 REM *** INSTRUCTIES ***
2920 GOSUB 80
2930 REM *** INITIALIZEREN ***
2940 GOSUB 660
2950 REM *** UITVOER ***
2960 IF VLAG = 0 THEN GOSUB 1110 ELSE GOSUB 1400
2970 REM *** INVOER ***
2980 GOSUB 2740:VLAG = 1
2990 REM *** GEGIST BESTEK ***
3000 GOSUB 1970
3010 REM *** FEITELIJKE KOERS ***
3020 GOSUB 2040
3030 REM *** NIEUWE LOCATIE ***
3040 GOSUB 2570
3050 IF H < 2 GOTO 2960
3060 REM *** UITVOER ***
3070 GOSUB 1120
3080 GOTO 3080
3090 REM *** SCHRIJVEN OP SCHERM 2 ***
3100 FOR ST = 1 TO LEN(T$)
3110 DRAW"BM=XX; ,=YY;"
3120 PRINT#1, MID$(T$, ST, 1)
3130 XX = XX + 6
3140 NEXT ST
3150 RETURN

```




```

3160 REM *** VERPLAATSEN VAN CURSOR ***
3170 PSET(XC,YC),5
3180 PRINT#1,AC$:XC = XC + 6
3190 PUT SPRITE 2,(XC,YC),1,2
3200 RETURN
3210 REM *** BACKSPACE ***
3220 XC = XC - 6
3230 IF XC <= XB THEN XC = XB
3240 PSET(XC,YC),5
3250 PUT SPRITE 2,(XC,YC),1,2
3260 LINE(XC,YC)-(XC+6,YC+6),5,BF
3270 IF AA$ = "" GOTO 3300
3280 AA$ = LEFT$(AA$,LEN(AA$)-1)
3290 AC$ = ""
3300 RETURN
3310 :
3320 REM *** EIND INVOER ***
3330 IF VC = 0 THEN K1 = VAL(AA$)
3340 IF VC = 1 THEN F1 = VAL(AA$):S = F1/30
:RELATIEVE SNELHEID
3350 AA$ = "":AC$ = ""
3360 PUT SPRITE 2,(XC,YC),0,2
3370 XC = 170:YC = 180
3380 IF VC = 1 THEN LINE(164,166)-(200,188),5,BF:RETURN
2990
3390 VC = 1:VLAG = 1
3400 RETURN

```

Voorbeeld



3.6 De slag bij de Shrixheuvels

Het gebied van de Shrixheuvels vormt de landlengte tussen het Noordercontinent en het Zuidercontinent. De landlengte was vroeger een stuk niemandsland, maar sinds de Schrammers het in bezit hebben genomen, hebben de handelskaravanen ontzettend veel last van de Schrammers.

Dat het gebied belangrijk is, laat zich gemakkelijk raden.

'Schrammers' is niet hun werkelijke naam, maar ze worden zo genoemd vanwege hun uiterlijk. Ze lijken veel op bytosaurussen.

De taak van de strateeg van de Grauwduivels is het gebied van de Shrixheuvels te zuiveren van Schrammers. Hiertoe zullen eerst de strategische punten van de Schrammers dienen te worden gezocht en daarna aangevallen.

De grauwduivels lijken wel wat op de Schrammers; hun ontwikkeling ligt echter iets hoger. Dat laatste klopt ook wel, omdat zij afstammen van de Compusaurussen. Het bijzondere van deze Grauwduivels is dat zij kunnen worden geprogrammeerd en wel op twee manieren: als Loerduivels of als Hellevegen. Eenmaal geprogrammeerd zijn veranderingen niet meer mogelijk.

De Loerduivels kunnen alleen als verkenners worden gebruikt, de Hellevegen alleen als aanvallers op de Schrammers.

Om de strategische punten van de Schrammers te ontdekken, moeten Loerduivels worden ingezet. Dit kan alleen vanuit het zuiden (Y -coördinaat=0). Wanneer de Loerduivels terugkomen, geven ze de strateeg een afstand op vanaf de ingezette locatie tot aan het gevonden strategisch punt.



Mochten er nog meer strategische punten zijn dan blijkt dit uit de opmerking 'STRATEGISCH PUNT+1'. Wil men met voldoende zekerheid een strategisch punt bepalen dan moet men twee verkenningen van verschillende locaties doen. Denkt de strateeg een punt te hebben gevonden dan kan hij de X- en de Y-coördinaat van het punt opgeven. Verschijnt er in het kaartje een 'O' dan klopt het en zou men een aanval kunnen wagen. Slaagt de aanval dan verandert de 'O' in '*'.

Wanneer de aanval niet slaagt, trekken de niet-vermiste Hellevegen zich direct terug. De Schrammers hebben natuurlijk ook verliezen geleden, wat dus de verdediging van het strategisch punt aantast. Slaagt een aanval op een strategisch punt dan vluchten de niet-vermiste Schrammers naar de overige nog niet bezette punten.

De strateeg volgt de strijd vanaf zijn commandopost met een blanco kaart waarop de coördinaten zijn aangetekend. Verder ligt er een liniaal en een passer. Het beeldscherm en de computer worden gebruikt om de Grauwduivels te programmeren. Tegelijkertijd levert de computer een aantal belangrijke gegevens die de terugkerende troepen hem melden. Rechtsmidden wordt aangegeven hoeveel Grauwduivels en welke soort voor een bepaalde actie zijn gezet. Wanneer Loerduivels zijn ingezet, verschijnt eronder de afstand tot een strategisch punt. Rechtsonder wordt bijgehouden hoeveel Schrammers en Grauwduivels er in het totaal vermist zijn en hoeveel tijdens de uitgevoerde actie.

Suggestie 1: Wanneer een strategisch punt eenmaal is bezet door de Hellevegen heeft de strateeg hier verder geen omkijken meer naar. Hij verliest een aantal Hellevegen in de strijd, maar verder niets. Om het spel echter te doen lijken, moet de strateeg ook het punt tegen de Schrammers verdedigen. Dit laatste kan alleen wanneer hij er een aantal Hellevegen ter verdediging levert. Is zo'n punt slecht verdedigd dan kan het weer door de Schrammers worden ingenomen. Het spel wordt er wel interessanter door.

Suggestie 2: Om het spel nog aantrekkelijker te maken, kan men de Loerduivels niet alleen de afstand tot een strategisch punt laten geven, maar ze ook nog een indicatie laten geven van de sterkte van dat punt. De nauwkeurigheid van de opgegeven 'sterkte' kan men af laten hangen van het aantal ingezette Loerduivels.

Programma

```
10 REM *** DE SLAG BIJ DE SHRIX-HEUVELS ***
20 'RANDOM
30 CLEAR 1000
40 DIM A(5,4), B$(5), C$(3)
50 GR$=CHR$(1)
60 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
70 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
80 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
90 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
100 GOTO 3230
110 :
```



```

120 REM *** INSTRUCTIES ***
130 KEY OFF:SCREEN 0:WIDTH 40:COLOR 3,1,1:CLS           : 'SCHERM
WISSEN
140 GOSUB 3090
150 LOCATE 0,5
160 PRINT" U bent aanvoerder van een horde GRAUW-"
170 PRINT" DUIVELS."
180 PRINT" Er zijn twee soorten Grauwduivels : "
190 PRINT" - LOERDUIVELS, die alleen als ver-"
200 PRINT" kenner gebruikt kunnen worden."
210 PRINT" - HELLEVEGEN, die alleen kunnen vech-"
220 PRINT" ten."
230 PRINT" U probeert het gebied van de Shrix-"
240 PRINT" heuvels in bezit te krijgen door de"
250 PRINT" strategische punten te ontdekken en"
260 PRINT" te bezetten."
270 GOSUB 3170
280 GOSUB 3620
290 A$ = INKEY$
300 IF A$ <> CHR$(13) GOTO 290
310 COLOR 7,1,1:CLS
320 GOSUB 3100
330 LOCATE 0,5
340 PRINT" De plaats van de strategische punten"
350 PRINT" moet U zien te bepalen met Uw LOER-"
360 PRINT" DUIVELS. De aanval op zo'n strategisch"
370 PRINT" punt moet U met de HELLEVEGEN doen."
380 PRINT
390 PRINT" Strategische punten worden verdedigd"
400 PRINT" door de SCHRAMMERS, de tegenhangers"
410 PRINT" van de hellevegen."
420 PRINT
430 PRINT" De rest van de informatie vindt U in"
440 PRINT" het boek."
450 GOSUB 3170
460 A$ = INKEY$
470 IF A$ <> CHR$(13) GOTO 460
480 RETURN
490 :
500 REM *** INITIALIZEREN ***
510 REM ** STRATEGISCHE PUNTEN **
520 S = INT(RND(-TIME) * 4 + 2)           : 'AANTAL
STRATEGISCHE PUNTEN
530 T = 0                                 : 'AANTAL
SCHRAMMERS
540 FOR Y = 1 TO S
550 I = INT(RND(-TIME) * 6 + 2)           : 'Y-COORD
STRAT. PUNT
560 J = INT(RND(-TIME) * 8)             : 'X-COORD
STRAT. PUNT
570 IF Y = 1 GOTO 670
580 H = 0
: 'STUURVARIABELE
590 REM * CONTROLE OP VOORKOMEN *
600 FOR X = 1 TO Y-1
610 IF (I <> A(X,1)) OR (J <> A(X,2)) GOTO 640
620 H = 1                                 : 'FOUT !

```

```

630 X = Y
640 NEXT X
650 IF H = 1 GOTO 550
660 REM * WAARDEN OPNEMEN *
670 A(Y,1) = I
680 A(Y,2) = J
690 A(Y,3) = INT(RND(-TIME) * 8 + 2) * 10      : 'AANTAL
SCHRAMMERS
700 A(Y,4) = 0                                  : 'NIET
ONTDEKT
710 T = T + A(Y,3)                              : 'TOTAAL
SCHRAMMERS
720 NEXT Y
730 REM * SORTEREN *
740 FOR Q = S TO 2 STEP (-1)
750 FOR R = 1 TO Q
760 P = A(Q,1) * 100 - A(Q,2)
770 L = A(R,1) * 100 - A(R,2)
780 IF P > L GOTO 820
790 FOR J = 1 TO 2
800 SWAP A(Q,J),A(R,J)
810 NEXT J
820 NEXT R
830 NEXT Q
840 REM ** STRINGS EN VARIABELEN **
850 R9$=R1$+R1$
860 A$ = R2$+"0"+R9$+"1"+R9$+"2"+R9$+"3"+R9$+"4"+R9$+"5"
+R9$+"6"+R9$+"7"+R3$
870 A1$ = R4$+"0"+R9$+"1"+R9$+"2"+R9$+"3"+R9$+"4"+R9$+"5"
+R9$+"6"+R9$+"7"+R5$
880 B$(1) = "Loerduivels"
890 B$(2) = "Hellevegen"
900 B$(3) = ""
910 B$(4) = CHR$(188)
920 B$(5) = CHR$(249)
930 C$(1) = "VERKENNING "
940 C$(2) = "AANVAL "
950 C$(3) = "STATUS QUO "
960 L = 3
: 'STARTWAARDEN
970 REM * GRAUWDUIVELS *
980 G = INT(RND(-TIME) * T/2 + T/2) * 8
: 'START GRAUWD.
990 G1 = 0
: 'TOT. VERMIST
1000 G2 = 0                                     : 'IN
SLAG VERMIST
1010 P(1) = 0
: 'LOERDUIVELS
1020 P(2) = 0
: 'HELLEVEGEN
1030 Q = 0
: 'INGEZET
1040 REM * SCHRAMMERS *
1050 T1 = 0
: 'TOT. VERMIST
1060 T2 = 0                                     : 'IN

```

```

SLAG VERMIST
1070 S1 = 0
: `GEVONDEN STRAT P.
1080 N = T/S
: `GEM. AANTAL SCHR.
1090 RETURN
1100 :
1110 REM *** VOORBEREIDING NIEUWE AKTIE ***
1120 Q = 0
: `INGEZET = 0
1130 T2 = 0
: `SCHRAMMERS VERMIST = 0
1140 G2 = 0
: `GRAUWD. VERMIST = 0
1150 D$ = ""
1160 REM ** WAT DOEN ? **
1170 LOCATE 2,21
1180 PRINT"Gaat U aanvallen of verkennen ? ";
1190 BEEP
1200 LINE INPUT Q$
1210 Q$ = LEFT$(Q$,1)
: `ALLEEN LINKER KARAKTER
1220 L = 1
: `STUURVARIABELE
1230 IF Q$ = "V" OR Q$ = "v" GOTO 1270
1240 L = 2
1250 IF Q$ = "A" OR Q$ = "a" GOTO 1270
1260 L = 3
1270 RETURN
1280 :
1290 REM *** AANTAL IN TE ZETTEN GRAUWDUIVELS ***
1300 Q = 0
1310 GOSUB 3470
1320 LOCATE 2,21
1330 PRINT"Hoeveel ";B$(L);" wilt U inzetten"
1340 LOCATE 2,22:PRINT"voor de ";C$(L);
1350 BEEP
1360 INPUT Z$
1370 IF LEFT$(Z$,1) > CHR$(57) GOTO 1310
1380 Q = VAL(Z$)
1390 IF Q <= 0 GOTO 1310
1400 H = 0
: `STUURVARIABELE
1410 REM ** CONTROLE **
1420 R = Q - P(L)
: `HOEVEEL AL GEBRUIKT
1430 IF R <= 0 GOTO 1620
1440 H = 1
1450 REM ** UIT GRAUWDUIVELS **
1460 IF R < G GOTO 1490
1470 REM ** TE WEINIG **
1480 R = G
1490 Q = P(L) + R
1500 GOSUB 3470
1510 LOCATE 1,21
1520 PRINT"U heeft";Q;B$(L);" ingezet accoord"
1530 LOCATE 10,22:PRINT"< JA / NEE > ? ";

```



```

1540 BEEP
1550 LINE INPUT Q$
1560 Q$ = LEFT$(Q$,1)
1570 IF Q$ = "N" OR Q$ = "n" GOTO 1300
1580 IF H = 0 GOTO 1620
1590 REM ** AANPASSING **
1600 G = G - Q
1610 P(L) = P(L) + Q
1620 RETURN
1630 :
1640 REM *** WELKE LOCATIE ***
1650 GOSUB 1930
1660 C$ = "Vanaf "
1670 IF L = 1 GOTO 1690
1680 C$ = "Op "
1690 LOCATE 2,21
1700 PRINT C$;"welke lokatie wilt U ";C$(L)
1710 LOCATE 2,22
1720 PRINT"doen <X-coord,Y-coord>";
1730 BEEP
1740 INPUT X,Y
1750 IF (X < 0) OR (X > 7) GOTO 1650
1760 IF (Y < 0) OR (Y > 7) GOTO 1650
1770 IF (L = 1) AND (Y <> 0) GOTO 1650
1780 RETURN
1790 :
1800 REM *** UITVOER ***
1810 CLS:COLOR 9,1,1:VR = 1
: `SCHERM WISSEN
1820 GOSUB 3100
: `TITEL
1830 :
1840 N = 4
1850 LOCATE 1,4:PRINT A$
1860 FOR I = 7 TO 0 STEP -1
1870 N = N + 1
1880 LOCATE 1,N:PRINT "V";SPACE$(22);R6$
1890 LOCATE 0,N:PRINT USING"#";I
1900 NEXT I
1910 LOCATE 1,13:PRINT A1$
1920 LOCATE 25,3:PRINT"<BESCHIKBAAR>"
1930 LOCATE 2,3:PRINT SPACE$(11)
1940 LOCATE 2,3:PRINT C$(L)
1950 LOCATE 26,5:PRINT"Grauwd. : ";USING"####";G
1960 LOCATE 26,6:PRINT"Loerd. : ";USING"####";P(1)
1970 LOCATE 26,7:PRINT"Hellev. : ";USING"####";P(2)
1980 LOCATE 25,9:PRINT"<INGEZET>"
1990 LOCATE 26,11:PRINT B$(L)
2000 LOCATE 29,12:PRINT Q
2010 LOCATE 1,14:PRINT"<VERMIST>          TOTAAL  ACTIE"
2020 LOCATE 2,15:PRINT"Grauwduivels"
2030 LOCATE 19,15:PRINT USING"####";G1
2040 LOCATE 28,15:PRINT USING"####";G2
2050 LOCATE 2,16:PRINT"Schrammers"
2060 LOCATE 19,16:PRINT USING"####";T1
2070 LOCATE 28,16:PRINT USING"####";T2
2080 GOSUB 3470
2090 LOCATE 2,18:PRINT D$

```

```

2100 LOCATE 2,19:PRINT"Afstand";TAB(21);V
2110 TE$=CHR$(249)+CHR$(249):IF LEFT$(D$,2) = TE$ THEN
GOSUB 3680
2120 RETURN
2130 REM *** BEREKENING AFSTAND ***
2140 D$ = "Niets te vinden"
2150 IF S1 => S GOTO 2280
2160 D$ = "Strategisch punt"
2170 IF S1 + 1 = S GOTO 2190
2180 D$ = "Strategisch punt + 1"
2190 FOR I = 1 TO S
2200 IF A(I,4) <> 0 GOTO 2230
2210 J = I
2220 I = S
: 'UIT
LUS
2230 NEXT I
2240 V = (A(J,2)-X)^2 + (A(J,1)-Y)^2
: 'BEPALING AFSTAND
2250 V = INT(SQR(V) * 10)/10
: 'AFRONDEN
2260 REM ** VOLDOENDE LOERDUIVELS INGEZET **
2270 P = (A(J,3)*(-.2)+N)/2
: 'BESPIEDINGSGRAAD
2280 R = INT(RND(-TIME) * Q)
2290 IF Q > P GOTO 2330
2300 R = Q
2310 V = 0
2320 D$ = "Niemand terug"
2330 GOSUB 2380
2340 GOSUB 2440
2350 RETURN
2360 :
2370 REM *** BIJWERKEN LOERDUIVELS VERMIST ***
2380 P(L) = P(L) - R
2390 G1 = G1 + R
2400 G2 = G2 + R
2410 RETURN
2420 :
2430 REM *** LOKATIE BEKEND ***
2440 IF S1 >= S GOTO 2640
2450 H = 3
2460 GOSUB 1930
2470 H = 0
2480 D$ = ""
2490 LOCATE 2,21
2500 PRINT"Lokatie bekend <Ja / Nee> ? ";
2510 BEEP
2520 LINE INPUT Q$
2530 IF LEFT$(Q$,1) <> "J" AND LEFT$(Q$,1) <> "j" GOTO
2650
2540 LOCATE 2,22
2550 PRINT"Geef X-coord, Y-coord ";
2560 BEEP
2570 INPUT X,Y
2580 R = 0
2590 GOSUB 2680
2600 IF R = 0 GOTO 2650

```

```

2610 IF A(R,4) <> 0 GOTO 2650
2620 A(R,4) = 1
2630 D$ = "Gevonden"
2640 S1 = S1 + 1
2650 RETURN
2660 :
2670 REM ** CONTROLE **
2680 FOR I = 1 TO S
2690 IF (X <> A(I,2)) OR (Y <> A(I,1)) GOTO 2720
2700 R = I
2710 I = S
2720 NEXT I
2730 RETURN
2740 :
2750 REM *** BEREKENING AANVAL ***
2760 R = 0
2770 D$ = "Hinderlaag"
2780 GOSUB 2680
2790 IF R = 0 GOTO 3050
2800 D$ = "Afgeslagen"
2810 D = INT(RND(-TIME) * Q/2 + Q/2)
2820 IF D <= A(R,3) GOTO 2870
2830 REM ** STRATEGISCH PUNT IN BEZIT **
2840 D$ = "PUNT BEZET"
2850 A(R,4) = 2
2860 REM ** VERMISTE SCHRAMERS **
2870 T2 = INT(RND(-TIME) * A(R,3))
2880 T1 = T1 + T2
2890 A(R,3) = A(R,3) - T2
2900 IF A(R,4) <> 2 GOTO 3050
2910 REM ** STRATEGISCHE PUNTEN IN HANDEN SCHRAMERS **
2920 D = 0
2930 FOR I = 1 TO S
2940 IF A(I,4) = 2 GOTO 2960
2950 D = D + 1
2960 NEXT I
2970 H = 9
2980 IF D = 0 GOTO 3050
2990 H = 0
3000 REM ** TERUGTREKKEN **
3010 FOR I = 1 TO S
3020 IF A(I,4) = 2 GOTO 3040
3030 A(I,3) = A(I,3) + INT(A(R,3)/D)
3040 NEXT I
3050 R = INT(RND(-TIME) * Q)
3060 GOSUB 2380
3070 RETURN
3080 :
3090 REM *** HOOFDREGEL INSTRUCTIES ***
3100 L1$ = R2$+LEFT$(S1$,74)+R3$
3110 L2$ = R4$+LEFT$(S1$,74)+R5$
3120 LOCATE 0,0:PRINT L1$
3130 PRINT R6$;" ** DE SLAG BIJ DE SHRIX--HEUVELS ** ";R6$
3140 PRINT L2$
3150 RETURN
3160 REM ** VRAAG OM EEN NIEUW SCHERM **

```

: 'WEER


```

3170 LOCATE 0,19:PRINT L1$
3180 PRINT R6$;" GELEZEN,DRUK DAN <RETURN-TOETS> IN. ";R6$
3190 PRINT L2$
3200 RETURN
3210 REM **** STUURMODULE ****
3220 REM *** INSTRUCTIE ***
3230 GOSUB 130
3240 REM *** INITIALIZEREN ***
3250 GOSUB 520
3260 REM *** UITVOER ***
3270 IF VR = 1 THEN GOSUB 1820 ELSE GOSUB 1810
3280 REM *** INVOER ***
3290 GOSUB 1120
3300 IF L = 3 GOTO 3270
3310 REM *** AANTAL IN TE ZETTEN GRAUWDUIVELS ***
3320 GOSUB 1300
3330 REM *** WELKE LOKATIE ***
3340 GOSUB 1650
3350 REM *** VERKENNING = 1 AANVAL = 2 ***
3360 ON L GOSUB 2140, 2760
3370 L = 3
3380 REM *** RESULTAAT ***
3390 R = P(2) + G
3400 IF (H <> 9) AND (R > 20) GOTO 3270
3410 D$ = "** VERLOREN **":KL = 13
3420 IF H <> 9 GOTO 3450
3430 D$ = "** GEWONNEN **":KL = 3
3440 REM *** UITVOER ***
3450 GOSUB 1830
3460 GOTO 3460
3470 FOR I = 8 TO 1 STEP -1
3480 FOR M = S TO 1 STEP -1
3490 IF A(M,1) <> I - 1 GOTO 3550
3500 IF A(M,4) = 0 GOTO 3550
3510 P = A(M,2) * 21/7 + 2
3520 R = A(M,4) + 3
3530 R1 = 12 - A(M,1)
3540 LOCATE P,R1:PRINT B$(R)
3550 NEXT M
3560 NEXT I
3570 FOR I = 18 TO 22
3580 LOCATE 0,I:PRINT SPACE$(39)
3590 NEXT I
3600 RETURN
3610 REM *** MUZIEK ***
3620 PLAY"V12T200","V10T200","V9T200"
3630 PLAY"O4GO5EC2O4EAG2B8R8BABO5CO4AG2","O3CGO2GO3GO3CGO2GO
3GDGO2GO3GCGO2GO3G","RO3ERERERERFRFRERE"
3640 PLAY"GO5EC2O4EAG2B8R8BABO5C1","O3CGO2GO3GCGO2GO3GDGO2GO
3GC1","RO3ERERERERFRFE1"
3650 PLAY"O5CO4BG+EAO5CE2RF2DE.R8E2","O3EO4DRDO3FO4EREO3DARA
EO4CRC","RO3G+RG+RO4CRCRO3FRFRARA"
3660 PLAY"O5D.C8O4BO5DCO4A.R8AB2O5E2O4A8R8A8R8AR","O3EO4DRDO
3AO4EREO3DAEO4DC8R8C8R8CR","RO3G+RG+RO4CRCRO3FRG+A8R8A8R8AR"
3670 RETURN
3680 OPEN"GRP:" AS#1
3690 SCREEN 2


```

```

3700 FOR I = 1 TO 120
3710 LINE(I-I, I-I)-(I+I, I+I), KL, B
3720 NEXT I
3730 LINE(68, 90)-(188, 104), KL, BF
3740 DRAW "BM72, 94"
3750 COLOR 1: PRINT #1, D$
3760 GOSUB 3620
3770 RETURN

```

Voorbeeld



4 Simulatiespellen

4.1 De kolenmijn

U staat bij de ingang van een oude kolenmijn waarin een schat verborgen ligt. U probeert deze schat te zoeken en naar de buitenwereld te brengen. Tot uw uitrusting behoort een lamp. Met het commando 'AAN' gaat de lamp aan; met het commando 'UIT' dooft de lamp. Tijdens de tocht door de mijn zullen aanwijzingen worden gegeven wat er te zien of te horen valt. Hooft u gepiep dan weet u dat er een vleermuis in de buurt is. Wordt deze in zijn slaap gestoord dan grijpt hij de indringer en smijt hem in een willekeurige grot diep in de mijn. U moet er wel rekening mee houden dat vleermuizen op het licht afkomen.

Wanneer u de lamp aan heeft, ziet u de luchtschachten. Is de lamp uit dan valt u er onherroepelijk in en komt in één van de lager gelegen grotten terecht. Heeft u ten slotte het goud gevonden dan bent u er nog niet. U moet ook nog naar buiten zien te komen door één van de uitgangen. Ook hierbij worden aanwijzingen gegeven. Andere schatzoekers zijn u voorgegaan, maar zij zijn er nooit levend uitgekomen.

Om zich door het grottenstelsel te verplaatsen, zijn de volgende commando's beschikbaar: NOORD, ZUID, OOST, WEST, NAARBOVEN en OMLAAG. Heeft men eenmaal een uitgang gevonden dan kan men met het commando VERLATEN weer in de buitenlucht komen.

Suggestie 1: Gezien de opbouw van het programma is het vrij eenvoudig om een andere locatie te kiezen, bijvoorbeeld een eiland met piraten.



Suggestie 2: Beschikt men over veel geheugenruimte dan kan men de grotten namen geven in plaats van nummers. Het spel wordt er zeker aantrekkelijker door.

Suggestie 3: In het programma wordt een lamp gebruikt. Als men valt, gaat de lamp niet uit en hij gaat ook niet kapot. Door nu de lamp als het ware met petroleum te vullen, kan worden gesuggereerd dat zo'n lamp ook brandstof verbruikt. Valt men tot overmaat van ramp in een schacht dan verliest men een gedeelte van de nog resterende brandstof.

Suggestie 4: Gezien de opbouw is het spel op een heel eenvoudige manier drastisch uit te breiden met allerlei gevaren die de schatzoeker bedreigen. Wat dacht u van plotseling verschijnende monsters, gifdampen en dergelijke zaken?

Programma

```
10 REM *** KOLENMIJN ***
20 TIME = RND(-TIME) * 1000
30 CLEAR 1000
40 DIM A(100,6), B(100), V(5), T$(9)
50 GR$=CHR$(1)
60 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
70 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
80 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
90 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
100 GOTO 3560
110 REM *** SPELREGELS ***
120 BD$=CHR$(188):BO$=CHR$(249)
130 GOSUB 3470
140 LOCATE 1,4
150 PRINT"De bedoeling van dit spel is een schat"
160 PRINT" te zoeken en vervolgens met de schat"
170 PRINT" de buitenwereld te bereiken."
180 PRINT" Voor het zich verplaatsen van de ene"
190 PRINT" grot naar de andere zijn de volgende"
200 PRINT" commando's beschikbaar: - NOORD, ZUID,"
210 PRINT" OOST, WEST, NAAR BOVEN en OMLAAG -."
220 PRINT" Je hebt een lamp bij je om wat te"
230 PRINT" kunnen zien. Met het commando 'AAN'"
240 PRINT" geeft de lamp licht, met het commando"
250 PRINT" 'UIT' is de lamp gedoofd."
260 PRINT
270 PRINT" LET OP! De vlemuizen komen op het"
280 PRINT" licht af."
290 LOCATE 0,20
300 PRINT T1$
310 PRINT R6$;" HEB JE DIT GELEZEN DRUK <RETURN> IN ";R6$
320 PRINT T2$
330 GOSUB 3720
340 A$ = INKEY$
350 IF A$ <> CHR$(13) GOTO 340
360 GOSUB 3470
```

```

370 LOCATE 0,7
380 PRINT" Heb je een uitgang gevonden dan kun je"
390 PRINT" de mijn verlaten met het commando:"
400 PRINT
410 PRINT"          `VERLATEN`"
420 PRINT
430 PRINT" Je bent niet de enige schatzoeker ge-"
440 PRINT" weest. Misschien kom je ze tegen ?"
450 PRINT
460 PRINT"          DOOD OF LEVEND !!"
470 LOCATE 0,19:PRINT T1$
480 PRINT R6$;" RUSTIG !!!                                ";R6$
490 PRINT R6$;" Er is een instorting in de kolenmijn";R6$
500 PRINT T2$
510 RETURN
520 :
530 REM *** INITIALIZEREN ***
540 REM ** TEKSTEN + COMMANDO'S **
550 T$(1) = "West "
560 T$(2) = "Oost "
570 T$(3) = "Noord "
580 T$(4) = "Zuid "
590 T$(5) = "Naar boven "
600 T$(6) = "Omlaag "
610 T$(7) = "Verlaten "
620 T$(8) = "Aan"
630 T$(9) = "Uit"
640 REM ** STUURVARIABLELE WAARDE GEVEN **
650 Z = 0 : `ZETTEN TELLER
660 Q = 0 : `GOUD LIGT
KLAAR
670 G = 1 : `GROTNUMMER
STARTPOSITIE
680 D = 1 : `VORIG
GROTNUMMER
690 REM ** STRING$ WISSEN **
700 GOSUB 2250
710 REM ** GROTTEEN SCHOONMAKEN **
720 FOR I = 1 TO 100
730 B(I) = 1
740 FOR J = 1 TO 6
750 A(I,J) = 0
760 NEXT J
770 NEXT I
780 REM ** GROTTEENSTELSEL IN 3-D RUIMTE BOUWEN **
790 REM * WEST = KOLOM 1 OOST = KOLOM 2 *
800 FOR X = 5 TO 100 STEP 5
810 I = X - 4
820 A(I,1) = X : `WEST
830 A(X,2) = I : `OOST
840 FOR Y = I + 1 TO I + 4
850 A(Y,1) = Y - 1
860 A(Y-1,2) = Y
870 NEXT Y
880 NEXT X
890 REM * NAAR BOVEN = KOLOM 5 OMLAAG = KOLOM 6 *
900 FOR I = 21 TO 100

```

```

910 A(I,5) = I - 20 : `NAAR BOVEN
920 A(I-20,6) = I : `OMLAAG
930 IF I > 80 GOTO 950
940 B(I) = 0 : `CONTROLE
950 NEXT I
960 REM * NOORD = KOLOM 3 ZUID = KOLOM 4 *
970 FOR Y = 16 TO 96 STEP 20
980 FOR X = 0 TO 4
990 H = X + Y - 15
1000 A(H,3) = Y + X : `NOORD
1010 A(H,4) = A(H,3) - 10 : `ZUID
1020 NEXT X
1030 FOR X = 5 TO 19
1040 H = X + Y - 15
1050 A(H,3) = X + Y - 20 : `NOORD
1060 A(H,4) = A(H,3) + 10 : `ZUID
1070 IF X < 15 GOTO 1090
1080 A(H,4) = A(H,3) - 10
1090 NEXT X
1100 NEXT Y
1110 REM ** WILLEKEURIGE DOORGANGEN DICHTMAKEN **
1120 Y = INT(RND(-TIME) * 100 + 1) : `AANTAL
DOORGANGEN = MAX.10
1130 FOR X = 1 TO Y
1140 J = INT(RND(-TIME) * 6 + 1)
: `KOLOMBEPALING
1150 I = INT(RND(-TIME) * 100 + 1)
: `REGELBEPALING
1160 IF B(I) >= 4 GOTO 1140
1170 B(I) = B(I) + 1
1180 A(I,J) = 0
1190 NEXT X
1200 FOR I = 1 TO 100
1210 B(I) = 0
1220 NEXT I
1230 REM ** SCHACHTEN PLAATSEN = 1 **
1240 Y = INT(RND(-TIME) * 20 + 1) : `AANTAL
SCHACHTEN
1250 FOR X = 1 TO Y
1260 I = INT(RND(-TIME) * 100 + 1) : `GROTBEPALING
1270 B(I) = 1
1280 NEXT X
1290 REM ** UITGANGEN MAKEN = 2 **
1300 B(1) = 2 : `HOOFDUITGANG
1310 A(1,5) = -10 : `DOORGANG
OMHOOG
1320 FOR X = 1 TO 5 : `5 UITGANGEN
1330 I = INT(RND(-TIME) * 18 + 2) : `GROTBEPALING
1340 B(I) = 2 : `UITGANG
PLAATSEN
1350 A(I,5) = -10
1360 NEXT X
1370 REM ** GOUD PLAATSEN = 3 **
1380 I = INT(RND(-TIME) * 90 + 11) : `GROTBEPALING
1390 IF B(I) <> 0 GOTO 1380 : `IS GROT VRIJ
?
1400 B(I) = 3 : `PLAATS VAN HET

```



```

GOUD
1410 REM ** 5 VLEERMUIZEN PLAATSEN IN MAT V **
1420 FOR X = 1 TO 5
1430 V(X) = INT(RND(-TIME) * 80 + 21)      : `SLAAPPLAATS
VLEERMUIZEN
1440 NEXT X
1450 RETURN
1460 :
1470 REM *** UITVOER ***
1480 COLOR 3,1,1:CLS
1490 LOCATE 0,0:PRINT T1$
1500 PRINT R6$;" --- KOLENMIJN ----- ZET :      ";R6$
1510 LOCATE 34,1:PRINT USING"###";Z
1520 PRINT T2$
1530 LOCATE 0,3:PRINT T1$
1540 FOR I = 4 TO 5
1550 LOCATE 0,I:PRINT R6$;SPACE$(37);R6$
1560 NEXT I
1570 LOCATE 0,6:PRINT T2$
1580 LOCATE 1,4:PRINT A$                      : `AANWIJZING
INVOER I
1590 LOCATE 1,5:PRINT B$                      : `AANWIJZING
INVOER II
1600 LOCATE 0,7
1610 PRINT" Je bevindt je in grotnummer : ";USING"###";G
1620 PRINT" Je was in grotnummer      : ";USING"###";D
1630 LOCATE 1,10
1640 PRINT"In volgende richtingen zijn doorgangen:"
1650 FOR J = 1 TO 6
1660 IF A(G,J) = 0 GOTO 1690
1670 IF J = 5 THEN LOCATE 0,13:PRINT" -";T$(J):GOTO 1690
1680 PRINT" -";T$(J);                          : `BESCHIKBARE
V-COMMANDO
1690 NEXT J
1700 LOCATE 34,11:PRINT"N"
1710 LOCATE 34,12:PRINT BO$
1720 LOCATE 32,13:PRINT"W";BO$;BO$;BO$;"O"
1730 LOCATE 34,14:PRINT BO$
1740 LOCATE 34,15:PRINT"Z"
1750 LOCATE 0,16:PRINT T1$
1760 FOR I = 17 TO 20
1770 PRINT R6$;SPACE$(37);R6$
1780 NEXT I
1790 LOCATE 0,21:PRINT T2$
1800 FOR I = 17 TO 20
1810 LOCATE 1,I:PRINT SPACE$(37)
1820 NEXT I
1830 IF C$ <> "" THEN LOCATE 2,17:PRINT C$
1840 IF D$ <> "" THEN LOCATE 2,18:PRINT D$
1850 IF E$ <> "" THEN LOCATE 2,19:PRINT E$
1860 IF F$ <> "" THEN LOCATE 2,20:PRINT F$
1870 RETURN
1880 :
1890 REM *** INVOER + CONTROLE ***
1900 LOCATE 1,22:PRINT"Wat is je keuze : ";
1910 LINE INPUT K$
1920 K$ = LEFT$(K$,2)                          : `EERSTE TWEE

```

```

KARAKTERS
1930 V = 0
1940 H = 0
1950 FOR J = 1 TO 9                                : 'H=0 FOUT
ANTWOORD
1960 IF LEFT$(T$(J),2) <> K$ GOTO 2190
1970 H = 1
1980 REM ** MOET LAMP AAN ? **
1990 IF J <> 9 GOTO 2030
2000 A$ = "Lamp is uit !!"
2010 L = 0                                          : 'LAMP IS UIT
2020 REM ** MOET LAMP AAN ? **
2030 IF J <> 8 GOTO 2070
2040 A$ = "Lamp is aan !!"
2050 L = 1                                          : 'LAMP IS AAN
2060 REM ** VERPLAATSEN ? **
2070 IF J <> 7 GOTO 2120
2080 V = -10                                       : 'RELATIEF
VERPLAATSEN
2090 IF B(G) = 2 GOTO 2120
2100 A$ = "Sufferd, er is hier geen uitgang !!"
2110 V = 0                                          : 'VERPLAATST.
ONGEDAAN
2120 IF J > 6 GOTO 2180
2130 V = A(G,J)                                    : 'VERPLAATST ?
2140 IF A(G,J) <> 0 GOTO 2180
2150 A$ = "Je bent tegen een muur aangeknald !!"
2160 B$ = "Er is hier geen doorgang."
2170 V = 0                                          : 'VERPLAATS.
ONGEDAAN
2180 J = 9                                          : 'UIT LUS
2190 NEXT J
2200 IF H <> 0 GOTO 2220
2210 A$ = "Ik begrijp je niet! Wat bedoel je!?"
2220 RETURN
2230 :
2240 REM *** OPHOGEN TELLER + STRING$ WISSEN ***
2250 Z = Z + 1                                     : 'ZETTEN
VERHOGEN
2260 D = G                                          : 'VORIGE =
HUIDIGE
2270 A$ = ""
2280 B$ = ""
2290 C$ = ""
2300 D$ = ""
2310 E$ = ""
2320 F$ = ""
2330 RETURN
2340 :
2350 REM *** VLEERMUIS ***
2360 REM ** AANWIJZING = C$ **
2370 REM ** OP LICHT AFKOMEN **
2380 IF L <> 1 GOTO 2490                            : 'L=1 LICHT
AAN!
2390 FOR X = 1 TO 5
2400 H = ABS(V(X) - G)                             : 'AFSTAND TOT
VLEERMUIS

```

```

2410 IF H > 30 GOTO 2470           : 'TE VER WEG ?
2420 REM ** VLEERMUIS HERKENT LICHT **
2430 IF V(X) => G GOTO 2450
2440 V(X) = V(X) + 1             : 'VLEERM. ONDER
JE
2450 IF V(X) <= G GOTO 2470
2460 V(X) = V(X) - 1           : 'VLEERM. BOVEN
JE
2470 NEXT X
2480 REM ** GRIJPEN VLEERMUIS **
2490 FOR J = 1 TO 6             : 'AANTAL
MOGELIJKE DOORGANGEN = 6
2500 IF A(G,J) <= 0 GOTO 2550
2510 FOR X = 1 TO 5
2520 IF A(G,J) <> V(X) GOTO 2540 : 'VLEERMUIS
ERBIJ ?
2530 C$ = "Er is gepiep te horen."
2540 NEXT X
2550 NEXT J
2560 REM ** GRIJPEN **
2570 FOR X = 1 TO 5
2580 IF V(X) <> G GOTO 2630     : 'VLEERMUIS
GRIJPT?
2590 C$ = "Je werd door een vleermuis gegrepen!"
2600 G = INT(RND(-TIME) * 90 + 11) : 'WILLEKEURIGE
GROT
2610 V(X) = INT(RND(-TIME) * 90 + 11) : 'NIEUWE
SLAAPPLAATS
2620 X = 5                      : 'UIT LUS
2630 NEXT X
2640 RETURN
2650 :
2660 REM *** SCHACHT ***
2670 REM ** AANWIJZING = D$ **
2680 IF B(G) <> 1 GOTO 2740     : 'DOORGANG ?
2690 D$ = "Een schacht, gelukkig was de lamp aan!"
2700 IF L <> 0 GOTO 2740       : 'LAMP AAN ?
2710 REM ** LAMP WAS UIT **
2720 G = G + INT(RND(-TIME) * (100 - G)) : 'GROTTEN
OMLAAG
2730 D$ = "Je bent door luchtschacht gevallen!!"
2740 RETURN
2750 :
2760 REM *** UITGANG ***
2770 REM ** AANWIJZINGEN = E$ **
2780 IF (G > 41) OR (Z < 11) GOTO 2880
2790 FOR J = 1 TO 6
2800 IF A(G,J) <= 0 GOTO 2840
2810 H = A(G,J)                : 'GROT
ERNAAST/BOVEN
2820 IF B(H) <> 2 GOTO 2840
2830 E$ = "Er is hier een zwak lichtschijnsel."
2840 NEXT J
2850 REM ** GROT MET UITGANG **
2860 IF B(G) <> 2 GOTO 2880
2870 E$ = "Je kunt de zon zien schijnen."
2880 RETURN

```



```

2890 :
2900 REM *** GOUD ***
2910 REM ** AANWIJZING = F$ **
2920 IF Q = 1 GOTO 3120 : 'HEEFT GOUD
?
2930 REM ** VLAKBIJ GOUD ? **
2940 FOR I = G - 1 TO G + 1
2950 IF (I < 1) OR (I > 100) GOTO 3060
2960 FOR J = 1 TO 6
2970 IF A(I,J) <= 0 GOTO 3050
2980 H = A(I,J) : 'GROT
ERNAAST/BOVEN
2990 IF B(H) <> 3 GOTO 3050
3000 F$ = "Er staan hier voetstappen."
3010 IF I <> G GOTO 3050
3020 F$ = "Er ligt hier een geraamte."
3030 J = 5 : 'UIT LUS
3040 I = G + 1
3050 NEXT J
3060 NEXT I
3070 REM ** GOUD GEVONDEN **
3080 IF B(G) <> 3 GOTO 3120
3090 F$ = "Je hebt het goud gevonden."
3100 Q = 1
3110 B(G) = 0 : 'GOUD
MEEGENOMEN
3120 RETURN
3130 :
3140 REM *** AKTIE ***
3150 G = V : 'NIEUWE GROT
3160 IF V <> 0 GOTO 3180
3170 G = D
3180 IF G <= 0 GOTO 3270
3190 REM ** VLEERMUIS **
3200 GOSUB 2380
3210 REM ** SCHACHT **
3220 GOSUB 2680
3230 REM ** UITGANG **
3240 GOSUB 2780
3250 REM ** GOUD **
3260 GOSUB 2920
3270 RETURN
3280 :
3290 REM *** RESULTAAT ***
3300 CLS:COLOR 14,4,4
3310 LOCATE 0,4:PRINT T1$
3320 FOR I = 5 TO 17
3330 LOCATE 0,I:PRINT"V";SPACE$(37);"V"
3340 NEXT I
3350 LOCATE 0,18:PRINT T2$
3360 LOCATE 1,8:PRINT"Eindelijk de frisse lucht!!"
3370 LOCATE 1,10:PRINT"Je bent er heelhuids uitgekomen,"
3380 A$ = "maar zonder het goud, STOMMELING."
3390 IF Q <> 1 GOTO 3410
3400 A$ = "met het goud... FANTASTISCH!!!!"
3410 LOCATE 1,11:PRINT A$
3420 LOCATE 1,14


```

```

3430 PRINT "In totaal had je er";Z;"zetten nodig."
3440 LOCATE 0,20
3450 RETURN
3460 :
3470 CLS
3480 T1$ = R2$+LEFT$(S1$,74)+R3$
3490 T2$ = R4$+LEFT$(S1$,74)+R5$
3500 LOCATE 0,0:PRINT T1$
3510 PRINT R6$;" --- KOLENMIJN ---          - SPELREGELS- ";R6$
3520 LOCATE 0,2:PRINT T2$
3530 RETURN
3540 REM **** STUURMODULE ****
3550 REM *** SPELREGELS ***
3560 GOSUB 120
3570 REM *** INITALIZEREN ***
3580 GOSUB 550
3590 REM *** UITVOER ***
3600 GOSUB 1480
3610 REM *** OPHOGEN TELLER ***
3620 GOSUB 2250
3630 REM *** INVOER ***
3640 GOSUB 1900
3650 REM *** AKTIE ***
3660 GOSUB 3150
3670 IF G > 0 GOTO 3600
3680 REM *** RESULTAAT ***
3690 GOSUB 3300
3700 END
3710 *** MUZIEK ***
3720 PLAY "V12T180","V10T180","V9T180"
3730 PLAY "O4C8E8G8R8GO5CO4A2B-8A8GO5CO4AF2","O3CB-8R8B-8R8FO
4C8R8C8R8O3CB-8R8B-8R8FO4C8R8C8R8","RO3E8R8E8R8RA8R8A8R8RE8R
8E8R8RA8R8A8R8"
3740 PLAY "O4C8E8G8R8GO5CO4A2B-8A8GO5CO4AF2","O3CB-8R8B-8R8FO
4C8R8C8R8O3CB-8R8B-8R8FO4C2","RO3E8R8E8R8RA8R8A8R8RE8R8E8R8R
A2"
3750 PLAY "O5F8R8C2F8R8C2E8D8C2E8D8CO4","O3FO4C8R8C8R8O3FO4C8
R8C8R8O3GO4C8R8C8R8O3CO4C8R8C8R8","RO3A8R8A8R8RA8R8A8R8RB-8R
8B-8R8RB-8R8B-8R8"
3760 PLAY "A8R8AF8R8F8E8F8A8G8F8E8F8G8C8D8E8F8R8F8","O3FO4C8R
8C8R8O3FO4C8R8C8R8O3CB-8R8B-8R8A8R8A8","RA8R8A8R8RA8R8A8R8RE
8R8E8R8F8R8F8"
3770 RETURN

```

Voorbeeld



4.2 Boerderij Zeldenrust

In dit eenvoudige simulatiespel speelt men een boer die twee gewassen verbouwt: gierknollen, die overigens erg lekker zijn, en blauwe kool. Het doel van dit spel is het binnen halen van een zo groot mogelijke oogst.

Gierknolplanten sterven af wanneer de temperatuur voor langere tijd onder de 10 graden Celsius blijft. Zij kunnen echter warm worden gehouden door giervliegen die in grote kolonies op de gierknolplant nestelen. Aan het eind van de oogst moeten de gierknollen wel worden schoongemaakt. Een nadeel is dat de giervliegen verzot zijn op blauwe koolstruiken.

Kaalspinnen jagen voornamelijk op giervliegen en bijen. De bijen zorgen voor de bestuiving van de blauwe kool. Wanneer er een sluipmuggenplaag is, vreten kaalspinnen ook deze insecten.

Als het gedurende een langere tijd droog is, moet men voor bevloeiing de irrigatiekanalen openen. De kanalen zijn tevens de broedplaatsen van de gevreesde sluipmug. De sluipmuggen zijn berucht vanwege hun vraatzucht. Beide gewassen hebben hier erg onder te lijden.

Om de insectenpopulaties onder controle te houden, zijn drie insecticiden beschikbaar die echter ook een schadelijke uitwerking op de gewassen hebben. Men heeft de keuze uit:

APV: 10% van de insecten en 3,5% van de gewassen sterft.

DAMAX: 40% van de insecten en 25% van de gewassen sterft.

ZWOP: 75% van de insecten en 50% van de gewassen sterft.

Men moet proberen om de verschillende insectenpopulaties in de hand te houden, om een zo groot mogelijke opbrengst te krijgen. Dit probleem kent zeer veel oplossingen. Dit komt onder andere doordat de weersgesteldheid een grote invloed uitoefent. Het betekent ook dat het spel steeds opnieuw kan worden gespeeld, zonder dat het gaat vervelen.

Suggestie: De groei van het gewas is niet te beïnvloeden; men is afhankelijk van het weer. In de moderne akkerbouwbedrijven wordt veel met kunstmest gewerkt. Dit element zou men ook nog aan het programma kunnen toevoegen. Men moet er wel op bedacht zijn dat er niet teveel kunstmest wordt gebruikt. Gebruikt men teveel dan verliest men een gedeelte van zijn oogst. Het is een kwestie een extra module toevoegen.

Programma

```
10 REM *** BOERDERIJ ZELDENRUST ***
20 CLEAR 2000
30 DIM A(3,2)
40 GR$=CHR$(1)
50 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
60 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
70 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
80 S9$="" :FOR RS=1 TO 40:S9$=S9$+R1$:NEXT RS
```

```

90 DATA 0.1, 0.04, 0.4, 0.25, 0.75, 0.5
100 TIME = RND(-TIME)
110 GOSUB 3380
120 GOTO 3160
130 :
140 REM *** INITIALIZEREN ***
150 A = 0 : `AANTAL
DROGE DAGEN
160 D = 0
: `DAGTELLER
170 G = 500
: `GRUWELKNOLLEN
180 B = 500 : `BLAUWE
KOOL
190 S = 0 : `SLUISJES
200 D2 = 0 : `DAGEN >
20 C
210 L = 0 : `SLUISJES
220 K = 0 : `SPEL
LOOPT
230 K1 = 0
240 REM ** BESTRIJDINGSMIDDELEN INLEZEN **
250 FOR I = 1 TO 3
260 READ A(I,1), A(I,2)
270 NEXT I
280 REM ** WEERSGESTELDHEID - STARTWAARDEN **
290 R = INT(RND(-TIME) * 10) + 1 : `KANS
REGEN/DROOGTE
300 T = INT(RND(-TIME) * 45) - 5
: `TEMPERATUUR
310 REM ** INSEKTENPOPULATIE - STARTWAARDEN **
320 H = INT(RND(-TIME) * 500) + 1 : `BLAUWE
HOMMELS
330 P = INT(RND(-TIME) * 500) + 1
: `KAALSPRINGHANEN
340 M = INT(RND(-TIME) * 500) + 1 : `RODE

```




```

GRUWELMIEREN
350 S = 0
: `SLUIPMUGGEN
360 RETURN
370 :
380 REM *** UITVOER NAAR BEELDSCHERM ***
390 CLS:COLOR 1,5,5:SCREEN 0:WIDTH 40:KEY OFF : `SCHERM
SCHOONMAKEN
400 POKE &HF3B1,26
410 GOSUB2840
420 LOCATE 12,3:PRINT R2$;S1$;R3$
430 LOCATE 12,4:PRINT R6$;" TEMPERATUUR : ";R6$
440 LOCATE 12,5:PRINT R6$;" WEERSVERWACHTING: ";R6$
450 LOCATE 12,6:PRINT R4$;S1$;R5$
460 LOCATE 1,7:PRINT"-- GEWASSEN --"
470 LOCATE 7,8:PRINT"Aantal Gruwelknollen : "
480 LOCATE 7,9:PRINT"Aantal Blauwe Kool : "
490 LOCATE 0,10:PRINT STRING$(39,249)
500 LOCATE 1,11:PRINT"-- INSEKTEN-POPULATIES --"
510 LOCATE 7,12:PRINT"Aantal Blauwe Hommels : "
520 LOCATE 7,13:PRINT"Aantal Kaalspringhanen : "
530 LOCATE 7,14:PRINT"Aantal Rode Gruwelmierren : "
540 LOCATE 7,15:PRINT"Aantal sluipmuggen : "
550 LOCATE 4,16:PRINT"** AANTAL SLUISJES : "
560 LOCATE 0,17:PRINT R2$+S$+R3$
570 FOR I = 18 TO 21
580 LOCATE 0,I:PRINT R6$;SPACE$(37);R6$
590 NEXT I
600 LOCATE 0,22:PRINT"Z"+S$+"["
610 IF K1 <> 0 THEN LOCATE 2,10:PRINT">> I N S E K T E N P
L A A G <<":ELSE GOTO 630
620 FOR CO = 2 TO 14:COLOR 1,CO,CO:FOR TT=1 TO 50:NEXT
TT:NEXT CO
630 COLOR 1,5,5:GOSUB 2900:VL = 1
640 RETURN
650 :
660 REM *** CONTROLE OP < 0 ***
670 IF G < 0 THEN G = 0
: `GRUWELKNOLLEN
680 IF B < 0 THEN B = 0 : `BLAUWE
KOOL
690 IF P < 0 THEN P = 0
: `SPRINKHANEN
700 IF H < 0 THEN H = 0 : `HOMMELS
710 IF M < 0 THEN M = 0 : `MIEREN
720 IF S < 0 THEN S = 0
: `SLUIPMUGGEN
730 RETURN
740 :
750 REM *** BEWERKINGSMODULE ***
760 D = D + 1
: `DAGTELLER
770 REM ** WEERGESTELDHEID **
780 Q = 0
790 REM * TEMPERATUUR *
800 IF T <= 10 THEN Q = Q + 1
810 IF T < 3 THEN Q = Q + .5

```



```

820 IF T < 0 THEN Q = Q + .75
830 IF T < -3 THEN Q = Q + 1
840 IF T > 35 THEN Q = Q + 1
850 IF T > 37 THEN Q = Q + 1.75
860 R = INT(RND(-TIME) * 10) + 1           : `KANS
REGEN/BUI/DROOGTE
870 A$ = "Droog"
880 A = A + 1
890 IF R <= 4 GOTO 960
900 A$ = "Bui"
910 A = A - 1
920 IF R < 7 GOTO 960
930 A = A - 2
940 A$ = "Regen"
950 REM ** NIEUWE TEMPERATUUR **
960 T1 = T                                 : `OUDE
TEMPERATUUR
970 X = INT(RND(-TIME) * 11) - 5
: `TEMPERATUURVERANDERING
980 IF X + T > 40 OR X + T < -5 THEN X = X * -1
990 T = T + X
1000 IF T > 20 THEN D2 = D2 + 1           : `WARME
DAGEN
1010 RETURN
1020 :
1030 REM *** INVLOEDEN OP GROEI EN INSEKTENPOPULATIES ***
1040 K1 = 0                                : `GEEN
INSEKTENPLAAG
1050 X = P + M + S + H                     : `TOTALE
INSEKTENPOPULATIE
1060 Y = (G + B) * 3 + 4500                : `GRENS
INSEKTENPLAAG
1070 IF X < Y GOTO 1110
1080 K1 = 1
: `INSEKTENPLAAG
1090 B = INT(B - RND(-TIME) * B/2 - B/2)   : `VRAAT
KOOL
1100 G = INT(G - RND(-TIME) * G/2 - G/2)   : `VRAAT
KNOL
1110 X = RND(-TIME) * H/1000
: `EFFECTIVITEIT HOMMELS
1120 B = B + INT(X * B)
1130 G = G - INT(RND(-TIME) * X * G)
1140 REM ** EFFECTIVITEIT RODE GRUWELMIEREN **
1150 X = INT(RND(-TIME) * M/2 + M/2)
1160 Y = RND(-TIME) * 2
1170 REM ** EFFECT VAN TEMPERATUUR OP KNOL **
1180 IF Q <= Y GOTO 1230
1190 IF G > X/200 GOTO 1220
1200 G = INT(G - (RND(-TIME) * G/6))
1210 GOTO 1230
1220 G = INT(RND(-TIME) * X/200 + (2/3) * (X/200))
1230 X = X/1500
1240 B = INT(B - (B * X))
1250 G = INT(G + (G * X))
1260 REM ** EFFECT KAALSPRINKHANEN **
1270 X = INT((RND(-TIME) * P)/5)           : `FACTOR

```

```

SPRINKHANEN
1280 H = H - X                               : `UITROEI
HOMMELS
1290 M = M - X                               : `UITROEI
MIEREN
1300 S = S - INT(X * RND(-TIME))
1310 REM ** EFFECT SLUIPMUGGEN **
1320 S = S + (L * INT(RND(-TIME) * 100 + 1))
: `SLUIPMUGGEN
1330 IF S < 5 GOTO 1390
1340 X = INT(((RND(-TIME) * S) + 1)/20)
1350 G = G - X
: `VRAATKNOLEN
1360 X = INT(((RND(-TIME) * S) + 1)/20)
1370 B = B - X
: `VRAATKOOL
1380 REM ** DROOGTESCHADE **
1390 IF A - L < 5 GOTO 1420                   : `VOLDOENDE
SLUIZEN
1400 G = INT(G - G * RND(-TIME))
1410 B = INT(B - B * RND(-TIME))
1420 GOSUB 670                               : `CONTROLE
OP NUL
1430 RETURN
1440 :
1450 REM *** VRAAGSTELLING ALGEMEEN ***
1460 I = 0
1470 GOSUB 3110
1480 LOCATE 1,18
1490 PRINT"0 > Niets doen      3 > Sluisje openen"
1500 LOCATE 1,19
1510 PRINT"1 > Inskt. vrij    4 > Sluisje sluiten"
1520 LOCATE 1,20
1530 PRINT"2 > Bestrijdingsmiddel spuiten"
1540 LOCATE 22,23:PRINT SPACE$(14)
1550 LOCATE 1,23:PRINT"*** WAT IS UW KEUZE : ";
1560 LINE INPUT I$
1570 I = VAL(I$)
1580 IF I < 0 OR I > 4 GOTO 1460
1590 IF I = 3 THEN L = L + 1
1600 IF I = 4 AND L > 0 THEN L = L - 1
1610 GOSUB 3330
1620 RETURN
1630 :
1640 REM *** INSEKTEN VRIJLATEN ***
1650 COLOR 1,3,3:GOSUB 2900
1660 GOSUB 3110
1670 LOCATE 2,18:PRINT"1 >> Blauwe Hommels"
1680 LOCATE 2,19:PRINT"2 >> Kaalsprinkhanen"
1690 LOCATE 2,20:PRINT"3 >> Rode gruwelmieren"
1700 LOCATE 2,21:PRINT"4 >> Sluipmuggen"
1710 LOCATE 0,23:PRINT"*** WELKE SOORT MOET WORDEN
UITGEZET:";
1720 INPUT X
1730 IF X < 1 OR X > 4 GOTO 1660
1740 LOCATE 0,23:PRINT SPACE$(39)
1750 LOCATE 0,23:PRINT"*** HOEVEEL VAN DEZE INSEKTEN:";

```

```

1760 BEEP:INPUT Y
1770 Y = ABS(INT(Y))
1780 IF Y < 2500 GOTO 1820
1790 K1 = K
: `INSEKTENPLAAG
1800 G = INT(G - (RND(-TIME)) * G/2 + G/2)
1810 B = INT(B - (RND(-TIME)) * B/2 + B/2)
1820 IF X = 1 THEN H = H + Y
: `HOMMELS
1830 IF X = 2 THEN P = P + Y
: `SPRINKHANEN
1840 IF X = 3 THEN M = M + Y
: `MIEREN
1850 IF X <> 4 GOTO 1920
1860 LOCATE 6,10
1870 PRINT"WAAROM";Y;"SLUIPMUGGEN!!"
1880 S = S + Y
1890 FOR TT = 1 TO 200
: `WACHTLUS
1900 NEXT TT
1910 LOCATE 0,10:PRINT STRING$(39,"*")
1920 GOSUB 670
1930 RETURN
1940 :
1950 REM *** VRAAGSTELLING BESTRIJDINGSMIDDELEN ***
1960 I = 0
1970 COLOR 1,8,8:GOSUB 2900
1980 GOSUB 3110
1990 LOCATE 1,18
2000 PRINT"1 > APV    10% Insecten    4% gewassen"
2010 LOCATE 1,19
2020 PRINT"2 > SLEM  40%    +-+    25%    +-+"
2030 LOCATE 1,20
2040 PRINT"3 > ZWOP  75%    +-+    50%    +-+"
2050 LOCATE 1,23
2060 PRINT"* WELK SPUITMIDDEL GEBRUIKEN:";
2070 INPUT I
2080 IF I < 1 OR I > 3 GOTO 1980
2090 I = INT(I)
2100 REM ** EFFECTEN INSECTICIDEN **
2110 Y = A(I,1) : `EFFECT OP
GEWAS
2120 X = A(I,2) : `EFFECT OP
INSECT
2130 G = INT(G - (G * X)) : `EFFECT OP
KNOL
2140 B = INT(B - (B * X)) : `EFFECT OP
KOOL
2150 H = INT(H - (H * Y)) : `EFFECT OP
HOMMEL
2160 M = INT(M - (M * Y)) : `EFFECT OP
MIER
2170 P = INT(P - (P * Y)) : `EFFECT OP
SPRINKHAAN
2180 S = INT(S - (S * Y)) : `EFFECT OP
SLUIPMUGGEN
2190 GOSUB 670

```



```

2200 RETURN
2210 :
2220 REM *** GROEI INSECTEN-POPULATIE ***
2230 S = INT(S + S * .1) : 'UITKOMEN
VAN SLUIPMUGGEN
2240 H = INT(H + H * .11) : 'UITKOMEN
VAN HOMMELS
2250 M = INT(M + M * .12) : 'UITKOMEN
VAN MIEREN
2260 P = INT(P + P * .15) : 'UITKOMEN
VAN SPRINKHANEN
2270 GOSUB 670
2280 RETURN
2290 :
2300 REM *** GROEI GEWASSEN ***
2310 X = D/5
2320 IF X <> INT(D/5) GOTO 2420 : 'GROEISTOT
ELKE 5 DAGEN
2330 REM ** GROEI AFHANKELIJK VAN VOCHTIGHEID EN WARMTE **
2340 Y = INT(G * (.06 + (RND(-TIME) * (D2 + L - A))/100))
2350 IF M < 5 THEN Y = 0
2360 G = G + Y
2370 Y = INT(B * (.06 + (RND(-TIME) * (D2 + L - A))/100))
2380 IF H < 5 THEN Y = -INT(RND(-TIME) * Y)
2390 B = B + Y
2400 D2 = 0
2410 GOSUB 670
2420 RETURN
2430 :
2440 REM *** VERLOREN ***
2450 CLS:COLOR 14,1,1 : 'SCHERM
WISSEN
2460 GOSUB 2840
2470 LOCATE 34,1:PRINT USING"###";D
2480 LOCATE 0,7:PRINT R2$+S$+R3$
2490 FOR LI = 8 TO 15
2500 LOCATE 0,LI:PRINT R6$;SPACE$(37);R6$
2510 NEXT LI
2520 LOCATE 0,16:PRINT R4$+S$+R5$
2530 LOCATE 1,8:PRINT"Je bent er in geslaagd het grootste"
2540 LOCATE 1,9:PRINT"gedeelte van de oogst te
vernietigen."
2550 LOCATE 1,11:PRINT"Je hebt nog maar";G;"gruwelknollen"
2560 LOCATE 1,12:PRINT"en";B;"blauwe kool."
2570 LOCATE 1,14:PRINT"Zo'n luie boer bestaat niet!?"
2580 IF K1 = 0 GOTO 2600
2590 LOCATE 1,15:PRINT"JE HEBT EEN INSEKTENPLAAG
ONTKETEND!"
2600 K = 1
2610 RETURN
2620 :
2630 REM *** GEWONNEN ***
2640 CLS
2650 FOR CO = 2 TO 12
2660 COLOR 1,CO,CO
2670 FOR TT = 0 TO 100:NEXT TT
2680 NEXT CO

```

```

2690 GOSUB 2840
2700 LOCATE 0,7:PRINT STRING$(39,240)
2710 FOR I = 8 TO 16
2720 LOCATE 0,I:PRINT"=";SPACE$(37);"="
2730 NEXT I
2740 LOCATE 34,1:PRINT USING"###";D
2750 LOCATE 0,17:PRINT STRING$(39,240)
2760 LOCATE 8,9:PRINT"DE OOGST IS BINNEN !!!!!"
2770 LOCATE 2,11:PRINT"Er zijn:"
2780 LOCATE 2,12:PRINT"-";G;"gruwelknollen en"
2790 LOCATE 2,13:PRINT"-";B;"blauwe kool geoogst."
2800 LOCATE 1,15:PRINT"JE VERDIENT EEN PRIJS VOOR
AKKERBOUW."
2810 K = 1
2820 RETURN
2830 :
2840 S$ = LEFT$(S9$,74)
2850 S1$ = LEFT$(S9$,50)
2860 LOCATE 0,0:PRINT R2$;S$;R3$
2870 LOCATE 0,1:PRINT R6$;" * * * BOERDERIJ * * *     DAG :
";R6$
2880 LOCATE 0,2:PRINT R4$;S$;R5$
2890 RETURN
2900 REM *** SCHOON MAKEN + VARIABELEN PRINTEN ***
2910 LOCATE 34,1:PRINT SPACE$(4)
2920 LOCATE 32,4:PRINT SPACE$(4)
2930 LOCATE 32,5:PRINT SPACE$(5)
2940 LOCATE 34,8:PRINT SPACE$(4)
2950 LOCATE 34,9:PRINT SPACE$(4)
2960 FOR I = 12 TO 16
2970 LOCATE 34,I:PRINT SPACE$(4)
2980 NEXT I
2990 LOCATE 34,1:PRINT USING"####";D
3000 LOCATE 32,4:PRINT USING"####";T1
3010 LOCATE 35,4:PRINT" C"
3020 LOCATE 32,5:PRINT A$
3030 LOCATE 34,8:PRINT USING"####";G
3040 LOCATE 34,9:PRINT USING"####";B
3050 LOCATE 34,12:PRINT USING"####";H
3060 LOCATE 34,13:PRINT USING"####";P
3070 LOCATE 34,14:PRINT USING"####";M
3080 LOCATE 34,15:PRINT USING"####";S
3090 LOCATE 34,16:PRINT USING"####";L
3100 RETURN
3110 REM *** EEN DEEL VAN SCHERM SCHOON ***
3120 FOR I = 18 TO 21
3130 LOCATE 1,I:PRINT SPACE$(37)
3140 NEXT I
3150 RETURN
3160 REM **** STUURMODULE ****
3170 GOSUB 150
3180 REM *** SPEL ***
3190 GOSUB 760
:WEERSGESTELDHEID
3200 IF VL = 1 THEN GOSUB 610 ELSE GOSUB 390      :`UITVOER
3210 GOSUB 1040                                   :`INVLOED
GROEI EN INSECT

```

```

3220 GOSUB 2300 : 'GROEI
GEWAS
3230 GOSUB 2230 : 'GROEI
INSECTEN-POPULATIE
3240 GOSUB 1460
: 'VRAAGSTELLING ALGEMEEN
3250 IF I = 1 THEN GOSUB 1650
: 'VRAAGSTELLING INSECTEN
3260 IF I = 2 THEN GOSUB 1960
: 'VRAAGSTELLING SPUITMIDDEL
3270 X = G + B : 'BEPALING
RESULTAAT
3280 IF X > 4000 THEN GOSUB 2640 : ' > 4000
GEWONNEN
3290 IF X < 50 THEN GOSUB 2450 : ' < 50
MISLUKT
3300 IF K = 0 GOTO 3190
3310 REM *** SPEL AFGELOPEN ***
3320 END
3330 PLAY"V13T150", "V9T150", "V9T150"
3340 PLAY"04E8F8G8R8G8R8E05C204E8F8GAFD2", "03CG8R8G8R8O2GO3G
8R8G8R8CG8R8G8R8DG8R8G8R8", "03RE8R8E8R8RE8R8E8R8RE8R8E8R8RF8
R8F8R8"
3350 PLAY"04D8E8F8R8F8R8DO5D204B.G8A8B8O5DCR", "02BO3G8R8G8R8
O2GO3G8R8G8R8O2GO3F8R8F8R8CG8R8G8R8", "03RF8R8F8R8RF8R8F8R8O2
RB8R8B8R8O3RE8R8E8R8"
3360 RETURN
3370 REM *** INTRO-SCHERM ***
3380 OPEN"GRP:" AS#1
3390 SCREEN 3
3400 COLOR 1,2,2:CLS
3410 PRESET(4,64)
3420 PRINT#1, "BOERDERY"
3430 GOSUB 3330
3440 IF PLAY(0) <> 0 THEN GOTO 3440
3450 RETURN

```


Voorbeeld



4.3 De stadhouder van Flipflopland

Een management-spel zou de beste karakterisering zijn voor dit grote spel. In dit spel moet een stadhouder een land gedurende 14 achtereenvolgende jaren goed besturen. Daarbij is het de bedoeling dat het areaal grond wordt uitgebreid. Om dit te berekenen, kan men jaarlijks aan het eind van de oogst een aantal beslissingen nemen. Men regeert als feodaal vorst.

Het zou een paar hoofdstukken kosten om alle grappen die in het spel zijn verwerkt uit te leggen. We zullen dan ook slechts een tip van de sluier oplichten. Wanneer men alles wil weten, zal men de bewerkingsmodule moeten analyseren.

Als stadhouder kan men onder andere beslissen over de hoeveelheid Flipflopers, de onderdanen. Men beslist verder hoeveel hectare grond er wordt ingezaaid, hoeveel grond er wordt gekocht of verkocht, hoeveel marktplaatsen er worden verbouwd en hoeveel er aan rattenbestrijding wordt uitgegeven.

Het spel is zodanig opgebouwd dat u eerst een aantal beslissingen moet nemen. De volgorde van deze beslissingen bepaalt u zelf aan de hand van het menu. Denkt u geen beslissingen meer te hoeven nemen dan geeft u in het menu een nul.

Men regeert voor een periode van 14 jaar. Na het verstrijken van de ambts-termijn wordt men beoordeeld op basis van de uitbreiding van de grond.



Om beslissingen te kunnen nemen, moet men op de hoogte zijn van een aantal feiten; anders wordt het een enorm zoektocht. De prijzen van landbouwgrond en marktplaatsen zijn prijzen die op een markt van volkomen mededinging ontstaan. Ze zijn wel conjunctuurgevoelig, maar niet beïnvloedbaar door de speler. De prijzen zijn afhankelijk van de prijzen van het vorige jaar, van het weer van het vorige jaar en van een willekeurige component.

Iedere Flipfopper dient aan het eind van een regeringsjaar zijn voedsel te krijgen. Men dient er rekening mee te houden dat hij vijf tins graan per jaar verorbert.

Een marktplaats kan maximaal 8000 tins graan bevatten. Kunnen de marktplaatsen de opbrengst niet bergen dan verrot de rest buiten. Een rattenplaag is dan niet denkbeeldig.

In de feodale tijd waarin dit spel speelt, was er nog geen sprake van bemesting. Men moest dus uitkijken dat men de grond niet uitputte. Daarom maakte men in die tijd gebruik van het drieslagstelsel, d.w.z. een derde gedeelte van de grond liet men altijd braak liggen.

Het weer is volkomen willekeurig: onvoorspelbaar dus. De oogst is afhankelijk van het weer.

Voor elke ha grond die men inzaait, heeft men één tin graan nodig. Zoals men al heeft opgemerkt, wordt alles uitgedrukt in tins graan.

Het zal niet meevallen om 14 jaar in het zadel te blijven terwijl misoogsten en epidemieën het stadhouderschap bedreigen.

Suggestie 1: Het programma is vrij groot en er wordt met vrij veel situaties rekening gehouden. In de module waar de Flipflopers hun voedsel krijgen, wordt een variabele gedefinieerd die de ondervoedingsgraad bijhoudt. Aan het begin van elk regeringsjaar wordt deze variabele in een vaste beginstand gebracht. Het gevolg is dat de ondervoeding alleen jaarlijks wordt bekeken en niet in de loop van de tijd. Men zou hier een wijziging in kunnen aanbrengen door de ondervoedingsgraad eveneens over de afgelopen jaren te bekijken.

Suggestie 2: In het programma wordt er zonder meer van uitgegaan dat men voldoende bewoners ter beschikking heeft om de grond in te zaaien. In werkelijkheid is dit natuurlijk niet zo. Men zou dus kunnen stellen dat per ha grond die moet worden gezaaid een bepaald aantal bewoners nodig is.

Suggestie 3: De prijsontwikkelingen van landbouwgrond en marktplaatsen volgen samen min of meer hetzelfde patroon. Het zou logisch zijn om tussen deze twee prijsverlopen een tijdsvertraging in te bouwen, zodat de prijs van een marktplaats mede afhankelijk wordt van de prijs van de landbouwgrond.

Programma

```
10 REM *** FLIPFLOPLAND ***
20 CLEAR 1000
30 TIME = RND(-TIME)
40 DIM A(7)
50 GR$=CHR$(1)
60 R1$=GR$+CHR$(87):R2$=GR$+CHR$(88):R3$=GR$+CHR$(89)
70 R4$=GR$+CHR$(90):R5$=GR$+CHR$(91):R6$=GR$+CHR$(86)
80 R7$=GR$+CHR$(81):R8$=GR$+CHR$(82)
```



```

90 S1$="":FOR RS=1 TO 40:S1$=S1$+R1$:NEXT RS
100 B0$=CHR$(188):BD$=CHR$(249)
110 GOTO 3490
120 REM *** SPELREGELS ***
130 SCREEN 0:WIDTH 40:KEY OFF:COLOR 1,5,5:CLS :`SCHERM
WISSEN
140 GOSUB 3450
150 LOCATE 0,0
160 PRINT R2$;L$;R3$
170 PRINT R6$;" * FLIPFLOPLAND *** Instructie * ";R6$
180 PRINT R4$;L$;R5$
190 PRINT
200 PRINT" U bent voor een periode van maximaal"
210 PRINT" 14 jaar STADHOUDER van FLIPFLOPLAND."
220 PRINT" Door goed beleid moet U proberen Uw"
230 PRINT" volk op Uw hand te houden."
240 PRINT" Daarbij dient U ook Uw bezittingen"
250 PRINT" verder uit te breiden."
260 PRINT" U dient er rekening mee te houden"
270 PRINT" dat elke FLIP-FLOPPER zo'n 5 Tins"
280 PRINT" graan verorbert. Ten overvloede"
290 PRINT" wijs ik U erop dat de prijzen van"
300 PRINT" houder niet beïnvloedbaar zijn."
310 PRINT" De prijzen worden door de conjunctuur"
320 PRINT" bepaald."
330 LOCATE 27,18:PRINT "<Uw dienaar>"
340 PRINT R2$;L$;R3$
350 PRINT R6$;" GELEZEN, DRUK DAN <RETURN> IN. ";R6$
360 PRINT R4$;L$;R5$
370 GOSUB 4060
380 A$ = INKEY$
390 IF A$ <> CHR$(13) GOTO 380
400 RETURN
410 :
420 REM *** INITIALIZEREN ***
430 A = 14
: `TIJDSCYCLUS
440 A1 = 360 / A : `EEN FASE
CONJ
450 C = 0 : `BEDRAG
TOT. RATBESTR310 H = 0
460 B1 = 0
: `GROEIPERCENTAGE
470 D = 0 : `DODEN
480 Q = 0 : `OPBREGST
490 B = INT(RND(-TIME) * 500+500) : `AANTAL
FIPFLOPPERS
500 X = INT(RND(-TIME) * 5 + 7)
510 V = INT(B * X * (RND(-TIME) + 1))
: `BEGINVOORR. GRAAN
520 M = INT(V / 8000 + 1)
: `AANT.MARKTPLAATSEN
530 J = INT(RND(-TIME) * 900 + 10)
: `STARTJAAR
540 T = INT(RND(-TIME) * 7 +1) : `TELLER
MAGERE/VETTE
550 R = INT(RND(-TIME) * B) : `AANTAL

```

```

RATTEN
560 IF R < 5 THEN 550
570 G = INT(RND(-TIME) * B/2 + B/2) * 4           : 'HECTAREN
GROND
580 A(7) = G
: 'BEGINSTAND ONTHOUDEN
590 W = INT(RND(-TIME) * 100 + 1)                 : 'WEER JAAR
N
600 W1 = INT(RND(-TIME) * 100 + 1)                : 'WEER JAAR
N-1
610 P1 = INT(RND(-TIME) * 50 + 50)                : 'PRIJS
GROND/HA
620 P2 = INT(RND(-TIME) * 600+300)                : 'PRIJS
MARKTPLAATS
630 GOSUB 690
640 D$ = ""
650 E$ = ""
660 RETURN
670 :
680 REM *** VARIABELEN MET EEN WAARDE VULLEN ***
690 G1 = 1                                         : '1 TIN
GEZAAID
700 V1 = -1
: 'ONDERVOEDING
710 FOR I = 1 TO 6
720 A(I) = 0
: 'STUURVARIABELEN
730 NEXT I
740 RETURN
750 :
760 REM *** BEREKENING EFFECTEN ***
770 D$ = ""
780 E$ = ""
790 J1 = J1 + 1                                    : 'JAAR
OPHOGEN
800 REM ** PRIJSONTWIKKELING **
810 REM * VETTE/MAGERE JAREN CYCLUS *
820 T = T + 1
830 IF T <= A THEN 850
840 T = 1                                          : 'BEGIN
CYCLUS
850 X = A1 * T                                     : 'WAARDE
CYCLUS
860 F = SIN( X * (3.1457 / 180))                   : 'FASE IN
CYCLUS
870 REM * PRIJS-ONTWIKKELING LANDBOUWGROND *
880 X = 1 + .5 * F + .5 * (W1 - 35)/100
: 'PRIJSWIJZIGING
890 P = P1                                         : 'OUDE P
ONTHOUDEN
900 P1 = P1 * .45 + P1 * X * .35 + P1 * (RND(-TIME) + .4) *
.2
910 P1 = INT((P + P1)/2)                           : 'NIEUWE
PRIJS
920 IF P1 > 0 THEN 950
930 P1 = INT((P + (RND(-TIME) * 50 + 1)) / 2)
940 REM * PRIJSONTWIKKELING BOUW MARKT *

```

```

950 P = P2                                     : `ONTHOUDEN
OUDE P.
960 P2 = P2*.45 + P2*X*.35 + P2*(RND(-TIME)+.4)*.2
970 X = RND(-TIME)                             : `RANDOM
VARIABELE
980 P2 = INT(X * P + (1-X) * P2)               : `NIEUWE
PRIJS
990 IF P2 > 0 THEN 1020
1000 P2 = INT(RND(-TIME) * 500 + 300)         : `NIEUWE
PRIJS
1010 REM ** WEER IN JAAR J **
1020 W1 = W                                     : `WEER
VORIG JAAR
1030 W = INT(RND(-TIME) * 100 + 1)           : `WEER DIT
JAAR
1040 F$ = "**DROOG**"
1050 IF W < 40 THEN 1110
1060 F$ = "**REDELIJK**"
1070 IF W < 70 THEN 1110
1080 F$ = "***GOED***"
1090 REM ** OPBRENGST GROND **
1100 REM * DRIESLAGSTELSEL *
1110 X = G1 - G * 2 / 3                         : `NIET TE
VEEL GROND
1120 IF X >= 0 THEN 1140
1130 G1 = G1 - INT(RND(-TIME) * G1 )          : `GROND
UITPUTTING
1140 Q = INT(RND(-TIME)*G1/3*10)+G1*2/3*10
: `OPBRENGST
1150 REM * INVLOED WEER *
1160 X = (W1 * .2 + .8 * W) / 100
: `GROEIFACTOR
1170 IF X > .4 THEN 1190
1180 X = X - .4
1190 Q = INT( Q * (1 + (X - .5)))             : `OPBRENGST
1200 REM ** RATTEN **
1210 REM * AANWAS RATTEN *
1220 X = (RND(-TIME) + V/(M * 8000) + Q / 10 / G1) / 3
1230 REM * RATTEN EXPLOESIE *
1240 IF V = 0 THEN 1280
1250 IF C/V > .02 THEN 1280
1260 IF RND(-TIME) > .25 THEN 1280
1270 X = X + RND(-TIME)
1280 R1 = INT( R/2 * X )                       : `AANWAS
RATTEN
1290 R = R + R1 / 2                           : `HELF
JONGE RATTEN
1300 REM * SCHADE VOORRAAD *
1310 X = (RND(-TIME) * R)                     : `%
ILVLOED
1320 V = V - INT(RND(-TIME) * X * 3)
1330 REM * SCHADE GEWAS *
1340 X = R - X
1350 Q = Q - INT(RND(-TIME) * X * 3)
1360 IF Q >= 0 THEN 1380
1370 Q = 0
1380 IF V >= 0 THEN 1400

```



```

1390 V = 0
1400 R = INT(R + R1 / 2)           : 'POPULATIE
RATTEN
1410 V = INT(V + Q)               : 'VOORRAAD
1420 GOSUB 1720
1430 REM ** BEVOLKING-ONTWIKKELING **
1440 REM * OVERLEDEN OUDERDOM *
1450 X = INT(RND(-TIME) * .1 * B)
1460 B1 = B1 - X
1470 B = B - X
1480 D = D + X
1490 REM * GEBOORTEN *
1500 X = INT((RND(-TIME)*.7+.3)*B/5*(1 + W1/100 -
.400003)*(1+V1))
1510 B = B + X
: 'BEVOLKINGSAANWAS
1520 B1 = B1 + X
1530 REM * EPIDEMIE *
1540 IF V1 < -.2 THEN 1570
1550 X = (W + W1)/2
1560 IF ((R*RND(-TIME))/B <=1) OR (X > 25) THEN 1640
1570 D$ = " ER IS EEN EPIDEMIE UITGEBROKEN! "
1580 COLOR 6,13,13
1590 X = INT(RND(-TIME)*70 + 30)/100
: 'DOODSPERCENTAGE
1600 D = (INT(X * B))             : 'AANTAL
DODEN
1610 B1 = B1 - D
1620 B = B - D
1630 REM * BEPALING GROEIPERCENTAGE *
1640 B2 = INT(B1/(B- B1)*1000+.5)/10 : 'GROEIPERC
BEVOLK
1650 IF B2 >= -.1 THEN 1680
1660 E$ = "DE BEVOLKING LOOPT TERUG !!"
1670 H = 2
: 'STUURVARIABELE
1680 GOSUB 690
1690 RETURN
1700 :
1710 REM *** VOLDOENDE RUIJTE MARKT ***
1720 IF V/M <= 8000 THEN 1770
1730 X = V - M * 8000
1740 V = M * 8000
1750 R = INT(R + RND(-TIME) * X/5) : 'RATTEN
EXTRA
1760 D$ = "EEN DEEL VAN OPBRENGST VERRROT BUITEN!"
1770 RETURN
1780 :
1790 REM *** GENOEG GRAAN ***
1800 H = 0
: 'STUURVARIABELE
1810 IF Y <= V THEN 1900
1820 FOR FR = 1 TO 10
1830 LOCATE 5,14:PRINT SPACE$(33)
1840 FOR T = 1 TO 50:NEXT T:BEEP
1850 LOCATE 5,14:PRINT"***ZOVEEL GRAAN IS ER NIET!***"
1860 FOR T = 1 TO 50:NEXT T

```

```

1870 NEXT FR
1880 LOCATE 26,21:PRINT SPACE$(6)
1890 H = 1
1900 RETURN
1910 :
1920 REM *** GENOEG GROND ***
1930 H = 0
: `STUURVARIABELE
1940 IF X <= G THEN 1980
1950 LOCATE 1,14:PRINT"**** ZOVEEL GROND BEZIT U NIET!
****"
1960 H = 1
1970 PLAY"S1M300005CDE", "S1M300006CDE", "S1M400007CDE"
1980 RETURN
1990 :
2000 REM *** RATTEN-CONTROLE = 5 ***
2010 LOCATE 1,21:PRINT "HOEVEEL TINS VOOR BESTRIJDNG";
2020 INPUT Y
2030 IF Y < 0 THEN 2010
2040 GOSUB 1800
2050 IF H <> 0 THEN 2010
2060 V = V - Y
2070 REM ** RATTENAFNAME I.V.M. BESTRIJDING **
2080 C = C + Y
: `TOT.BESTEED RAT
2090 R = R - INT((RND(-TIME) * .3 + .7 * Y/V) * R)
2100 IF R/M > 60 THEN 2120
2110 R = INT(RND(-TIME) * 60 ) * M : `MIN.
RATTEN
2120 A(5) = 1 : `VRAAG
GEBRUIKT
2130 V = V - Y : `AANPASS
GRAAN
2140 RETURN
2150 :
2160 REM *** GROND KOPEN = 1 ***
2170 LOCATE 1,21:PRINT "HOEVEEL HA WILT U KOPEN";
2180 INPUT X
2190 IF X < 0 THEN 2170
2200 IF X = 0 THEN 2310
2210 Y = X * P1
2220 GOSUB 1800
2230 IF H <> 0 THEN 2170
2240 REM ** TRANSACTIE **
2250 G = G + X : `GROND
TOEVOEGEN
2260 V = V - Y : `GROND
BETALEN
2270 REM ** AANTAL BEWONERS PER HA **
2280 Y = X * INT(RND(-TIME) * 4 + 1)
: `BEWONERS/HA
2290 B1 = B1 + Y
: `BEVOLK.TOENAME
2300 B = B + Y : `BEWONERS
BIJ
2310 A(1) = 1 : `VRAAG
GEBRUIKT

```

```

2320 RETURN
2330 :
2340 REM *** GROND VERKOPEN = 2 ***
2350 LOCATE 1,21:PRINT "HOEVEEL GROND WILT U VERKOPEN";
2360 INPUT X
2370 IF X < 0 THEN 2350
2380 IF X = 0 THEN 2450
2390 GOSUB 1930
2400 IF H <> 0 THEN 2350
2410 REM ** TRANSACTIE **
2420 G = G - X
: `GRONDVERKOOP
2430 V = X * P1 + V
2440 GOSUB 1720
2450 A(2) = 1 : `VRAAG
GEBRUIKT
2460 RETURN
2470 :
2480 REM *** MARKTPLAATS = 3 ***
2490 LOCATE 1,21:PRINT "MARKTPLAATSEN BOUWEN 0-1-2-3";
2500 INPUT X
2510 IF (X < 0) OR (X > 3) THEN 2490
2520 REM ** GENOEG TINS JA/NEE **
2530 Y = X * P2 * 10
2540 GOSUB 1800
2550 IF H <> 0 GOTO 2490
2560 REM ** TRANSACTIE **
2570 V = V - Y : `AANPASSEN
VOORR
2580 M = M + X : `MARKT
ERBIJ
2590 A(3) = 1 : `VRAAG
GEBRUIKT
2600 RETURN
2610 :
2620 REM *** VOEDSEL FLIPFLOPPERS = 4 ***
2630 LOCATE 1,21:PRINT "HOEVEEL TINNEN WILT U GEVEN";
2640 INPUT Y
2650 IF Y < 0 GOTO 2630
2660 REM ** GENOEG GRAAN **
2670 GOSUB 1800
2680 IF H <> 0 THEN 2630
2690 REM ** ONDERVOEDING ? **
2700 V1 = (Y - B * 5)/(B * 5) : `%
ONDERVOED
2710 IF V1 => 0 THEN 2750
2720 D$ = "* DIT KAN LEIDEN TOT ONDERVOEDING. *"
2730 IF V1 > -.3 THEN 2750
2740 V1 = V1 - RND(-TIME)
2750 V = V - Y : `VOORR
AANP
2760 A(4) = 1 : `VRAAG
BEANTWOORD
2770 RETURN
2780 :
2790 REM *** LAND INZAAIEN = 6 ***
2800 LOCATE 1,21:PRINT "HOEVEEL HA WILT U INZAAIEN";

```



```

2810 INPUT X
2820 IF X < 0 THEN 2800
2830 GOSUB 1930
2840 IF H <> 0 THEN 2800
2850 REM ** VOLDOENDE ZAAIGOED **
2860 Y = INT(X)                                :`1 TIN PER
HA
2870 GOSUB 1800
2880 IF H <> 0 THEN 2800
2890 G1 = X                                     :`IN TE
ZAAIEN GROND
2900 V = V - Y
2910 A(6) = 1                                  :`VRAAG
BEANTWOORD
2920 RETURN
2930 :
2940 REM *** BESLISSING ***
2950 LOCATE 0,15:PRINT R2$;L$;R3$
2960 PRINT R6$;" 1..KOPEN                      2..VERKOPEN ";R6$
2970 PRINT R6$;" 3..MARKT BOUWEN              4..VOEDSEL ";R6$
2980 PRINT R6$;" 5..RATTENBESTRIJDING        6..ZAAIEN ";R6$
2990 PRINT R6$;" M..MEER INFORMATIE          0..STOP ";R6$
3000 LOCATE 0,20:PRINT R4$;L$;R5$
3010 K$ = INKEY$
3020 IF K$ = "M" THEN GOTO 3830
3030 IF K$ = "" GOTO 3010
3040 K = VAL(K$)
3050 IF (K <= 0) OR (K > 6) THEN 3080
3060 IF A(K) = 1 THEN 3080
3070 ON K GOSUB 2170 , 2350 , 2490 , 2630 , 2010 , 2800
3080 RETURN
3090 :
3100 REM *** UITVOER ***
3110 CLS:VL = 1                                :`SCHERM
WISSEN
3120 PRINT R2$;L$;R3$
3130 PRINT R6$;" * FLIPFLOPLAND ***** JAAR : ";R6$
3140 PRINT R4$;L$;R5$
3150 COLOR 4,11,11
3160 LOCATE 0,4
3170 PRINT" Bevolking                          :`"
3180 PRINT" Aantal doden                       :`"
3190 PRINT" Ratten-populatie                   :`"
3200 PRINT
3210 PRINT" Grondbezit in HA                    :`"
3220 PRINT" Voorraad tins graan                :`"
3230 PRINT" Aantal marktplaatsen              :`"
3240 LOCATE 34,1:PRINT USING"##";J1
3250 FOR I = 12 TO 14
3260 LOCATE 1,I:PRINT SPACE$(38)
3270 NEXT I
3280 LOCATE 0,21:PRINT SPACE$(39)
3290 LOCATE 27,4:PRINT USING"#####";B
3300 LOCATE 27,5:PRINT USING"#####";D
3310 LOCATE 27,6:PRINT USING"#####";R
3320 LOCATE 27,8:PRINT USING"#####";G
3330 LOCATE 27,9:PRINT USING"#####";V

```

```

3340 LOCATE 27,10:PRINT USING"#####";M
3350 IF D$ = "" GOTO 3420
3360 FOR FR = 1 TO 10
3370 LOCATE 1,12:PRINT SPACE$(38)
3380 FOR T = 1 TO 50:NEXT T:PLAY"S1M200005D"
3390 LOCATE 1,12:PRINT D$
3400 FOR T = 1 TO 50:NEXT T:PLAY"S1M200006D"
3410 NEXT FR:D$ = ""
3420 LOCATE 1,14:PRINT E$
3430 IF DD$ <> "" THEN DD$ = "":GOTO 3010
3440 RETURN
3450 L$ = LEFT$(S1$,74)
3460 RETURN
3470 REM **** STUURMODULE ****
3480 REM *** INSTRUCTIES ***
3490 GOSUB 130
3500 REM *** INITIALIZEREN ***
3510 GOSUB 430
3520 FOR N = 1 TO 14
3530 REM *** UITVOER ***
3540 IF VL = 0 THEN GOSUB 3110 ELSE GOSUB 3240
3550 REM *** BESLISSING ***
3560 GOSUB 2950
3570 IF K <> 0 THEN 3540
3580 REM *** BEREKENING ***
3590 GOSUB 770
3600 IF H <> 2 THEN 3630
3610 N = 14
3620 E$ = " UW AUTORITEIT WORDT NIET AANVAARD. "
3630 NEXT N
3640 REM *** UITVOER ***
3650 GOSUB 3240
3660 REM *** RESULTAAT ***
3670 IF H = 2 THEN 3820
3680 X = INT(G / A(7) * 100 + .5) /10
3690 D$ = "***** EEN KNAP HEERSER *****"
3700 Z$ = " HET GRONDBEZIT IS VERGROOT "
3710 COLOR 1,3,3
3720 IF X > 10 THEN 3800
3730 D$ = "***** EEN MATIG HEERSER *****"
3740 Z$ = " HET GRONDBEZIT IS VERGROOT "
3750 COLOR 4,5,5
3760 IF X > 0 THEN 3800
3770 D$ = "EEN SLECHT HEERSER. HET GRONDBEZIT IS
VERKLEIND"
3780 Z$ = " HET GRONDBEZIT IS VERKLEIND "
3790 COLOR 15,6,6
3800 LOCATE 1,12:PRINT D$
3810 LOCATE 0,14:PRINT Z$;"MET";ABS(X);"%
3820 LOCATE 0,20:END
3830 REM *** MEER INFORMATIE SCHERM ***
3840 FOR I = 3 TO 14
3850 LOCATE 1,I:PRINT SPACE$(38)
3860 NEXT I
3870 COLOR 1,7,7
3880 LOCATE 0,4
3890 PRINT" HET WEER : ";F$


```

```

3900 PRINT
3910 PRINT" OPBRENGST           : " USING"#####";Q
3920 PRINT" GROEI-PERCENTAGE    : " USING"#####";B2
3930 PRINT
3940 PRINT" PRIJS PER HECTAR     : " USING"#####";P1
3950 PRINT" BOUWPRIJS           : " USING"#####";P2*10
3960 PRINT
3970 PRINT" VOORRAAD TINS GRAAN : " USING"#####";V
3980 PRINT" AANTAL MARKTPLAATSEN: " USING"#####";M
3990 DD$ = INKEY$
4000 IF DD$ <> "M" GOTO 3990
4010 FOR I = 3 TO 14
4020 LOCATE 1,I:PRINT SPACE$(38)
4030 NEXT I
4040 GOTO 3150
4050 REM *** MUZIEK ***
4060 PLAY"V15T125","V9T125","V9T125"
4070 PLAY"O4G8F+8DG8A8O5CO4B8A8G8B8A8F+8DG8F+8DG8A8O5CO4B8A8
G8B8A8F+8G","O3BDO4CO3DBDO4CO3DBDO4CO3DBDO4CO3B","O3GRARGRAR
GRARGRG2"
4080 RETURN

```

Voorbeeld



De vele computerspellen in dit boek maken optimaal gebruik van de mogelijkheden die MSX-computers bieden. Zo zijn bijna alle spellen voorzien van muziek, geluid en fraaie graphics. De listings kunnen op elke MSX-computer worden ingetypt en variëren van eenvoudige korte programma's tot behoorlijk uitgebreide spellen met een hogere moeilijkheidsgraad. Ondanks het luchtige karakter van deze programma's hebben ze de serieuze programmeur meer te bieden. Er is door de auteur naar gestreefd om een zo duidelijk mogelijke structuur in de programma's aan te brengen. Men is daardoor in staat om snel de werking ervan te doorzien. Na enige oefening zijn de gegeven suggesties voor uitbreiding of aanpassing goed uit te voeren.